
イントロダクション

この講義ノートは 1994 年に Xavier Leroy が高等師範学校で基礎・応用数学及び情報
科学の修士課程 1 年生に対して行ったシステムプログラミングの講義が元になってい
ます。初期の版では Caml-Light [1] 言語が使われていました。

2003年から 2006年にエコール・ポリテクニークで開かれた情報科学の修士課程学
生向け講義のために、Didier Rémy はノートを改変し、OCaml[2] 言語が使われるよう
になりました。この期間に、Gilles Roussel, Fabrice Le Fessant と Maxence Guesdon
は講義を助けるとともに、このノートにも貢献しました。この新しい版では要素の追

加や更新が行われています。

最初の版からの約 10 年間のうちに、扱われる数字の桁はコンマ 1 つ分大きくな
りました。また web は大きく発達し、1994 年当時には先駆的だった httpリレーの
例は現在では平凡なものです。そしてなんといってもこの間に OCaml言語は成熟し、
Unison [18] のような実際のシステムアプリケーションに使われるようになりました。
伝統的に Unix システムプログラミングは C で行われなければならないという風

潮があります。しかしこのコースで私達はより高レベルな言語 — 具体的には OCaml
— のほうが Unix システムプログラミングの基礎を説明するのに適していることを発
見しました。

OCamlの Unix システムコールに対するインターフェースはより抽象的です。C
では全てが整数とビットフィールドに変換されますが、OCamlでは ML の型システ
ムを使うことで、システムコールの引数と返り値が明確になります。そのため引数と

返り値がどのようにデコード/エンコードされるかを説明する必要がなく、システム
コールの意味を説明するのが簡単になります。(例えば 39 ページにある wait システ

ムコールの説明を見てみてください。)
加えて OCamlは静的な型システムを持ち、基本型が明確なので、C よりも安全に

プログラムを書けます。熟練した C プログラマーはこの利点を必要のない贅沢である
とみなすかもしれませんが、このコースが対象とする熟練していないプログラマーに

とっては重要なことです。

システムプログラミングに関するこのノートの二つ目の目標は、定理証明やコン

パイラ、記号計算といったOCamlの一般的な応用とは異なる分野における OCamlの
利用例を見せることです。OCamlの持つ強固な命令的カーネルに加えてパラメトリッ
ク多相や高階関数、例外といった C にない優れた機能のおかげで、このノートで行
われた試みの結果は上々です。他にも関数型プログラミングと命令型プログラミング

を互いに排他的にではなく組み合わせて使うことで、同じプログラムの中にオペレー

ティングシステムとの優れたインターフェースと複雑な記号計算を共存させることが

可能なことをこのノートは示しています。

このノートは読者が OCamlと Unix のシェルコマンドを知っていることを仮定し
ます。OCamlに関する疑問は OCaml System documentation [2] を、Unix について
の疑問は Unix manual のセクション 1 や [5, 6] などの入門書を参考にしてください。
このノートは Unix システムのプログラミング上のインターフェースしか示すこと

しかせず、その実装や内部構造には触れません。bsd 4.3 の内部構造は [8] で、System
v の内部構造は [9] で説明されています。Tanenbaum の本 [13, 14] はネットワークと
オペレーティングシステムの構造に関する全体像が示されています。

このノートで説明される Unixインターフェースは http://caml.inria.fr/ocaml/

でフリーソフトウェアとして配布されている OCaml System の一部です。

1

http://caml.inria.fr/ocaml/


1 基礎

1.1 Sys モジュールと Unix モジュール

OCamlからシステムにアクセスするときに使われる関数は Sys と Unix の二つのモ

ジュールにまとめられています。一つ目の Sys モジュールは OCamlが実行される
Unix およびその他のオペレーティングシステムで一般的な関数を含みます。二つ目
の Unix モジュールは Unix に特有なものを全て含みます。
これ以降 Sys と Unix モジュールにある識別子はどちらのモジュールのものかを

示すことなく使うことにします。つまり open Sys および open Unix を実行した状態

であるということです。完全な例を示すときには、open Sys と open Unix を明示的

に書くことにします。

Sys および Unix モジュールは Pervasives モジュールに定義されている変数

を上書きし、元の定義を隠してしまうことがあるので注意してください。例えば、

Pervasives.stdin と Unix.stdin は別物です。隠された定義にはプリフィックスを

つけることでアクセスできます。

Unix ライブラリを使う OCamlのプログラムをコンパイルするには、次のように
します:
ocamlc -o prog unix.cma mod1.ml mod2.ml mod3.ml

ここで prog というプログラムは mod1, mod2 そして mod3 という三つのモジュールか

ら成ります。モジュールは別々にコンパイルすることもできます:
ocamlc -c mod1.ml

ocamlc -c mod2.ml

ocamlc -c mod3.ml

この場合、次のようにしてリンクします:
ocamlc -o prog unix.cma mod1.cmo mod2.cmo mod3.cmo

両方の例において、引数 unix.cma は OCamlで書かれた Unix ライブラリを表します。

バイトコードコンパイラではなくネイティブコードコンパイラを使うには、ocamlc を

ocamlopt に、unix.cma を unix.cmxa に置き換えてください。

コンパイルツール ocamlbuild を使っている場合、次の内容を _tags ファイルに

追加してください:
<prog.{native,byte}> : use_unix

“toplevel”を言われる対話環境から Unix システムにアクセスすることもできます。実
行している環境が C ライブラリの動的リンクに対応している場合、OCamlトップレ
ベルを起動して次のディレクティブを入力します:
#load "unix.cma";;

動的リンクに対応していない場合、システム関数がプリロードされた対話環境を作る

必要があります:
ocamlmktop -o ocamlunix unix.cma

このトップレベルは次のコマンドで起動できます:
./ocamlunix

プログラムを呼ぶためのインターフェース

1.2 プログラムを呼ぶためのインターフェース

シェル (コマンドインタープリタ) からプログラムを実行する場合、シェルは 引数 と
環境 を実行するプログラムに渡します。引数とはコマンドライン上でプログラムの

名前の後ろに続く語です。環境とは variable=value の形をした文字列の集まりであ

り、環境変数のバインディングを表します。このバインディングは csh では setenv

var=val で、sh では var=val; export varでセットされます。

プログラムに渡された引数は文字列の配列 Sys.argv に格納されます:

val argv : string array

プログラムの環境は Unix.environment 関数で取得できます:

val environment : unit -> string array

Sys.getenv 関数を使えば環境をより簡単に検索できます。

val getenv : string -> string

Sys.getenv v は v という環境変数に結び付けられた値を返します。環境変数が見つ

からなかった場合は Not_found 例外を出します。

例 最初の例として、引数を出力する echo プログラムを示します。これは同じ名前

の Unix コマンドと同じ動作です。

let echo () =

let len = Array.length Sys.argv in

if len > 1 then

begin

print_string Sys.argv.(1);

for i = 2 to len - 1 do

print_char ' ';

print_string Sys.argv.(i);

done;

print_newline ();

end;;

echo ();;

* * *
プログラムは exit を呼ぶことで任意の場所で終了させることができます。

val exit : int -> 'a

引数は呼び出し元のプログラムに送られる返り値です。問題のない場合には 0 を、エ
ラーが起こった場合には 0でない値を返すという慣習があります。プログラムの実行
結果が条件として使われた場合、sh シェルは返り値 0 をブール値 “true” に、0 でな
い全ての返り値を “false” として解釈します。
プログラムが全ての式を実行し終わって終了する場合、そのプログラムは暗黙的

に exit 0 を呼びます。プログラムが補足されない例外によって途中で実行を終了す

る場合、そのプログラムは暗黙的に exit 2 を呼びます。

exit 関数は呼ばれたときに書き込み用にオープンされている全てのチャンネルの

バッファをフラッシュします。at_exit 関数を使うと、プログラムが終了するときに

これ以外の動作をさせることができます。

エラー処理

val at_exit : (unit -> unit) -> unit

最後に登録された関数が最初に実行されます。at_exit 関数を使って登録された関数

は登録を解除することができませんが、これが本質的な制限になることはありません。

グローバル変数を使って実行を変えることができるからです。

1.3 エラー処理

他に明示されていない限り、Unix モジュールの全ての関数はエラーが起きたときに

Unix_error 例外を出します。

exception Unix_error of error * string * string

Unix_error例外の第二引数はエラーが起こったシステムコールの名前です。第三引数は

エラーが起こったオブジェクトの名前を (可能な場合には)表します。例えば、ファイル
の名前を引数として取るシステムコールの場合には、このファイルの名前が Unix_error

の第三引数となります。最後に、第一引数はエラーの種類を表すエラーコードを表し

ます。エラーコードは error というヴァリアント型に属しています。

type error = E2BIG | EACCES | EAGAIN | ... | EUNKNOWNERR of int

この型のコンストラクタには posix で定義されるエラーが同じ名前と意味ですべて含
まれ、加えて unix98, bsd のエラーの一部が含まれます。その他の全てのエラーは
EUNKOWNERR というコンストラクタになります。

例外が発生したとき、try によって補足されないエラーはプログラムの一番上まで

上がっていき、プログラムを実行の途中で終了させます。小さいアプリケーションで

は予見できないエラーを致命的なものとみなすことは良い習慣です。しかしその場合、

エラーを分かりやすく表示することが望ましいです。エラーを分かりやすく表示する

ために、Unix モジュールには handle_unix_error 関数があります:

val handle_unix_error : ('a -> 'b) -> 'a -> 'b

handle_unix_error f x はまず引数 x を関数 f に適用します。この適用が Unix_error

を出した場合、エラーを説明するメッセージが表示され、exit 2 によってプログラム

は終了します。次のプログラムは典型的な使用例です:

handle_unix_error prog ();;

ここで関数 prog : unit -> unit がプログラム本体を実行します。参考のために、

handle_unix_error の実装を以下に示します。

1 open Unix;;

2 let handle_unix_error f arg =

3 try

4 f arg

5 with Unix_error(err, fun_name, arg) ->

6 prerr_string Sys.argv.(0);

7 prerr_string ": \"";

8 prerr_string fun_name;

9 prerr_string "\" failed";

10 if String.length arg > 0 then begin

11 prerr_string " on \"";

12 prerr_string arg;

13 prerr_string "\""

ライブラリ関数

14 end;

15 prerr_string ": ";

16 prerr_endline (error_message err);

17 exit 2;;

prerr_xxx の形をした関数は基本的には print_xxx 関数と同じ動作をしますが、書き

込み先は stdout ではなく stderr となります。

error -> string 型をもつ error_message は引数の番号が表すエラーを説明する

メッセージを返します (第 16 行)。プログラムに渡される第 0 引数 Sys.argv.(0) に

はプログラムを起動するのに使われたコマンドが格納されます (第 6 行)。
handle_unix_error 関数がプログラムの実行を終了させるような致命的なエラー

を処理します。OCamlを使うことの利点は全てのエラーを明示的に処理することが求
められ、エラーが発生するとプログラムの実行が終了することになるトップレベルで

さえこれが求められることです。実際システムコールによるどんなエラーも例外を発

生させるので、プログラムの実行は中断され、例外は明示的に補足・処理されるまで

上に登る事になります。これによってプログラムが不整合状態で実行が続くことを防

ぐことができます。

Unix_error 型のエラーにはもちろんパターンマッチを使うことができます。次の

関数はこれからよく目にすることになります:

let rec restart_on_EINTR f x =

try f x with Unix_error (EINTR, _, _) -> restart_on_EINTR f x

このコードは関数を実行してもし中断された場合にはもう一度繰り返すという処理を

行います (4.5 節を参照)。

1.4 ライブラリ関数

これから例を通して見ていくことですが、システムプログラミングでは同じパターン

の処理が繰り返し出てきます。アプリケーションのコードが本質的な部分だけを含む

ように、共通する処理をまとめたライブラリ関数を定義しておくことが望ましいです。

自分で書いて自分で実行するプログラムではどんなエラーが出て、そのうちどれ

が実行を終了させるような致命的なエラーかが分かるものですが、ライブラリ関数の

場合には実行されるコンテキストが分からないのでどれが致命的なエラーなのかは通

常分かりません。かといって全てのエラーが致命的であると仮定することもできませ

ん。そのためプログラムを止めるのか、それとも無視するのか呼び出し元に判断させ

るために、エラーを呼び出し元に伝えることが必要になります。

しかし、ライブラリ関数を普通に実装すると発生したエラーをそのまま呼び出し元

に伝えることができません。システムを整合状態に置くことが求められるためです。例

えばファイルを開いてそのファイルディスクリプタを使って操作を行うライブラリ関

数は、ファイルへの操作でエラーが生じた場合を含めた全ての場合においてファイル

ディスクリプタを閉じる処理を行う必要があります。ファイルディスクリプタがリー

クしてファイルディスクリプタを使いきってしまうことを防ぐためです。

ファイルに対する操作を引数として受け取る場合もあります。この場合、いつど

のように操作が失敗するかを知ることは (呼び出した側でなければ) できません。そ
のため操作の本体は “最終処理” コードで守ることが必要になります。このコードは
関数が例外を出したかどうかにかかわらず、関数が帰る直前に実行されます。

try . . . finalize 構文は OCamlにビルトインでは用意されていませんが、簡単に
定義することができます 1。

1構文がビルトインで用意されれば関数を定義して使うよりも便利でしょう。

ライブラリ関数

let try_finalize f x finally y =

let res = try f x with exn -> finally y; raise exn in

finally y;

res

この関数はメインの処理 f と最終処理 finally およびそれらの引数 x と y を受け取

ります。最初にプログラムの本体 f x が実行され、その結果は最終処理 finally y が

実行されてから返されます。プログラムの実行で例外 exn が起こった場合、最終処理

が実行されてからもう一度 exn を出します。メインの処理と最終処理の両方が失敗し

た場合、最終処理の例外が出されます (メインの処理の例外が出されるようにするこ
ともできます)。

ノート これからこのコースでは例でよく使う try_finalize などの関数をまとめた

補助ライブラリ Misc を使います。必要に応じてライブラリの関数を紹介するほか、イ

ンターフェースは付録にあります。このコースに出てくる例をコンパイルするには、

Misc モジュールの定義をまとめておく必要があります。

Misc モジュールにはこのコースでは直接使用しない可視化のための関数も含まれ

ています。これらの関数は Unix ライブラリを強化するためのもので、いくつかの関

数の振る舞いを上書きします。そのため Misc ライブラリを使う場合は Unix の後に読

み込まれる必要があります。

例 このコースにはたくさんの例が含まれています。これらはバージョン 4.05.0 の

OCamlでコンパイルされることを確認しています。古いバージョンではプログラムを
若干改変する必要があります。

例には二つの種類があります: とても一般的で再利用が可能な “ライブラリ関数”
と小さなアプリケーションです。これら二つを区別することは重要です。ライブラリ関

数の場合には実行時のコンテキストをできるだけ一般的なものと仮定して、インター

フェースを熟慮し、全ての特殊ケースを扱うようにします。一方小さなアプリケーショ

ンの場合には、多くのエラーは致命的なものであり、プログラムの実行を停止させま

す。そのためエラーが起きた時にはその原因を伝えるだけで十分であり、システムを

整合状態へと戻す処理は必要ありません。

2 ファイル

Unix において “ファイル” という言葉はいくつかのものを表します:

• 通常のファイル: テキストまたはバイナリ情報を含んだ有限のバイト列。“通常ファイ
ル” とも呼ばれる。

• ディレクトリ

• シンボリックリンク

• 特殊ファイル (デバイス): 主にコンピュータの周辺機器にアクセスするために使わ
れる。

• 名前付きパイプ

• 名前付き Unix ドメインソケット

ファイルという表現にはファイルが保持するデータだけではなく、その種類やアクセ

ス権限、最終更新日時といったファイルそのものに関するデータ (メタ属性と呼ばれ
ます) も含まれます。

2.1 ファイルシステム

大ざっぱにいって、ファイルシステムは木と考えることができます。根 (ルート)は /

で表され、枝は '\000' と / を除く文字列からなるファイルの名前でラベル付けされ

ます (ただし空白文字と印字できない文字は避けたほうが良いとされます) 。終端で
ないノードは ディレクトリ です: これらのノードは必ず二つの枝 . と .. を含み、そ

れぞれこのディレクトリそのものと親のディレクトリを表します。ディレクトリでな

いノードのことを ファイル と呼ぶことがありますが、木のどのノードもファイルで

あることを考えると、これは曖昧です。曖昧さを避けるために、このノートではこれ

らのことを 非ディレクトリファイル と呼ぶことにします。

木のノードはパスを使って表すことができます。パスの始点がファイル階層の頂

上である場合、そのパスは 絶対 です。一方始点がディレクトリである場合にはパス

は 相対 です。より正確に言うと、相対パス とはファイルの名前を / で区切った文字

列であり、絶対パス とは先頭に / のついた相対パスです。ここでは同じ文字 / が区

切り文字と根ノードという二つの意味で使われています。

Filename モジュールを使うとパスをポータブルに扱うことができます。例えば

concat は / という文字を与えることなく二つのパスを結合するので、他のオペレー

ティングシステム (windows では区切り文字は \ です) でも同じような動作をさせる
ことができます。Filename モジュールには current_dir_name と parent_dir_name が

あり、それぞれ . と .. という枝を表します。basename 関数と dirname 関数はパス p

を受け取ってそれぞれディレクトリ名 d と非ディレクトリファイル名 bを返します。

このとき p と d/p が表すファイルは同じになります。Filename モジュールの関数は

パスの操作だけを行うので、実際にそのパスが存在するかどうかは考慮しません。

ファイル階層は厳密には木ではありません。. と .. というディレクトリが自分自

身や上の階層のディレクトリを指しているからです。さらに、非ディレクトリファイ

ルは複数の親を持つことができます (ハードリンク と言います)。また他のファイルへ
のパスを保持する非ディレクトリファイルとみなすことができる シンボリックリンク

もあります。概念上は、シンボリックリンクの保持するパスは通常ファイルと同じよ

ファイル名とファイルディスクリプタ

1

7

?

/

bin

usr

tmp

ls

cd

cc

bin

lib

foo

bar

gcc

/usr

../gnu

• 逆向きのリンクは省略した。

• ハードリンク 7 の親は 2 と 6。

• シンボリックリンク 10 は 5 を
指す。

11 はどんなノードも指さない。

• 9 から 8 への等価なパス:
../usr/lib

./../usr/lib

foo/lib

Figure 1 — ファイル階層の例

うにその内容を読むことで取得できます。パスの途中でシンボリックリンクに当たっ

た場合、そのたびにパスをたどります。s が l へのシンボリックリンクならば、p/s/q

というパスは l が絶対パスのときは l/q を、相対パスのときは p/l/q を表します。

図 1 にファイル階層の例を示します。/tmp/bar というパスにあるシンボリックリ

ンク 11 は ../gnu という相対パスを指していますが、このファイルはこの段階では存

在していません。

一般的に、次の規則に従えばファイル階層の再帰的な探索は終了します:

• ディレクトリ . と .. を無視する。

• シンボリックリンクをたどらない。

シンボリックリンクをたどる場合には木ではなく一般のグラフを走査することになる

ので、たどったノードを覚えておかないとループを避けることができません。

それぞれのプロセスはワーキングディレクトリを持ちます。ワーキングディレク

トリは getcwd 関数で取得することができ、chdir で変えることができます。chroot p

を使えばファイル階層のビューを制限することができます。これによってディレクト

リ p が制限されたビューのルートになります。それ以降は絶対パスが新しいルート p

からのものとして解釈されます (新しいルートからの .. は p 自身になります)。

2.2 ファイル名とファイルディスクリプタ

ファイルにアクセスする方法は二つあります。一つ目はファイルシステム階層の ファ

イル名 (あるいは パス名) を利用する方法です。ハードリンクがあるので、全てのファ
イルは複数のファイル名を持つことができます。ファイル名は string 型の値です。例

えばシステムコール unlink, link, symlink そして rename はどれもファイル名を使い

ます。

val unlink : string -> unit

val link : string -> string -> unit

val symlink : string -> string -> unit

val rename : string -> string -> unit

以下のような効果を持ちます:

• unlink f はファイル f を削除する。Unix コマンド rm -f f と同じ。

• link f1 f2 は f2 という名前で f1 というファイルを指すハードリンクを作成する。

Unix コマンド ln f1 f2 と同じ。

• symlink f1 f2 は f2 という名前で f1 というファイルを指すシンボリックリンクを作

成する。Unix コマンド ln -s f1 f2 と同じ。

• rename f1 f2 はファイル f1 をファイル f2 にリネームする。Unix コマンド mv f1 f2

と同じ。

ファイルにアクセスする二つ目の方法はファイルディスクリプタを使うものです。

ファイルディスクリプタはファイルへのポインタであり、ファイルの名前の他にも現

在の読み込み/書き込み位置、アクセス権限 (読み込み/書き込みは可能か?)、入出力
を管理するためのフラグ (ブロッキング/ノンブロッキングや上書き/追記など)といっ
た情報を含みます。ファイルディスクリプタは抽象型 file_descr の値です。

ファイルへの名前を使ったアクセスはファイルディスクリプタを使ったアクセス

と独立しています。例えばあるファイルのファイルディスクリプタを取得したとき、

そのファイルを消去したりリネームしたりすることは可能ですが、そうした場合でも

ファイルディスクリプタは元のファイルを指したままです。

プログラムが実行されると 3 つのディスクリプタが確保され、Unix モジュールの

stdin, stdout, stderr という変数に割り当てられます。

val stdin : file_descr

val stdout : file_descr

val stderr : file_descr

これらのディスクリプタはそれぞれプロセスの標準入力、標準出力、標準エラー出力

に対応します。

プログラムがコマンドラインから実行されリダイレクトされることがない場合、三

つのディスクリプタは端末を表します。しかし例えば入力がシェルの cmd < f を使っ

てリダイレクトされている場合、cmd を実行している間はディスクリプタ stdin は f

というファイルを指します。同様に、cmd > f と cmd 2> f はコマンドの実行中にそれ

ぞれ stdout と stderr をファイル f に割り当てます。

2.3 ファイルのメタ属性、種類、権限

システムコール stat, lstat および fstat はファイルについてのメタ属性、つまりそ

のファイルの内容についてではなくそのノード自身についての情報を返します。例え

ばファイルの識別子、ファイルの種類、アクセス権限、最終更新日時といった情報な

どが含まれます

val stat : string -> stats

val lstat : string -> stats

val fstat : file_descr -> stats

システムコール stat と lstat はファイル名を引数として受け取りますが、fstat は

それまでに開かれたディスクリプタを受け取りそのディスクリプタが指しているファ

イルの情報を返します。stat と lstat はシンボリックリンクに対して異なった動作

ファイルのメタ属性、種類、権限

をします。lstat はシンボリックリンクそのものの情報を返しますが、stat はシンボ

リックリンクが指すファイルに関する情報を返します。これら 3 つのシステムコール
の返り値は stats 型のレコードです。そのフィールドは図 1 に説明されています。

識別子

ファイルはデバイス番号 st_dev (大抵の場合はファイルのあるディスクパーティショ
ンの番号) と inode 番号 st_ino で一意に識別できます。

所有者

ファイルは所有者 st_uid と所有グループ st_gid を持ちます。マシン上の全てのユー

ザとグループは通常 /etc/passwd と /etc/gourps に保存されています。ユーザとグ

ループを文字列からポータブルに検索するには getpwnam 関数と getgrnam 関数が使

えるほか、getpwuid 関数と getgrgid 関数を使うと id から検索できます。

val getpwnam : string -> passwd_entry

val getgrnam : string -> group_entry

val getpwuid : int -> passwd_entry

val getgrgid : int -> group_entry

プロセスを実行しているユーザの名前とそのユーザが属している全てのグループ

番号はそれぞれ getlogin と getgroups 関数で取得できます。

val getlogin : unit -> string

val getgroups : unit -> int array

chown 関数はファイル (第一引数) の所有者 (第二引数) と所有グループ (第三引
数) を変えます。ファイルディスクリプタを持っているならば、代わりに fchown 関

数が使えます。任意のファイルの所有者と所有グループを変更できる権限を持つのは

スーパーユーザだけです。

val chown : string -> int -> int -> unit

val fchown : file_descr -> int -> int -> unit

プログラムの実効 uid がファイルと等しいとき、あるいは実効 gid または実効ユーザ

の属する補助グループの一つがファイルの所有グループと等しい場合は、特権無しで

所有ユーザ/グループの変更が可能です。

アクセス権限

アクセス権限は整数の中にビット列として格納されており、file_permは intの別名に

過ぎません。そこには所有ユーザ、所有グループおよびその他のユーザの読み込み、書

き込みおよび実行のための権限を表すビットとスペシャルビットが保存されています:

Special︷ ︸︸ ︷
−−−

User︷ ︸︸ ︷
−−−

Group︷ ︸︸ ︷
−−−

Other︷ ︸︸ ︷
−−−︸ ︷︷ ︸

0oSUGO

ここでユーザ (User) 、グループ (Group)、その他 (Other) というフィールドの中に
は、読み込み (r)、書き込み (w) そして実行 (x) の権限がこの順番で保存されていま
す。ファイルの権限はこれらの権限を合わせたものであり、例を表 2 に示します。
非ディレクトリファイルに対して、読み込み、書き込みおよび実行の権限が意味

することは明らかです。ディレクトリに対する実行権限とはそのディレクトリに入る

フィールド名 説明

st_dev : int ファイルが保存されているデバイスの ID を表す。

st_ino : int パーティションにおけるファイルの ID (inode 番号と
呼ばれます) を表す。(st_dev, st_ino) の組でファイ

ルシステム内のファイルを識別できる。

st_kind : file_kind ファイルの種類を表す。file_kind 型は列挙型であり、

以下のコンストラクタを持つ:

S_REG 通常ファイル

S_DIR ディレクトリ

S_CHR キャラクタデバイス

S_BLK ブロックデバイス

S_LNK シンボリックリンク

S_FIFO 名前付きパイプ

S_SOCK ソケット

st_perm : int ファイルへのアクセス権限を表す。

st_nlink : int ファイルがディレクトリの場合はディレクトリ内の要

素の数を表す。ファイルがディレクトリ出ない場合、こ

のファイルに対するハードリンクの数を表す。

st_uid : int ファイルの所有ユーザを表す。

st_gid : int ファイルの所有グループを表す。

st_rdev : int ファイルが特殊ファイルの場合、ファイルに関連付け

られた周辺機器の ID を表す。

st_size : int ファイルのサイズ (バイト) を表す。

st_atime : int ファイルが最後にアクセスされた時間を、1970年 1月
1日深夜 0時 gmt からの経過秒数で表す。

st_mtime : int ファイルが最後に更新された日時を表す (単位は同上)。

st_ctime : int ファイルの状態が最後に更新された日時を表す。ファ

イルへの書き込み、アクセス権限の変更、所有ユーザ/
グループの変更、リンク数の変更などがファイルの状

態を変化させる。

Table 1 — stats 構造体のフィールド

ビット (8進表記) ls -l の表記 アクセス権限

0o100 --x------ 所有ユーザによって実行可能

0o200 -w------- 所有ユーザによって書き込み可能

0o400 r-------- 所有ユーザによって読み込み可能

0o10 -----x--- 所有グループのメンバーによって実行可能

0o20 ----w---- 所有グループのメンバーによって書き込み可能

0o40 ---r---- 所有グループのメンバーによって読み込み可能

0o1 --------x その他のユーザによって実行可能

0o2 -------w- その他のユーザによって書き込み可能

0o4 ------r-- その他のユーザによって読み込み可能

0o1000 --------t グループに対する t ビット (スティッキービット)

0o2000 -----s--- グループに対する s ビット (SGID)

0o4000 --s------ ユーザに対する s ビット (SUID)

Table 2 — 権限ビット

(chdir する) ための権限であり、読み込み権限とはディレクトリの内容を一覧で表示
するための権限です。ただしディレクトリ内のファイルやサブディレクトリの名前を

知っている場合、それらを読み込むためにはディレクトリの読み込み権限は必要では

ありません。

スペシャルビットは xビットが立っていない場合には意味を持ちません (xが立っ
ていないならば、追加の権限を与えません) 。スペシャルビットの場所が x と同じで、

x が設定されていないときには s, t の代わりに S, T が使われるのはこのためです。
t フラグはスティッキービットと呼ばれ、このフラグが付いたディレクトリでは全

てのユーザがファイルとディレクトリの作成を行えますが、削除が行えるのは所有者

とルートだけです。

s ビットが立っている実行可能ファイルを実行すると、ファイルの所有者または

所有グループとしてファイルが実行されます。さらにプログラムの実行時にシステム

コール setuid と setgid を呼ぶことで、実効ユーザ識別子とグループを本来のユー

ザ/グループに切り変えることができます。

val setuid : int -> unit

val setgid : int -> unit

setuid と setgid が呼ばれたとき、プロセスは元のユーザ/グループ識別子を保存しま
す。元の識別子が保存されるのは実効識別子を特別な権限なしに後で戻すことができる

ようにするためです。システムコール getuid と getgid は元の識別子を返し、geteuid

と getegid は実効識別子を返します。

val getuid : unit -> int

val geteuid : unit -> int

val getgid : unit -> int

val getegid : unit -> int

ただしスーパーユーザが setuid と setgid を実行した場合は別で、この場合は実効

ユーザ/グループ識別子と実ユーザ/グループ識別子の両方を変更します。
プロセスは他にもファイル作成マスクを持ちます。これはファイル権限と同じよ

うにエンコードされ、名前が示すように、禁止する操作を表します。ファイルを作成

するとき、ファイル作成マスクで 1になっているビットは作成されるファイルの権限
では 0 になります。ファイル作成マスクはシステムコール umask で取得および変更で

きます。

val umask : int -> int

システム変数を変更する多くのシステムコールと同じように、ファイル作成マスクを

変更する umask は古い値を返します。そのため、この関数を二回呼べば現在の値を確

認できます。一回目は適当な値を入力して変数の現在の値を手に入れ、二回目でその

値を入力して変数を元の値に戻します。例えば:

let m = umask 0 in ignore (umask m); m

ファイルアクセス権限はシステムコール chmod と fchmod で変更できます。

val chmod : string -> file_perm -> unit

val fchmod : file_descr -> file_perm -> unit

これらの関数が動作していることはシステムコール access によって “動的に” 確認で
きます。

type access_permission = R_OK | W_OK | X_OK | F_OK

val access : string -> access_permission list -> unit

ここでアクセスされるファイルへの権限の問い合わせは access_permission 型の値の

リストで表されます。F_OK はファイルが存在しているかどうかを (他の権限を確認せ
ずに) 確認します。他の値の意味は明らかです。

access によって調べられる情報は lstat で得られる情報よりも制限的なことがあ

ることに注意してください。これはファイルシステムが制限された権限— 例えば、読

み込み専用モード — のもとにマウントされる場合があるためです。動的な 情報 (プ
ロセスが実際にできることへの制限) と 静的な 情報 (ファイルシステムが指定する制
限) を区別したのはこのためです。

2.4 ディレクトリに対する操作

ディレクトリに書き込めるのカーネルだけ (そしてファイルを作成するときだけ) で
す。そのため、ディレクトリを書き込みモードで開くことは禁止されています。Unix
の特定のバージョンでは、読み込み専用モードでディレクトリを開いて readで読む

ことが許されていますが、別のバージョンでは禁止されています。しかしディレクト

リエントリのフォーマットは Unix のバージョンによって異なり、複雑なことが多い
ので、仮にディレクトリへの書き込みができたとしても行うべきではありません。次

の関数を使うとポータブルにディレクトリを走査することができます:

val opendir : string -> dir_handle

val readdir : dir_handle -> string

val rewinddir : dir_handle -> unit

val closedir : dir_handle -> unit

システムコール opendir はディレクトリのディレクトリディスクリプタを返します。

readdir はディスクリプタの次のエントリを読んで同じディレクトリ内のファイルの

名前を返すか、ディレクトリの終端に到達した場合には End_of_file 例外を出します。

rewinddir はディスクリプタをディレクトリの最初に移動し、closedir はディレクト

リディスクリプタを閉じます。

例 Misc モジュールに含まれる次のライブラリ関数はディレクトリ dirname 内のエ

ントリーについて、関数 f を繰り返し適用します。

let iter_dir f dirname =

let d = opendir dirname in

try while true do f (readdir d) done

with End_of_file -> closedir d

* * *
ディレクトリの作成と空ディレクトリの削除には mkdir と rmdir を使います。

val mkdir : string -> file_perm -> unit

val rmdir : string -> unit

mkdir の第二引数には新しく作られるディレクトリのアクセス権限を指定します。す

でに空であるディレクトリしか削除することはできません。そのためディレクトリと

その要素を削除するにはまず再帰的にディレクリの要素を削除してからディレクトリ

を削除することが必要になります。

2.5 完全な例: ファイル階層の検索

Unix コマンド find はファイル階層にあるファイルで一定の条件 (ファイル名、タイ
プ、権限など) に合致するものを一覧で表示します。このセクションではこの探索を
実装したライブラリ関数 Findlib.find と、-follow そして -maxdepth オプションに

対応した find コマンドを作成します。

Findlib.find に対するインターフェースを以下のように定めます:

val find :

(Unix.error * string * string -> unit) ->

(string -> Unix.stats -> bool) -> bool -> int -> string list ->

unit

関数呼び出し

find handler action follow depth roots

はリスト roots で指定されるファイル (絶対パスまたは関数が呼ばれたときのプロセ
スのカレントディレクトリからの相対パス)をルートとするファイル階層を最大 depth

の深さまで、フラグ follow がセットされているならばシンボリックリンクをたどって

探索します。探索を開始したパスを r とすると、探索結果のパスは r を先頭に持ちま

す。探索で見つかったパス p は Unix.lstat p (follow が true の場合は Unix.stat p)
の結果とともに action 関数に渡されます。ディレクトリに対しては、action 関数は

探索をそのディレクトリの探索を続けるべきか (true) かやめるべきか (false) を返
します。

handler 関数は探索中に起こった Unix_error 型のエラーを報告します。エラーが

起こった場合は、例外の引数が handler 関数に渡され探索は続行されます。例外が

action 関数または handler 関数の内部で起こった場合にはその時点で探索は終了し、

例外は呼び出し側に伝わります。action と handler の中で出される Unix_error 例外

を探索中に起こったエラーと区別するために、Hidden 例外でラップします (hide_exn
と reveal_exn 参照)。

1 open Unix;;

2

3 exception Hidden of exn

4 let hide_exn f x = try f x with exn -> raise (Hidden exn);;

5 let reveal_exn f x = try f x with Hidden exn -> raise exn;;

6

7 let find on_error on_path follow depth roots =

8 let rec find_rec depth visiting filename =

9 try

10 let infos = (if follow then stat else lstat) filename in

11 let continue = hide_exn (on_path filename) infos in

12 let id = infos.st_dev, infos.st_ino in

13 if infos.st_kind = S_DIR && depth > 0 && continue &&

14 (not follow || not (List.mem id visiting))

15 then

16 let process_child child =

17 if (child <> Filename.current_dir_name &&

18 child <> Filename.parent_dir_name) then

19 let child_name = Filename.concat filename child in

20 let visiting =

21 if follow then id :: visiting else visiting in

22 find_rec (depth-1) visiting child_name in

23 Misc.iter_dir process_child filename

24 with Unix_error (e, b, c) -> hide_exn on_error (e, b, c) in

25 reveal_exn (List.iter (find_rec depth [])) roots;;

ディレクトリはデバイス番号と inode 番号の組 id によって識別されます ( 12 行
目)。リスト visiting がそれまでに訪問したディレクトリを記録します。この情報が

必要になるのはシンボリックリンクをたどる時だけです ( 21 行目)。
ここまでくれば、find コマンドを作るのは簡単です。このコードの主な処理は Arg

モジュールを使ってコマンドライン引数をパースすることです。

let find () =

let follow = ref false in

let maxdepth = ref max_int in

let roots = ref [] in

let usage_string =

("Usage: " ^ Sys.argv.(0) ^ " [files...] [options...]") in

let opt_list = [

"-maxdepth", Arg.Int ((:=) maxdepth), "max depth search";

"-follow", Arg.Set follow, "follow symbolic links";

] in

Arg.parse opt_list (fun f -> roots := f :: !roots) usage_string;

let action p infos = print_endline p; true in

let errors = ref false in

let on_error (e, b, c) =

errors := true; prerr_endline (c ^ ": " ^ Unix.error_message e) in

Findlib.find on_error action !follow !maxdepth

(if !roots = [] then [ Filename.current_dir_name ]

else List.rev !roots);

if !errors then exit 1;;

Unix.handle_unix_error find ();;

この find の機能は少ないですが、これからの練習問題で示されるように、ライブラ

リ関数 FindLib.find ははるかに多機能です。

練習問題 1 Findlib.findを使って、以下のUnixコマンドと同じコマンド find_out_CVS

を書いてください。

find . -type d -name CVS -prune -o -print

このコマンドはカレントディレクトリから始まりファイルの名前を再帰的に表示しま

すが、CVS という名前のディレクトリについては表示することもディレクトリに入る

こともしません。解答は 112 ページ。

* * *

練習問題 2 getcwd 関数はシステムコールではありませんが、Unix モジュールで定

義されています。getcwd の “原始的な” 実装を与えてください。まずアルゴリズムの
原理を言葉で説明してから実装するようにしてください (同じシステムコールを何度
も呼ぶのは避けたほうが良いでしょう)。解答は 112 ページ。

* * *

2.6 ファイルのオープン

openfile 関数を使うと指定した名前のファイルに対するディスクリプタを得ることが

できます (対応するシステムコールは open ですが、これは ocaml の予約語なので使

うことができません) 。

val openfile :

string -> open_flag list -> file_perm -> file_descr

第一引数は開くファイルの名前です。第二引数は open_flag 列挙型のフラグのリスト

であり、ファイルが開かれるモードおよびファイルが存在しなかったときの動作を指

定します。file_perm 型の第三引数はファイルが作られるときのファイルのアクセス

権限を指定します。返り値はファイルへのディスクリプタであり、入出力位置は最初

ファイルの先頭にあります。

第二引数のフラグのリストは以下のうちちょうど一つだけを含む必要があります。

O_RDONLY 読み込み専用モードで開く。

O_WRONLY 書き込み専用モードで開く。

O_RDWR 読み込み/書き込みモードで開く。

これらのフラグは読み込みと書き込み命令がディスクリプタに対して行えるかどうか

を指定します。読み込み/書き込み権限のないファイルを読み込み/書き込みモードで

開こうとした場合、openfile は失敗します。このため、全てのファイルを O_RDWR を

使って開こうとするのは避けるべきです。

openfile の第二引数のフラグには以下の値を一つ以上含むことができます。

O_APPEND 追記モードで開く。

O_CREAT ファイルが存在しない場合作成する。

O_TRUNC ファイルが存在する場合、内容を切り捨てる。

O_EXCL ファイルがすでに存在しているなら失敗する。

O_NONBLOCK ノンブロッキングモードで開く。

O_NOCTTY 端末モードでは機能しない。

O_SYNC ファイルに関する同期モードで書き込みを行う。

O_DSYNC データに関する同期モードで書き込みを行う。

O_RSYN 同期モードで読み込みを行う。

最初のグループはファイルが存在に関連した動作を決めます:

• O_APPEND が指定された場合、全ての書き込み処理の前に入出力位置がファイルの末尾

にセットされる。これによって書き込まれたデータはファイルの末尾に付け足される

ようになる。O_APPEND が無い場合、書き込みは入出力位置 (初期位置はファイルの先
頭) で行われる。

• O_TRUNC が指定された場合、ファイルは開かれたときに切り捨てられる。ファイルの

内容は失われてファイルの長さは 0 になり、最初の書き込みは空ファイルに行われる。
O_TRUNC が無い場合、書き込みはファイルの先頭からそこにあるデータを上書きしな

がら行われる。

• O_CREAT が指定された場合、ファイルが存在しなければ作成される。作成されるファ

イルは空で、その権限は openfile の第三引数とプロセスの作成マスクによって決ま

る (作成マスクは umask によって取得および確認できる) 。

• O_EXCL が指定された場合、ファイルがすでに存在しているなら openfile は失敗する。

このフラグと O_CREATE を併せると、ファイルを ロック1 として使うことができる。
ロックを取得したいプロセスは O_EXCL と O_CREAT を指定して openfile を呼ぶ。ファ

イルが存在していた場合、これは他のプロセスがロックを取得済みであることを意味

し、openfile はエラーを出す。ファイルが存在せず openfile がエラーを出さずに値

を返しファイルが作られた場合、他のプロセスはロックを取得することができなくな

る。ロックを開放するにはプロセスはロックファイルに unlink を行う。ファイルの

作成はアトミックな演算である: もし二つのプロセスが O_EXCL と O_CREAT を指定し

て同時に同じファイルを作成しようとした場合、多くとも一つのプロセスしか成功し

ない。この手法の欠点はプロセスが現在使用中のロックを得るためにビジーウェイト

する必要があることと、プロセスの異常終了がロックを開放しない場合があることで

ある。

1ロックファイルが NFS パーティションに存在している場合はこれは不可能です。NFS が open に

対する O_CREAT を実装していないためです。

例 たいていのプログラムは openfile の第三引数として 0o666 を使います。これは

文字列でいうと rw-rw-rw- を意味します。デフォルトファイル作成マスクが 0o022 の

場合、ファイルは rw-r--r-- の権限で作成されます。マスクがより寛大で 0o002 の場

合には、ファイルは rw-rw-r-- の権限で作成されます。

* * *
例 ファイルから読み込むには以下のようにします:

openfile filename [O_RDONLY] 0

O_CREAT が指定されていないならば、第三引数は何でも構いません。0 がよく使われ
ます。

それまでの内容にかかわらず空ファイルに書き込むには以下のようにします:

openfile filename [O_WRONLY; O_TRUNC; O_CREAT] 0o666

ファイルが実行可能なコード (例えば ld によって作られるファイルやスクリプトな

ど) を含む場合、ファイルを実行権限付きで作成します:

openfile filename [O_WRONLY; O_TRUNC; O_CREAT] 0o777

ファイルが機密情報 (例えば mail が既読メールを保存する “メールボックス”) であ
る場合、書き込み権限を所有ユーザのみとして作成します:

openfile filename [O_WRONLY; O_TRUNC; O_CREAT] 0o600

存在するファイルの末尾にデータを付け足すか、ファイルが存在しない場合には作成

するには以下のようにします:

openfile filename [O_WRONLY; O_APPEND; O_CREAT] 0o666

* * *
O_NONBLOCK フラグはファイルが名前付きパイプまたはスペシャルファイルの場合

に、ファイルのオープンとその後の読み込みがノンブロッキングであることを保証し

ます。

O_NOCTYY フラグはファイルが制御端末 (キーボードやウィンドウなど) の場合に、
そのファイルが呼び出しプロセスの制御端末にならないことを保証します。

フラグの最後のグループは読み込みと書き込み処理をどのように同期するかを指

定します。デフォルトでは処理は同期しません。

• O_DSYNC が指定された場合、データの書き込みは同期されプロセスは全ての書き込み

が物理的にメディア (通常はディスク) に行われるまでブロックされる。

• O_SYNC が指定された場合、ファイルのデータとメタ属性の書き込みが同期される。

• O_RSYNC が O_DSYNC と共に指定された場合データの読み込みも同期される: 読み込み
が起こる前にそれまでの全ての書き込み (要求されたが実行されていないものを含む)
が本当にメディアに書き込まれることが保証される。O_RSYNC が O_SYNC と共に指定

された場合上記のことがメタ情報にも適用される。

2.7 読み込みと書き込み

システムコール read と write はバイト列をファイルに書き込みます。歴史的な理由

から、システムコール write は OCamlでは single_write という名前です。

val read : file_descr -> bytes -> int -> int -> int

val single_write : file_descr -> bytes -> int -> int -> int

二つの関数 read と single_write は同じインターフェースを持ちます。第一引数は操

作を行うファイルディスクリプタです。第二引数2は read の場合は読み込んだバイト

列を収める文字列で、single_write の場合は書き込むバイト列です。第三引数は文字

列の中で入出力を行う最初のバイトの位置で、第四引数は入出力を行うバイト数です。

第三、四引数は第二引数の部分文字列を定めています (この部分文字列はもちろん有
効なものである必要がありますが、read と single_write はこのことをチェックしま

せん) 。

fd

read fd s d n write fd s d n

s

d n

read と single_write は実際に読み込んだ/書き込んだバイト数を返します。
入出力の命令はファイルディスクリプタの現在の入出力位置から行われます (ファ

イルが O_APPEND モードで開かれた場合、この位置は書き込み命令の前にファイルの

末尾にセットされます)。システムコールの後、現在位置は読み込み/書き込みを行っ
たバイト数だけ進みます。

書き込みでは実際に書き込むバイト数は要求されたバイト数と普通一致しますが、

いくつか例外があります: (i) バイト列を書き込めなかった場合 (例えばディスクが満
杯なとき) (ii) ディスクリプタがノンブロッキングモードで開かれたパイプまたはソ
ケットな場合 (iii) 書き込む文字列が OCamlの持つバッファより大きい場合

(iii) の理由は OCamlが最大値の制限された補助バッファを使っているためです。
バッファの最大値よりも書き込みが大きかった場合、書き込みは部分的になります。

この問題を解決するために、OCamlには エラーが出るか全てのデータが書き込まれ
るまで書き込みを繰り返す write があります。しかしこの関数を使うとエラーが起

こった場合に書き込まれたバイト数を知ることができません。single_writeを使うと

書き込みの原始的になり (何が書き込まれたかが分かる)、オリジナルの Unix システ
ムコールにより忠実になるので single_write を使うべきです。single_write の実装

はセクション 5.7 で説明されています。

例 fd が書き込み専用モードで開かれたディスクリプタだとします。

write fd "Hello world!" 3 7

は "lo worl" という文字列を対応するファイルに書き込み、7 を返します。

* * *
読み込みでは実際に読み込んだバイト数が読むように要求されたバイト数よりも

小さいことがありえます。例えばファイルの終端が近いときは現在位置からファイル

の終端までのバイト数が要求されたバイト数よりも小さくなります。特に現在位置が

ファイルの終端なとき read は 0 を返します。“ゼロはファイルの終端と等しい” と
いう慣習はスペシャルファイルやパイプ、ソケットに対しても成り立ちます。例えば

ctrl-D を端末に入力すると read は 0 を返します。
read が要求した値よりも小さい値を返すもうひとつの例は端末から読み込む場合

です。この場合 read はまず行の入力が利用可能になるまでブロックします。行が入

2訳注: バージョン 4.02以降の OCamlでは、変更可能なバッファとして使われる文字列の型は string

ではなく bytes となります。

ディスクリプタのクローズ

力され、その長さが要求されたバイト数よりも短い場合、read は要求されたバイト数

に達しようと次のデータを待つことをせずに行の入力が利用可能になった時点で値を

返します (これは端末のデフォルトの動作ですが、文字ごとに読み込むように変える
こともできます。セクション section 2.13 と terminal_io 型を参考にしてください)。
例 次のプログラムは標準入力から最大 100文字を読み込み、文字列として返します。

let buffer = Bytes.create 100 in

let n = read stdin buffer 0 100 in

Bytes.sub buffer 0 n

* * *
例 以下の関数 really_read は read と同じインターフェースを持ちますが、要求さ

れたバイト数を取得するために追加の読み込みを行います。読み込み中にファイルの

終端に達した場合には End_of_file 例外が出ます。

let rec really_read fd buffer start length =

if length <= 0 then () else

match read fd buffer start length with

| 0 -> raise End_of_file

| r -> really_read fd buffer (start + r) (length - r);;

* * *

2.8 ディスクリプタのクローズ

システムコール close はファイルディスクリプタを閉じます。

val close : file_descr -> unit

ディスクリプタが閉じられると読み込みや書き込みなどのディスクリプタに関す

る操作は全て失敗します。ディスクリプタは必要なくなった時点で閉じられるべきで

すが、閉じることは必須ではありません。write 関数による書き込みの要求が即時に

カーネルに伝わるために、Pervasives モジュールのチャンネルとは違って全ての書き

込みが実行されたことを保証するためにチャンネルを閉じる必要はありません。一方

プロセスが確保できるディスクリプタの数はカーネルによって (数百から数千に) 制
限されていることから、使わないディスクリプタを close で開放しないとディスクリ

プタが枯渇します。

2.9 完全な例: ファイルのコピー

引数として与えられる f1 と f2 について、f1 のバイト列を f2 にコピーするコマンド

file_copy を作ります。

open Unix;;

let buffer_size = 8192;;

let buffer = Bytes.create buffer_size;;

let file_copy input_name output_name =

let fd_in = openfile input_name [O_RDONLY] 0 in

let fd_out = openfile output_name [O_WRONLY; O_CREAT; O_TRUNC] 0o666 in

システムコールのコストとバッファ

let rec copy_loop () = match read fd_in buffer 0 buffer_size with

| 0 -> ()

| r -> ignore (write fd_out buffer 0 r); copy_loop ()

in

copy_loop ();

close fd_in;

close fd_out;;

let copy () =

if Array.length Sys.argv = 3 then begin

file_copy Sys.argv.(1) Sys.argv.(2);

exit 0

end else begin

prerr_endline

("Usage: " ^ Sys.argv.(0) ^ " <input_file> <output_file>");

exit 1

end;;

handle_unix_error copy ();;

作業の多くは file_copy 関数によって実行されます。最初に入力ファイルのディ

スクリプタを読み込み専用で開き、次に出力ファイルのディスクリプタを書き込み専

用モードで開きます。

出力ファイルがすでに存在している場合ファイルは切り捨てられ (O_TRUNC オプ
ション)、存在しない場合には作成されます (O_CREAT オプション)。作成されるファ
イルの権限は rw-rw-rw- をファイル作成マスクで改変したものですが、これは十分で

はありません。実行可能ファイルをコピーする場合は、コピー先も実行可能であるべ

きだからです。コピー先のファイルと元のファイルの権限を同じにする方法は後述し

ます。

copy_loop 関数の中で buffer_size バイトのコピーを行います。まず buffer_size

の読み込みを行い、これが 0 を返した場合はファイルの終端に到達しているのでコ
ピーを終了します。そうでなければ読み込んだ r バイトを出力ファイルに書き込んで

同じことを繰り返します。

最後に二つのディスクリプタを閉めます。プログラム本体となる copy はコマンド

が二つの引数を受け取ったことを確認し、その引数を file_copy 関数に渡します。

コピー中に起きた Unix_error は handle_unix_error によって補足され、エラーの

内容が表示されます。ここで起こるエラーの例としては入力ファイルが存在しないた

めに開くことができない、権限が足りなくてファイルを開くことができない、ディス

クに容量がなくて書き込むことができない、などがあります。

練習問題 3 file_copy -a f1 f2 が f1 の内容を f2 の末尾に付け足すように、-a オ

プションを追加してください。解答は 112 ページ。

* * *

2.10 システムコールのコストとバッファ

file_copy の例では読み込みを 8192 バイトごとに行ないました。どうして 1 バイト
ごとや 1 メガバイトごとに読み込みをしないのでしょうか? 理由は効率です。図 2 に

システムコールのコストとバッファ

Figure 2 — ブロックサイズの関数としてのコピー速度

file_copy の速度を示します。一秒間にコピーできるバイト数を縦軸に、ブロックサ

イズ (buffer_size の値) を横軸に示しています。転送されるデータの総量はブロック
サイズに関わらず一定です。ブロックサイズが小さい時は、コピー速度はブロックサ

イズにほぼ比例しています。実行時間の多くがはデータの転送ではなく、copy_loop

のループと read と write の呼び出しに使われているということがわかります。更に

詳細に実行時間を計測すると、ほとんどが read と write の呼び出しに使われている

ことがわかります。システムコールは処理が大きくない場合でも (テストに使われた
PC — 2.8 GHz Pentium 4 — では) 最低 4 マイクロ秒、一般的には 1 から 10 マイ
クロ秒程度かかります。そのため入出力のブロックサイズが小さい場合にはシステム

コールの時間が支配的になります。

ブロックが大きくて 4KB から 1MB の場合、コピー速度は最大値で一定です。こ
こではシステムコールとループにかかる時間がデータ転送にかかる時間に比べて小さ

くなっているということです。加えてバッファのサイズがシステムのキャッシュよりも

大きくなるためにデータの転送がシステムコールのコストを上回るようになります3。
最後に、ブロックがとても大きい (8 MB 以上) ときにはコピー速度は最大値より

も少しだけ小さくなります。ここで影響するのはブロックを確保してメモリのページ

を割り当てるのを書き込み中に行う時間です。

以上のことから学べることは、システムコールはほとんど何も処理をしていない

場合でも大きな — 通常の関数呼び出しよりもはるかに大きな — コストがかかると

いうことです。アーキテクチャによって違いますが、だいたい 2 から 20 マイクロ秒
が呼び出しごとにかかります。そのためシステムコールの数を減らすことが重要にな

ります。読み込みと書き込みに関して言えば、一文字ごとではなくある程度のサイズ

のブロックごとに行われるべきです。

file_copy の例では大きなブロックで入出力を行うのは難しくありません。しかし

ある種のプログラムでは一文字ごとに入出力を行うことが自然なことがあります (例え
ばファイルから一行ずつ読む処理、字句解析、数字の印字など)。このようなプログラム
のために、ほとんどのシステムにはアプリケーションとオペレーティングシステムの

間にソフトウェアのレイヤーを追加する入出力ライブラリがあります。例えば OCaml
3実際には OCamlはデータ転送を (現在のバージョンでは) 16KB に制限し全体の転送が終わるま

で write システムコールを繰り返します — セクション 5.7 参照。しかしこの制限はシステムのキャッ
シュサイズよりも大きいので無視できます。

完全な例: 簡単な入出力ライブラリ

には Pervasives モジュールにファイルディスクリプタと似た抽象型 in_channel と

out_channelが定義されていて、この型に関する関数 input_char,input_line、output_char
あるいは output_string があります。このレイヤーはバッファを使って複数回の一文

字ごとの読み込みと書き込みを一回のシステムコールにまとめます。これによって一

文字ごとに処理をするプログラムの効率が良くなります。さらにこのレイヤーによっ

てプログラムがよりポータブルになります。Pervasives モジュールを使うプログラム

を新しいオペレーティングシステムに移植するには、このライブラリをそのシステム

上で使えるシステムコールを使って実装すれば良いからです。

2.11 完全な例: 簡単な入出力ライブラリ

バッファを使った入出力のテクニックの例として、OCamlの Pervasives ライブラリ

の一部を実装します。次のようなインタフェースを持ちます:

exception End_of_file

type in_channel

val open_in : string -> in_channel

val input_char : in_channel -> char

val close_in : in_channel -> unit

type out_channel

val open_out : string -> out_channel

val output_char : out_channel -> char -> unit

val close_out : out_channel -> unit

“読み込み” の部分から始めます。抽象型 in_channel は次のように定義します:

open Unix;;

type in_channel =

{ in_buffer: bytes;

in_fd: file_descr;

mutable in_pos: int;

mutable in_end: int };;

exception End_of_file

文字列 in_buffer は文字通りのバッファです。フィールド in_fd は読み込むファイル

に開かれた (Unix の) ファイルディスクリプタです。フィールド in_pos は読み込み

の現在位置を示します。フィールド in_end は事前にバッファへ読み込まれた文字列

のうち有効な部分長さです。

完全な例: 簡単な入出力ライブラリ

in_fd

読み込んだ

文字列

バッファされ

これから読み

込まれる文字列

in_buffer

読み込み位置 in_fd

in_pos in_end

in_pos と in_end のフィールドは読み込み処理中に更新されるので mutable として宣

言します。

let buffer_size = 8192;;

let open_in filename =

{ in_buffer = Bytes.create buffer_size;

in_fd = openfile filename [O_RDONLY] 0;

in_pos = 0;

in_end = 0 };;

読み込みのためにファイルを開いたとき、同時に合理的なサイズの (システムコールが
多くなりすぎない程度に大きく、メモリを無駄遣いしない程度に小さい) バッファを
作ります。その後 in_fd フィールドを読み込み専用で開いたファイルに対する Unix
のファイルディスクリプタで初期化します。バッファは最初空です (ファイルからの
どんな文字列も含んでいません) 。そのため in_end フィールドは 0 で初期化します。

let input_char chan =

if chan.in_pos < chan.in_end then begin

let c = chan.in_buffer.[chan.in_pos] in

chan.in_pos <- chan.in_pos + 1;

c

end else begin

match read chan.in_fd chan.in_buffer 0 buffer_size

with 0 -> raise End_of_file

| r -> chan.in_end <- r;

chan.in_pos <- 1;

chan.in_buffer.[0]

end;;

in_channel から文字を読むとき、次の二つのうち一つを行います。一つ目はバッファ

に一つ以上まだ読んでいない文字がある、つまり in_pos フィールドの値が in_end

フィールドの値よりも小さい場合です。このときはバッファの in_pos にある文字を

返し、in_pos をインクリメントします。もう一つはバッファが空の場合で、このとき

は read を呼んでバッファにもう一度文字列を読み込みます。read が 0 を返したなら
ファイルの終端に達したということなので End_of_file 例外を出します。そうでなけ

れば in_end に呼んだ文字の数を代入します。

let close_in chan =

close chan.in_fd;;

完全な例: 簡単な入出力ライブラリ

in_channel を閉じる処理は対応する Unix のファイルディスクリプタを閉じるだけ
です。

“書き込み” の部分は “読み込み” の部分にとても良く似ています。唯一異なるの
はバッファがまだ読んでいない読み込み (バッファされたが読み込まれていない文字
列) を保持するのではなくて、まだ完了していない書き込み (バッファされたがファ
イルディスクリプタに書き込まれていない文字列) を保持する点です。

out_fd

書き込んだ

文字列

バッファされ

これから書き

込まれる文字列

out_buffer

書き込み位置 out_fd

out_pos

type out_channel =

{ out_buffer: bytes;

out_fd: file_descr;

mutable out_pos: int };;

let open_out filename =

{ out_buffer = Bytes.create 8192;

out_fd = openfile filename [O_WRONLY; O_TRUNC; O_CREAT] 0o666;

out_pos = 0 };;

let output_char chan c =

if chan.out_pos < Bytes.length chan.out_buffer then begin

chan.out_buffer.[chan.out_pos] <- c;

chan.out_pos <- chan.out_pos + 1

end else begin

ignore (write chan.out_fd chan.out_buffer 0 chan.out_pos);

chan.out_buffer.[0] <- c;

chan.out_pos <- 1

end;;

let close_out chan =

ignore (write chan.out_fd chan.out_buffer 0 chan.out_pos);

close chan.out_fd;;

out_channel に文字を書き込むには次の二つのうち一つを行います。一つ目はバッ

ファが満杯ではない場合で、このときは文字をバッファを out_pos の位置に保存して

out_pos をインクリメントします。もう一つはバッファが満杯の場合で、このときは

write を呼んでバッファを空にしてからバッファの先頭に文字を読み込みます。

入出力の位置

out_channel を閉めるときにはバッファの内容 (位置 0 から out_pos - 1 までの

文字列) を書き込むことを忘れないでください。これを忘れると最後にバッファが空
になってからチャンネルに書き込まれた内容が失われます。

練習問題 4 次の関数を実装してください:

val output_string : out_channel -> string -> unit

この関数は output_char をそれぞれの文字に対して複数回呼んだときと同じ動作をし

ますが、より効率的です。解答は 112 ページ。

* * *

2.12 入出力の位置

システムコール seek はファイルディスクリプタの現在の入出力位置を変更します。

val lseek : file_descr -> int -> seek_command -> int

第一引数はファイルディスクリプタで第二引数は移動させる位置です。第二引数は

seek_command 型の第三引数に基づいて解釈されます。この列挙型は位置の種類を指定

します:

SEEK_SET 第二引数は関数を呼び出した後の入出力位置を表す。ファイルの最

初の文字は位置 0 である。

SEEK_CUR 第二引数は現在の入出力位置からの相対的なオフセットを表す。正

の値のとき前に、負の値のとき後ろに動く。

SEEK_END 第二引数はファイルの終端からの相対的なオフセットを表す。SEEK_CUR

と同様にオフセットは正負どちらにもなれる。

lseek の返り値は関数を実行した後の入出力位置 (絶対位置) です。
負の絶対位置が指定された場合はエラーとなります。ファイルの終端よりも後ろ

の位置を指定することは可能です。このとき read は (ファイルの末尾に達しているの
で) 0 を返し、write はまずファイルの終端から入出力位置まで 0 を書き込んでから
データを書き込みます。

例 カーソルを 1000 番目の文字に移動させるには以下のようにします:

lseek fd 1000 SEEK_SET

一文字巻き戻すには以下のようにします:

lseek fd (-1) SEEK_CUR

ファイルのサイズを求めるには以下のようにします:

let file_size = lseek fd 0 SEEK_END in ...

* * *
ファイルディスクリプタが O_APPEND モードで開かれている場合、入出力位置は毎

回の書き込みの前にファイルの終端にセットされます。そのため書き込み位置を指定

するために lseek を読んでも意味がありません。一方読み込みを指定することには使

えます。

ファイルの種類に特有の操作

コミュニケーションデバイス (パイプ、ソケット) や端末を始めとする多くのスペ
シャルファイルなどの、入出力の絶対位置が意味を持たないタイプのファイルについ

ては lseek の動作は未定義です。Unix のほとんどの実装ではこれらのファイルに対
する lseek は無視されます (入出力位置はセットされますが、入出力処理は入出力位
置を無視します)。いくつかの実装ではパイプとソケットに対する lseek はエラーを

出します。

練習問題 5 tail コマンドはファイルの末尾 n 行を表示します。tail を通常ファイ

ルに対して効率よく実装するにはどうすればよいでしょうか ? -f オプションはどの

すれば実装できるでしょうか (参考: man tail) ? 解答は 113 ページ。

* * *

2.13 ファイルの種類に特有の操作

Unix ではデータのやり取りはディスクリプタを通して行われ、ディスクリプタは永
続性のファイル (通常ファイル、周辺機器) または揮発性のファイル (パイプとソケッ
ト、 5 章と 6 章参照) を表します。ファイルディスクリプタはデータのやり取りのた
めの統一されたメディアによらないインタフェースを提供します。もちろんファイル

ディスクリプタに対する操作の実際の実装は背後にあるメディアによって異なります。

しかしあるメディアの全ての機能を使う必要がある場合は他のファイルと同じよ

うに扱うことはできません。ファイルのオープンや読み込み、書き込みなどの一般的

な操作はほとんどのディスクリプタで同じ動作をします。しかしこのような一般的な

操作であっても、周辺機器とパラメータで決まるアドホックな動作をするスペシャル

ファイルが存在します。またあるメディアに対してだけ可能な操作もあります。

通常ファイル

システムコール truncate と ftruncate を使うと通常ファイルを短くすることができ

ます。

val truncate : string -> int -> unit

val ftruncate : file_descr -> int -> unit

第一引数は切り捨てるファイルで第二引数は切り捨てた後のサイズです。これより後

ろの位置にある全てのデータは失われます。

シンボリックリンク

シンボリックリンクに対するほとんどの操作はリンクを “たどり” ます。つまり、操
作はリンクそのものではなくリンクが指すファイルに適用されます (例えば openfile,
stat, truncate, opendir などはこのような動作をします) 。
二つのシステムコール symlink と readlink はシンボリックリンクそのものを操

作します。

val symlink : string -> string -> unit

val readlink : string -> string

symlink f1 f2 は f1 へのシンボリックリンク f2 を作成します (Unix コマンド
ln -s f1 f2 と同様です) 。readlink はシンボリックリンクの内容、つまりリンク

が指すファイルの名前を返します。

スペシャルファイル

スペシャルファイルは “キャラクタ” または “ブロック” に分類されます。前者は文字
のストリームです。つまり文字の入出力は逐次的にしか行うことができません。例と27

ファイルの種類に特有の操作

して端末やサウンドデバイス、プリンターなどがあります。後者は永続的な媒体を持

つものであり、ディスクが典型です。文字はブロック単位で読み込むことができ、現

在位置からの相対位置にシークすることができます。

スペシャルファイルは以下のように分類できます:

/dev/null あらゆるものを飲み込み何も出てこないブラックホール。プロセスの

出力を /dev/null にリダイレクトすることでプロセスの出力を無視

できる。

/dev/tty* 制御端末。

/dev/pty* 本物の端末ではないが端末をシミュレートし同じインターフェース

を持つ擬似端末。

/dev/hd* ディスク。

/proc システム変数。Linux ではシステム変数はファイルシステムで管理さ
れ、入出力が許されている。

多くのファイルに対するシステムコールはスペシャルファイルに対して違った動

作をします。しかし read と write に関しては、ほとんどのスペシャルファイル (端
末、テープドライバ、ディスクなど) が通常ファイルと同じ動作をします (読み書きす
るバイト数に制限があることがあります)。ただしそのようなスペシャルファイルの多
くが lseek を無視します。

通常のファイルシステムに加えて、周辺機器を表すスペシャルファイルは動的に

制御また設定される必要があります。例えばテープドライブには巻き戻しや早送りが、

端末には行の編集モードや特殊文字による制御、シリアル通信用変数 (スピード、パ
リティなど) があります。Unix ではこれらデバイスのパラメータの設定は全てシステ
ムコール ioctl を通して行います。しかし OCamlにはこのシステムコールが提供さ
れていません。ioctl は引数の形が特殊なので統一的に扱うことができないためです。

端末

端末と擬似端末はキャラクタタイプのスペシャルファイルで、OCamlから設定を変
更することができます。システムコール tcgetattr はオープンされたスペシャルファ

イルのファイルディスクリプタを受け取り posix 規格に基づいて端末の状態を表す
terminal_io 型の構造体を返します。

type terminal_io =

{ c_ignbrk : bool; c_brk_int : bool; ...; c_vstop : char }

val tcgetattr : file_descr -> terminal_io

この構造体を変更してシステムコール tcsetattr を呼ぶことで周辺機器の設定を変更

できます。

val tcsetattr : file_descr -> setattr_when -> terminal_io -> unit

第一引数は操作する周辺機器のファイルディスクリプタです。最後の引数は terminal_io

型の構造体で、周辺機器への引数となります。第二引数は変更がいつ起きるべきかを

指定する setattr_when 列挙型の値です。即時 (TCSANOW)、データを全て送ってから
(TCSADRAIN)、データを全て受け取ってから (TCAFLUSH) の三つを指定できます。書き込
みに関するパラメータを変更するときには TCSADRAIN が、読み込みに関するパラメー

タを変更するときには TCSAFLUSH が推奨されます。

例 標準入力が端末または擬似端末の場合、パスワードを読む処理を行っている間は

ユーザが打ち込んだ文字列を表示するべきではありません。この処理は以下のように

実装できます:

let read_passwd message =

match

try

let default = tcgetattr stdin in

let silent =

{ default with

c_echo = false;

c_echoe = false;

c_echok = false;

c_echonl = false;

} in

Some (default, silent)

with _ -> None

with

| None -> input_line Pervasives.stdin

| Some (default, silent) ->

print_string message;

flush Pervasives.stdout;

tcsetattr stdin TCSANOW silent;

try

let s = input_line Pervasives.stdin in

tcsetattr stdin TCSANOW default; s

with x ->

tcsetattr stdin TCSANOW default; raise x;;

read_passwd 関数は stdin につながっている端末の現在の設定を取得するところから

始まります。その後文字を表示しないように変更した設定を定義します。もしこの処

理が失敗した場合標準入力は制御端末ではないので普通に一行の入力を受け取ります。

そうでなければメッセージを表示し、端末の設定を変え、パスワードを読み込み、端

末の設定を元に戻します。読み込みが失敗した後でも端末の設定が元に戻るように注

意が必要です。

* * *
プログラムが別のプログラムを起動しその標準入力を端末 (もしくは擬似端末) に

つなげる必要がある場合があります。OCamlはこれをサポートしていません4。その
ため擬似端末 (一般に /dev/tty[a-z][a-f0-9] という名前のついたファイル) の中か
らすでに開いているものを手動で探す必要があります。そしてその擬似端末のファイ

ルをオープンすれば、新しいプログラムを標準入力がこのファイルになった状態で始

めることができます。

端末のデータの流れを制御する関数が 4 つあります (割り込みを送る、送信の終
了を待つ、待っているデータをフラッシュする、やり取りを再開する)。

val tcsendbreak : file_descr -> int -> unit

tcsendbreak 関数は周辺機器に割り込みを送ります。第二引数は割り込みの長さです

(0 は周辺機器のデフォルト値と解釈されます)。
4Case ライブラリ [3] にはこのような関数が含まれます。

ファイルのロック

val tcdrain : file_descr -> unit

tcdrain 関数は全ての書き込みデータが送信されるのを待ちます。

val tcflush : file_descr -> flush_queue -> unit

第二引数の値にもとづいて、tcflush 関数は書き込まれたが送信されていないデータ

(TCIFLUSH) か受け取ったが読み込まれていないデータ (TCOFLUSH) 、あるいはその両
方 (TCIOFLUSH) を捨てます。
val tcflow : file_descr -> flow_action -> unit

第二引数の値にもとづいて、tcflow 関数はデータの送信を止める (TCOOFF) か、デー
タの送信を再開する (TCOON) か、制御文字 stop あるいは start を送って送信を止
める (TCIOFF) かをします。
val setsid : unit -> int

setsid 関数はプロセスを新しいセッションに移し端末から切り離します。

2.14 ファイルのロック

二つのプロセスは同じファイルに同時に書き込むことができますが、書き込みが衝突

した場合データの一貫性が失われることがあります。O_APPEND を使って常にファイル

の末尾に書き込むようにしてこれを回避できることがあります。log ファイルにはこ

の方法で良いですが、データベースのように任意の場所に書き込みが起こるときは上

手くいきません。そのような場合にはファイルを使うプロセスは他人のつま先を踏ま

ないように協調する必要があります。ファイル全体に対するロックは補助ファイルを

使うことで実装できます (17 ページ参照) が、システムコール lockf を使うとファイ

ルの一部分をロックするより良い同期パターンを利用できます。

val lockf : file_descr -> lock_command -> int -> unit

2.15 完全な例: 再帰的なファイルのコピー

file_copy (セクション 2.9) を拡張して通常ファイルだけではなくシンボリックリン
クとディレクトリにも対応させます。ディレクトリについてはその中身も再帰的にコ

ピーすることにします。

通常ファイルのコピーにはすでに定義した 2.9 関数を再利用します。
open Unix

...

let file_copy input_name output_name =

...

次の set_infos 関数はファイルの所有者とアクセス権限、最終アクセス/変更日時を
変更します。コピー先のファイルの情報をコピー元と同じにするためにこの関数を使

います。

let set_infos filename infos =

utimes filename infos.st_atime infos.st_mtime;

chmod filename infos.st_perm;

try

chown filename infos.st_uid infos.st_gid

with Unix_error(EPERM,_,_) -> ()

完全な例: 再帰的なファイルのコピー

システムコール utime は最終アクセス/更新日時を、chmod と chownがアクセス権限と

所有者を変更します。通常ユーザが chown を実行すると “permission denied” エラー
が出て失敗することがありますが、このエラーは捕捉した上で無視します。

処理の本体である再帰関数は以下のようになります:

let rec copy_rec source dest =

let infos = lstat source in

match infos.st_kind with

| S_REG ->

file_copy source dest;

set_infos dest infos

| S_LNK ->

let link = readlink source in

symlink link dest

| S_DIR ->

mkdir dest 0o200;

Misc.iter_dir

(fun file ->

if file <> Filename.current_dir_name

&& file <> Filename.parent_dir_name

then

copy_rec

(Filename.concat source file)

(Filename.concat dest file))

source;

set_infos dest infos

| _ ->

prerr_endline ("Can't cope with special file " ^ source)

source ファイルの情報を読むところから処理が始まります。ファイルが通常ファイル

の場合、file_copy によってデータを、set_infos によって情報をコピーします。ファ

イルがシンボリックリンクの場合、リンクがどこを指しているを読み取りそのファイ

ルを指すリンクを作成します。ファイルがディレクトリの場合、目的となるディレクト

リを作成しディレクトリのエントリを読み、各エントリに対して再帰的に copy\_rec

を呼び出します。このときディレクトリそのものと親ディレクトリのエントリは無視

します。これ以外のファイルについては警告を出して無視します。

メインプログラムは単純です:

let copyrec () =

if Array.length Sys.argv <> 3 then begin

prerr_endline ("Usage: " ^Sys.argv.(0)^ " <source> <destination>");

exit 2

end else begin

copy_rec Sys.argv.(1) Sys.argv.(2);

exit 0

end

;;

handle_unix_error copyrec ();;

完全な例: TAR

練習問題 6 ハードリンクを賢くコピーしてください。同じファイルが n 個の異なる
場所に存在する場合、上記のプログラムでは copy_rec は同じファイルを n 個作成し
ます。このような状況を検出し、コピーを一度だけして他の場所にはハードリンクを

作るようにしてください。解答は 113 ページ。

* * *

2.16 完全な例: TAR

tar ファイルフォーマット (tape archive の略です) はファイル階層を一つのファイ
ルに保存します。tar ファイルは小さなファイルシステムと見ることができます。

このセクションでは tar ファイルを読み書きする関数を定義します。そのほかに

readtarという、readtar aでアーカイブ aに含まれるファイルを表示し、readtar a f

でアーカイブ a に含まれるファイル f を取り出すコマンドも作ります。ファイル階層

全体を取り出すこととファイル階層からアーカイブを作ることは練習問題として読者

に残します。

ファイルフォーマットの仕様 tar アーカイブは複数のレコードから成ります。それ

ぞれのレコードがファイルを表します。レコードはファイルについての情報 (名前、種
類、サイズ、所有者など) をエンコードするヘッダから始まり、ファイルの内容がその
後に続きます。ヘッダは 512 バイトのブロックで、表 3 のような構造をしています。
ファイルの内容はヘッダのすぐ後ろに保存され、サイズは 512 バイトの倍数まで

0 で拡張されます。レコードの後には別のレコードが続きます。ファイルは最低 20 ブ
ロック (1 ブロックは 512 バイト) を持つように空のブロックでパディングされます。

tar アーカイブは脆い媒体に保存されて何年もしてから読み込まれることを想定

しているので、ヘッダが傷ついたことを検出するための checksum フィールドがあり

ます。その値はヘッダ内の全てのバイトの和です (チェックサムを計算するときには
checksum フィールド自身は 0 として計算します)。
ヘッダの kind フィールドはファイルの種類を以下のように 1 バイトにエンコー

ドします5:

'\0' or '0' '1' '2' '3' '4' '5' '6' '7'

REG LINK LNK CHR BLK DIR FIFO CONT

ほとんどの場合 kind フィールドの値は stats 構造体の st_kind フィールドに保存さ

れているUnix のファイルの種類 file_kind に対応します。LINK はアーカイブに保存

されたファイルに対するハードリンクを表します。CONT はメモリの連続した領域に保

存された通常ファイルを表します (これはいくつかのファイルシステムが持つ機能で
あり、通常ファイルと同じように扱うことができます)。
ヘッダの kind フィールドが LINK または LNK のとき、link フィールドにはリン

クの指す先のファイル名が保存されます。kind フィールドが CHR または BLK のとき、

major と minor フィールドには周辺機器のメジャー番号とマイナー番号が保存されま

す。これらのフィールドはそれ以外のとき使用されません。

kind フィールドの値はヴァリアント型によって、ヘッダはレコードによって自然

に表現されます。

5このフィールドは他のフィールドの値が大きすぎて予約されたサイズより大きくなった場合や、tar

フォーマットの拡張などの例外的なケースをエンコードするための値も取ります。

オフセット 長さ コードの種類 名前 説明

0 100 文字列 name ファイルの名前

100 8 8進 perm ファイルの権限

108 8 8進 uid 所有ユーザの ID

116 8 8進 gid 所有グループの ID

124 12 8進 size ファイルのサイズ (単位はバイト)

136 12 8進 mtime 最終更新日

148 8 8進 checksum ヘッダのチェックサム

156 1 文字 kind ファイルの種類

157 100 8進 link リンク

257 8 文字列 magic シグネチャ ("ustar\032\032\0")

265 32 文字列 user 所有ユーザの名前

297 32 文字列 group 所有グループの名前

329 8 8進 major 周辺機器のメジャー番号

337 8 8進 minor 周辺機器のマイナー番号

345 167 パディング

注意 フィールドの長さの単位はバイト。全てのフィールドはヌル文字 '\000' で終わる文字

列でエンコードされるが、フィールド kind と size については終端の '\000' は無くても

良い。

Table 3 — ヘッダの構造

type kind =

| REG | LNK of string | LINK of string | CHR of int * int

| BLK of int * int | DIR | FIFO | CONT

type header =

{ name : string; perm : int; uid : int; gid : int; size : int;

mtime : int; kind : kind; user : string; group : string }

ヘッダの読み込み ヘッダの読み込みはあまり面白い処理ではありませんが、無視す

ることもできません。

exception Error of string * string

let error err mes = raise (Error (err, mes));;

let handle_error f s =

try f s with

| Error (err, mes) ->

Printf.eprintf "Error: %s: %s" err mes;

exit 2

let substring s offset len =

let max_length = min (offset + len + 1) (Bytes.length s) in

let rec real_length j =

if j < max_length && s.[j] <> '\000' then real_length (succ j)

else j - offset in

Bytes.sub s offset (real_length offset);;

let integer_of_octal nbytes s offset =

let i = int_of_string ("0o" ^ substring s offset nbytes) in

if i < 0 then error "Corrupted archive" "integer too large" else i;;

let kind s i = match s.[i] with

| '\000' | '0' -> REG

| '1' -> LINK (substring s (succ i) 99)

| '2' -> LNK (substring s (succ i) 99)

| '3' -> CHR (integer_of_octal 8 s 329, integer_of_octal 8 s 329)

| '4' -> BLK (integer_of_octal 8 s 329, integer_of_octal 8 s 337)

| '5' -> DIR | '6' -> FIFO | '7' -> CONT

| _ -> error "Corrupted archive" "kind"

let header_of_string s =

{ name = substring s 0 99;

perm = integer_of_octal 8 s 100;

uid = integer_of_octal 8 s 108;

gid = integer_of_octal 8 s 116;

size = integer_of_octal 12 s 124;

mtime = integer_of_octal 12 s 136;

kind = kind s 156;

user = substring s 265 32;

group = substring s 297 32; }

let block_size = 512;;

let total_size size =

block_size + ((block_size -1 + size) / block_size) * block_size;;

アーカイブの終端は本来なら新しいレコードが始まるべき場所にあるファイルの終端

か、完全で空のブロックです。そのためヘッダを読み込むときに読むブロックは空な

ものか完全なものです。そこで really_read を再利用します。アーカイブが壊れてい

ない限り、1 ブロックを読み込もうとしたときにファイルの終端を読むことはありま
せん。

let buffer_size = block_size;;

let buffer = Bytes.create buffer_size;;

let end_of_file_error () =

error "Corrupted archive" "unexpected end of file"

let without_end_of_file f x =

try f x with End_of_file -> end_of_file_error ()

let read_header fd =

let len = read fd buffer 0 buffer_size in

if len = 0 || buffer.[0] = '\000' then None

else begin

if len < buffer_size then

without_end_of_file (really_read fd buffer len) (buffer_size - len);

Some (header_of_string buffer)

end;;

アーカイブの読み込み アーカイブに操作を行うには、操作の対象を見つけるまでレ

コードを順に読んでいく必要があります。通常はそれぞれのレコードのヘッダだけを

読みこむだけですみますが、前に読み込んだアーカイブに戻ってその内容を読む必要

があることもあります。そのような場合のためにそれぞれのレコードごとにそのヘッ

ダとアーカイブ内の位置を記録しておきます。

type record = { header : header; offset : int; descr : file_descr };;

アーカイブのレコード (ファイルの内容は除く) を読み込んで記録する一般的なイテ
レータを定義します。イテレータを一般的にするために、蓄積のための関数 f は抽象

的なものにしておきます。こうすることでレコードの表示や破壊などの処理にも同じ

イテレータ関数を使うことができます。

let fold f initial fd =

let rec fold_aux offset accu =

ignore (without_end_of_file (lseek fd offset) SEEK_SET);

match without_end_of_file read_header fd with

Some h ->

let r =

{ header = h; offset = offset + block_size; descr = fd } in

fold_aux (offset + total_size h.size) (f r accu)

| None -> accu in

fold_aux 0 initial;;

fold_aux 関数は処理を offset の位置から開始し、accu の中に途中経過が含まれてい

ます。レコードが始まる位置 offset まで移動し、ヘッダを読み、レコード r を構築

し、同じ処理を新しい (より処理の進んだ) 途中結果 f r accu とともにレコードの末

尾から行います。この処理はヘッダが無くなるまで、つまりアーカイブの終端に達す

るまで繰り返されます。

レコードの名前の表示 fold 関数の使用例として、レコードの名前を保存すること無

く表示する処理を示します:

let list tarfile =

let fd = openfile tarfile [ O_RDONLY ] 0o0 in

let add r () = print_string r.header.name; print_newline () in

fold add () fd;

close fd

レコードの内容を表示する コマンド readtar a f はアーカイブの中のファイル f を

探索し、もしそれが通常ファイルならばその内容を表示します。f がアーカイブ内の

ファイル g に対するハードリンクであれば、アーカイブの中では別になっていたとし

ても本当は二つのファイルは同一なので、そのリンクをたどって g の内容を表示しま

す。g と f のどちらがリンクでどちらがリンク先であるかはアーカイブが作られると

きにどちらが先に探索されたかのみに依存します。ここではシンボリックリンクを追

うことはしません。

ハードリンクの解決は以下の相互再帰関数によって行われます:

let rec find_regular r list = match r.header.kind with

| REG | CONT -> r

| LINK name -> find_file name list

| _ -> error r.header.name "Not a regular file"

and find_file name list = match list with

| r :: rest ->

if r.header.name = name then find_regular r rest

else find_file name rest

| [] -> error name "Link not found (corrupted archive)";;

find_regular 関数はレコード r に対応する通常ファイルを探します。r が通常ファ

イルならば r を返します。r がハードリンクならば find_file 関数を使ってリンクの

指すファイルをアーカイブにすでに保存されているレコード list の中から探します。

それ以外の場合は関数は失敗します。

レコードが見つかった場合はその内容を表示します。ディスクリプタをレコード

の開始地点に移動させた後は file_copy とよく似た処理になります。

let copy_file file output =

ignore (lseek file.descr file.offset SEEK_SET);

let rec copy_loop len =

if len > 0 then

match read file.descr buffer 0 (min buffer_size len) with

| 0 -> end_of_file_error ()

| r -> ignore (write output buffer 0 r); copy_loop (len-r) in

copy_loop file.header.size

これらの関数を組み合わせれば完成です:

exception Done

let find_and_copy tarfile filename =

let fd = openfile tarfile [ O_RDONLY ] 0o0 in

let found_or_collect r accu =

if r.header.name = filename then begin

copy_file (find_regular r accu) stdout;

raise Done

end else r :: accu in

try

ignore (fold found_or_collect [] fd);

error "File not found" filename

with

| Done -> close fd

まずターゲットのファイル名が見つかるまでアーカイブのレコード (内容は除く) を読
みます。その後 find_regular 関数で実際にそのファイルの内容を含んでいるレコー

ドを探します。この二回目の逆順の探索はアーカイブが矛盾なく作られている限り成

功します。しかし一回目の探索はファイルがアーカイブに存在しない場合に失敗する

ので、処理が失敗した場合でも二つのエラーを区別するようになっています。

readtar コマンドを実装したメイン関数は以下のようになります:

let readtar () =

let nargs = Array.length Sys.argv in

if nargs = 2 then list Sys.argv.(1)

else if nargs = 3 then find_and_copy Sys.argv.(1) Sys.argv.(2)

else

prerr_endline ("Usage: " ^Sys.argv.(0)^ " <tarfile> [ <source> ]");;

handle_unix_error (handle_error readtar) ();;

練習問題 7 readtar コマンドを拡張して、シンボリックリンクが指しているファイ

ルがアーカイブ内に存在する場合はそのファイルの内容を表示するようにしてくださ

い。解答は 114 ページ。

* * *

練習問題 8 untar a がアーカイブ a の全てのファイル (スペシャルファイルを除く)
を抽出し新しいディレクトリとして作成するようなコマンド untar を作成してくださ

い。ファイルについての情報 (所有者、権限) は可能ならばアーカイブのものを復元し
てください。

ファイル階層は untar コマンドが実行されたディレクトリに再構築されるべきで

す。コマンドが現在のワーキングディレクトリのサブディレクトリではない場所に書

き込むことは許されません。アーカイブにレコードが無いディレクトリはユーザのデ

フォルト権限で作られるようにしてください。解答は 116 ページ。

* * *

練習問題 9 tar -xvf a f1 f2 ... が f1, f2 ... とそのサブディレクトリを含むアー
カイブ a を作成するようなコマンド tar を書いてください。解答は 118 ページ。

* * *

3 プロセス

プロセスとはオペレーティングシステム上で実行されるプログラムのことです。プロ

セスはプログラム (機械語) とその状態 (現在の実行位置、変数の値、関数呼び出しス
タック、開いているファイルディスクリプタなど) からなります。
この章では新しいプロセスを作ったり新しいプログラムを実行したりするための

Unix のシステムコールを紹介します。

3.1 プロセスの作成

システムコール fork はプロセスを作成します。

val fork : unit -> int

fork を呼び出した 親プロセス のほぼ完璧な複製である 子プロセス が新しく作られ

ます。二つのプログラムは同じプログラムを同じ実行位置 (fork から返った位置) か
ら実行します。このとき全ての変数は同じ値を持ち、スタックは同一で、開かれてい

るファイルディスクリプタも同じです。二つのプロセスを区別する唯一のものは fork

の返り値です。子プロセスでは fork は 0 を返し、親プロセスでは 0 でない整数を返
します。fork の返り値を確認することで、プログラムは自分が親なのか子なのかを確

認してそれによって動作を変えることができます。

match fork () with

| 0 -> (* 子プロセスだけが実行するコード *)

| pid -> (* 親プロセスだけが実行するコード *)

fork によって親プロセスに返される 0 でない整数は子プロセスの プロセス ID です。
カーネルはプロセス IDを使ってプロセスを一意に識別します。プロセスは getpid 関

数を呼ぶことでプロセス ID を取得できます。
子プロセスは親プロセスと同じ状態 (同じ変数の値、同じファイルディスクリプタ)

に初期化されます。この状態は親と子で共有されるのではなく、fork が呼ばれたとき

にコピーされます。例えば fork の前に定義した参照変数があった場合、fork の後に

は親と子プロセスはこの参照を互いに影響を及ぼすこと無く独立に変更できます。

同様にファイルディスクリプタも fork が呼ばれたときにコピーされます。そのた

め一方を閉じたとしてももう一方は開いたままです。ただし二つのディスクリプタは

(システムメモリにある) ファイルテーブル内の同じエントリを指すので、入出力の現
在位置を共有します。親と子のどちらかが読み込みを行った場合、その次に読み込む

のがどちらであっても読み込み位置は変化します。また lseek による入出力位置の変

更はもう一方のプロセスにすぐに伝わります。

3.2 完全な例: leave コマンド

leave hhmm コマンドは時刻 hhmm に利用を終える時間だとユーザに報告するバックグ

ラウンドプロセスをフォークしてすぐに終了します。このコマンドを作成します。

1 open Sys;;

2 open Unix;;

3

4 let leave () =

5 let hh = int_of_string (String.sub Sys.argv.(1) 0 2)

6 and mm = int_of_string (String.sub Sys.argv.(1) 2 2) in

7 let now = localtime(time ()) in

8 let delay = (hh - now.tm_hour) * 3600 + (mm - now.tm_min) * 60 in

9

10 if delay <= 0 then begin

11 print_endline "Hey! That time has already passed!";

12 exit 0

13 end;

14 if fork () <> 0 then exit 0;

15 sleep delay;

16 print_endline "\007\007\007Time to leave!";

17 exit 0;;

18

19 handle_unix_error leave ();;

プログラムは最初にコマンドラインをパースして時刻を取得し、報告するまでの秒数

を計算します (9 行目) 。time 関数はエポック (1970 年 1 月 1 日 午前 0 時 0 分 0
秒) から現在時刻までの経過秒数を返します。localtime を使うとこの値から年、月、
日、時、分、秒を計算することができます。プログラムはその後 fork で新しいプロセ

スを作ります。親プロセス (fork の返り値が 0 でないプロセス) はすぐに終了するた
め、leave を起動したシェルの制御はすぐにユーザに戻ります。子プロセス (fork の
返り値が 0 のプロセス) の実行は続き、sleep を呼んで指定された時間まで待ってか

らメッセージを表示して終了します。

3.3 プロセスの終了を待つ

システムコール wait は fork によって作られた子プロセスの一つが終了するまで待

ち、そのプロセスがどのように終了したかについての情報を返します。これは親子間

の同期メカニズムであり、子から親へのとても原始的な形のコミュニケーションです。

val wait : unit -> int * process_status

val waitpid : wait_flag list -> int -> int * process_status

基礎となるシステムコールは waitpidであり、wait ()という呼び出しは waitpid [] (-1)

の短縮形に過ぎません。waitpid [] p の動作は p の値によって異なります。

• p > 0 ならば、プロセス ID が p である子プロセスの終了を待つ。

• p = 0 ならば、同じグループ ID を持つ任意の子プロセスの終了を待つ。

• p = −1 ならば、任意の子プロセスの終了を待つ。

• p < −1 ならば、グループ ID が -p である子プロセスの終了を待つ。

返り値の最初の要素は wait によって終了を捕捉された子プロセスのプロセス ID で
す。2番目の要素は以下に示す process_status 型の値です:

WEXITED r 子プロセスは exit が呼ばれるかプログラムの終端に達することに

よって通常の方法で終了した。r はリターンコード (exit の引数) を
表す。

WSIGNALED s 子プロセスはシグナル (ctrl+C, kill など。4 章参照) によって終了
した。s がシグナルの種類を表す。

WSTOPPED s 子プロセスはシグナル s によって停止された。これが起こるのはあ

るプロセス (典型的にはデバッガ) が他のプロセスの実行を (ptrace
を使って) モニターしているという特殊なケースに限られる。

子プロセスの一つが wait を呼んだ時点ですでに終了していた場合は呼び出しはすぐ

に返ります。そうでなければ親プロセスは子プロセスのどれかが終了するまでブロッ

クします (“ランデブー”と呼ばれる動作です)。この子プロセスの終了を待つには wait

を n 回呼ぶ必要があります。
waitpid 関数は二つのオプショナルなフラグを第一引数に受け取ります。一つ目の

WNOHANG フラグは終了していない子プロセスが無い場合に待たないことを指示します。

子プロセスが無かった場合の返り値は第一要素が 0 で第二要素は未定義です。もう一
つの WUNTRACED フラグは sigstop シグナルによって停止させられた子プロセスを返

すことを指示します。waitpid は p に該当する子プロセスがないとき (あるいは p が

-1 で現在のプロセスが子プロセスを持たないとき)には例外を出します。

例 以下の fork_search 関数は二つのプロセスを使って線形探索を行います。線形探

索には simple_search 関数を使っています。

1 open Unix;;

2 exception Found;;

3

4 let simple_search cond v =

5 try

6 for i = 0 to Array.length v - 1 do

7 if cond v.(i) then raise Found

8 done;

9 false

10 with Found -> true;;

11

12 let fork_search cond v =

13 let n = Array.length v in

14 match fork () with

15 | 0 ->

16 let found = simple_search cond (Array.sub v (n/2) (n-n/2)) in

17 exit (if found then 0 else 1)

18 | _ ->

19 let found = simple_search cond (Array.sub v 0 (n/2)) in

20 match wait () with

21 | (pid, WEXITED retcode) -> found || (retcode = 0)

22 | (pid, _) -> failwith "fork_search";;

fork された子プロセスはテーブルの上半分を探索し、cond を満たす要素を見つけた

場合は 1 を、それ以外の場合は 0 をリターンコードとして終了します ( 16 行目と 17
行目)。親プロセスはテーブルの下半分を探索し、wait を呼んで子プロセスと同期し

ます ( 21 行目と 22行目)。子プロセスが通常の方法で終了した場合、そのリターン
コードとテーブルの下半分の探索結果を組み合わせます。そうでなければエラーが起

こっているので、fork_search 関数は失敗します。

* * *
wait はプロセス間の同期を行いますが、それ以外に子プロセスが持つリソースの

完全な開放も行います。終了した子プロセスは “ゾンビ” 状態となり大部分のリソー
ス (メモリなど) が開放されますが、子プロセスは wait を呼んだ親プロセスに返り

値を伝える必要があるので、プロセステーブルのスロットには乗ったままです。親プ

ロセスが wait を呼べば、子プロセスはプロセステーブルからも削除されます。この

テーブルの大きさは固定なので、リークを防ぐためにもフォークした全てのプロセス

を wait することが重要です。

親プロセスが子プロセスよりも先に終了した場合、子の親はプロセス ID が 1 の
プロセス (通常は init) に移ります。このプロセスは wait の無限ループを含むので、

子プロセスは終了するとすぐに回収されます。この仕組みによって “ダブルフォーク”
という便利なテクニックが使えるようになります。このテクニックは子プロセスの終

了をブロックして待つことができないときなどに使われます。

match fork () with

| 0 -> if fork () <> 0 then exit 0;

(* 子プロセスの処理を行う *)

| _ -> wait ();

(* 親プロセスの処理を行う *)

子プロセスは二回目のフォークの後すぐに終了します。これによって孫プロセスは親

を失うので、init の養子となります。この方法ではゾンビプロセスが生まれることは

ありません。親はフォーク後すぐに wait を呼んで子を回収します。子はすぐに終了

するので、この wait が長い間ブロックすることはありません。

3.4 プログラムの起動

システムコール execve、execv、そして execvp は現在のプロセスでプログラムを起動

します。現在のプログラムの実行を止めて新しいプログラムに移るので、エラーの場

合を除いてこの呼び出しが返ることはありません。

val execve : string -> string array -> string array -> unit

val execv : string -> string array -> unit

val execvp : string -> string array -> unit

第一引数は実行するプログラムを含むファイルの名前です。execvp を使った場合ファ

イルの名前は (環境変数 PATH で指定される) 探索パスのディレクトリから探索され
ます。

第二引数はプログラムを実行するときに渡されるコマンドライン引数の配列です。

実行するプログラムの中ではこの配列が Sys.argv となります。

execve を使うと第三引数にプログラムが実行される環境を渡すことができます。

execv と execvp では現在の環境がそのまま使われます。

execve と execv、そして execvp が結果を返すことはありません。エラーが起こ

ること無くプロセスが指定されたプログラムを実行するか、実行ファイルが見つから

ないなどのエラーが起きて呼び出し元のプログラムに Unix_error を出すかのどちら

かです。

完全な例: ミニシェル

例 次の三つは同じ動作をします:

execve "/bin/ls" [|"ls"; "-l"; "/tmp"|] (environment ())

execv "/bin/ls" [|"ls"; "-l"; "/tmp"|]

execvp "ls" [|"ls"; "-l"; "/tmp"|]

* * *
例 受け取った grep コマンドへの引数に -i オプション (大文字と小文字を区別しな
い) を追加して起動するための “ラッパー” コマンドは以下のように書けます:

open Sys;;

open Unix;;

let grep () =

execvp "grep"

(Array.concat

[ [|"grep"; "-i"|];

(Array.sub Sys.argv 1 (Array.length Sys.argv - 1)) ])

;;

handle_unix_error grep ();;

* * *
例 emacs コマンドをターミナルのタイプを変えて起動するための “ラッパー” コマ
ンドは以下のように書けます:

open Sys;;

open Unix;;

let emacs () =

execve "/usr/bin/emacs" Sys.argv

(Array.concat [ [|"TERM=hacked-xterm"|]; (environment ()) ]);;

handle_unix_error emacs ();;

* * *
exec を呼んだプロセスは新しいプログラムを実行するプロセスと同じです。その

ため新しいプログラムは exec を呼んだプログラムの実行環境の一部を引き継ぎ、以

下に上げるものは同じになります:
• プロセス ID と親プロセス

• 標準出力、標準入力、標準エラー出力

• 無視されるシグナル (4 章を参照)

3.5 完全な例: ミニシェル

次のプログラムは単純なコマンドインタープリターです。標準入力から行入力を読み、

単語ごとに区切り、コマンドを起動し、標準入力から EOF を受け取るまでこれを繰
り返します。文字列を単語のリストに分割する関数から始めます。このひどい処理に

ついてはどうかノーコメントとさせてください。

open Unix;;

open Printf;;

完全な例: ミニシェル

let split_words s =

let rec skip_blanks i =

if i < String.length s & s.[i] = ' '

then skip_blanks (i+1)

else i in

let rec split start i =

if i >= String.length s then

[String.sub s start (i-start)]

else if s.[i] = ' ' then

let j = skip_blanks i in

String.sub s start (i-start) :: split j j

else

split start (i+1) in

Array.of_list (split 0 0);;

次はインタープリターのメイン処理です:

let exec_command cmd =

try execvp cmd.(0) cmd

with Unix_error(err, _, _) ->

printf "Cannot execute %s : %s\n%!"

cmd.(0) (error_message err);

exit 255

let print_status program status =

match status with

| WEXITED 255 -> ()

| WEXITED status ->

printf "%s exited with code %d\n%!" program status;

| WSIGNALED signal ->

printf "%s killed by signal %d\n%!" program signal;

| WSTOPPED signal ->

printf "%s stopped (???)\n%!" program;;

exec_command 関数がコマンドを実行とエラーの対処を行います。リターンコード 255
はコマンドが実行されなかったことを意味します (これは通常の慣習ではありません。
リターンコード 255 で終了するプログラムはほとんど無いはずだという想定からこの
ようにしています)。print_status は終了プロセスが返した状態をデコードして出力

します。

let minishell () =

try

while true do

let cmd = input_line Pervasives.stdin in

let words = split_words cmd in

match fork () with

| 0 -> exec_command words

| pid_son ->

let pid, status = wait () in

print_status "Program" status

done

完全な例: ミニシェル

with End_of_file -> ()

;;

handle_unix_error minishell ();;

input_line 関数は EOF に達すると End_of_file を出すので、これをもってループの

終了とします。その後は行入力を単語に区切ってから fork を呼び出します。子プロ

セスは exec_command を呼んでコマンドを実行します。親プロセスは wait を使ってコ

マンドの終了を待った後、wait の返す子プロセスの状態を出力します。

練習問題 10 コマンドの最後に & が付いている場合にコマンドをバックグラウンド

で実行する機能を追加してください。解答は 122 ページ。

* * *

4 シグナル

シグナル、あるいはソフトウェア割り込みはプログラムの処理を切り替えるためのプ

ログラム外部からの非同期イベントです。シグナルはプログラムの実行中のどんなと

きにでも起こりえます。この点において、read 関数やパイプ (5 章参照) のような外
部からのメッセージを明示的に待つプロセス間通信とは異なっています。

シグナルによって伝わるのはシグナルの種類という最小限の情報だけです。シグ

ナルはもともとプロセス間通信を目的とするものではありませんでしたが、外部装置

(つまりシステムや他のプロセス) の状態の原始的な情報を送ることでプロセス間通信
を可能にしています。

4.1 デフォルトの動作

プロセスがシグナルを受け取ったときの動作としてありえるのは次の四つです:

• シグナルがプロセスを終了させる。加えてシステムはプロセスの状態のイメージをコ
アファイルに書き込むことができる (コアダンプと呼ばれ、あとでデバッガを使って
調べられる)。

• シグナルはプロセスの実行を停止させるが、メモリー上に保持する。親プロセス (た
いていはシェル) は終了しないので、あとで追加のシグナルを送ることでプロセスの
実行をフォアグラウンドまたはバックグラウンドで再開させることができる。

• シグナルは完全に無視され、何も起きない。

• シグナルを受け取ったプロセスで関連づいた関数が実行される。それまでのプロセス
の実行は関数が返ってから再開される。

シグナルにはいくつか種類があり、それぞれ特定のイベントと結びついています。

表 4 にシグナル (の一部) とそのデフォルトの動作を示します。
プロセスが受け取るシグナルはいくつかの方法で送られます:

• ユーザがキーボードで送る。コンソールで ctrl-C と打つことで sigint シグナルを

ターミナルがフォアグラウンドで実行しているプロセスに送ることができる。同様に

ctrl-\ は sigquit シグナルを送る1。端末が自分自身を終了するか、ネットワークの
リンクが切れることで端末が終了したとき、sighup シグナル送られる。

• ユーザがシェルコマンド kill で送る。kill を使うと特定のプロセスに特定のシグナ

ルを送ることができる。例えば kill -KILL 194 は sigkill シグナルを ID が 194 の
プロセスに送り、プロセスを停止させる

• 他のプログラムがシステムコール kill で送る (一つ前のケースと同じ)。

• プロセスの誤った動作を受けてシステムが送る。例えばゼロによる除算を行おうとし
たプロセスには sigfpe が送られる。

• システムが実行環境が変化したことを通知するために送る。例えば子プロセスが終了
すると、親には sigchld シグナルが送られる。

1これらは端末のデフォルトのキーであり、端末の設定を返ることで変更できます。2.13 節参照。

シグナルの利用

名前 イベント デフォルトの動作

sighup ハングアップ (接続の終了) 終了

sigint 割り込み (ctrl-C) 終了

sigquit 終了 (ctrl-\) 終了 & コアダンプ

sigfpe 算術エラー (0 による除算) 終了 & コアダンプ

sigkill 中止 (無視できない) 終了

sigsegv 不正なメモリ参照 終了 & コアダンプ

sigpipe 読み込み先のいないパイプへの書き出し 終了

sigalrm タイマー割り込み 無視

sigtstp 一時中断 (ctrl-Z) 中断

sigcont 中断したプロセスの再開 無視

sigchld 子プロセスが終了あるいは停止した 無視

Table 4 — シグナル (一部) とそのデフォルトの動作

4.2 シグナルの利用

システムコール kill を使うとプロセスにシグナルを送ることができます。

val kill : int -> int -> unit

第一引数はシグナルを送るプロセスの ID で、第二引数は送るシグナルの番号です。
kill を呼んだユーザによって所有されていないプロセスにシグナルを送ろうとすると

エラーとなります。プロセスは自分自身にシグナルを送ることができます。システム

コール kill が値を返した場合、シグナルが目的のプロセスに届いたことが保証され

ます。

プロセスが同じシグナルを短い間に何度も受け取った場合、そのシグナルに対応す

るコードは一度しか実行しないことがあります。そのためプロセスは受け取ったシグ

ナルの数を数えることはできず、数えられるのはシグナルに反応した回数となります。

システムコール alarm を使うとシステムクロックを使って割り込みを予約するこ

とができます。

val alarm : int -> int

alarm s という呼び出しはすぐに返りますが、その後少なくとも s 秒後 (最大で何秒
後かについての保証はありません) に sigalrm シグナルが送られます。この呼び出し

が返すのはこれまでの alarm によって設定されたアラームまでの残り秒数です。s が

0 の場合、これまでのアラームをキャンセルします。

4.3 シグナルに対する動作の変更

システムコール signal を使うとプロセスが指定した種類のシグナルを受け取ったと

きの動作を変更できます。

シグナルに対する動作の変更

val signal : int -> signal_behavior -> signal_behavior

第一引数は動作を変更するシグナルの番号で、signal_behavior 型の第二引数はその

シグナルに対応する動作を指定します。signal_behavior には以下の値があります:

Signal_ignore シグナルを無視する。

Signal_default デフォルトの動作を行う。

Signal_handle f シグナルを受け取ると f を実行する。

システムコール fork によってプロセスをフォークしてもシグナルに関する動作は

引き継がれます。フォーク直後の子プロセスのシグナルへの動作は fork が実行され

た時点での親プロセスのものと同じです。システムコール execve は無視されている

シグナルは無視されたままにし、それ以外のシグナルについては Signal_default に

動作を変更します。

例 ログオフしたりセッションを終了した後でもバックグラウンドでタスク (大規模
な計算、あるいは “スパイウェア” など)を実行したままにしておきたいことがありま
す。プロセスの sighup シグナル (ユーザが接続を終了すると送られます) に対するデ
フォルトの動作はプロセスの終了なので、このシグナルを無視するように変更すれば

プロセスを実行されたままにすることができます。Unix コマンド nohup はまさにこ

のことを行います:
nohup cmd arg1 ... argn

nohup はコマンド cmd arg1 ... argn を fighup に影響されない状態で実行します

(バックグラウンドで起動された全てのプロセスに対して自動的に nohup を実行する

シェルもあります)。この nohup は 3 行で実装できます:
open Sys;;

signal sighup Signal_ignore;;

Unix.execvp argv.(1) (Array.sub argv 1 (Array.length argv - 1));;

システムコール execvp は sighup が無視される状態を引き継ぎます。

* * *
例 異常動作をしたプログラムの注意深い終了処理。例えば tar のようなプログラム

は、異常動作によって終了することになっても終了する前にファイルの重要な情報を

書き込んだり壊れたファイルを削除することが望ましいです。以下のコードをプログ

ラムの最初に書けばプログラムが終了するときの動作を設定することができます。

signal sigquit (Signal_handle quit);

signal sigsegv (Signal_handle quit);

signal sigfpe (Signal_handle quit);

ここで quit は以下のような形をしています:
let quit _ =

(* ファイルの重要な情報の書き込みを試みる *)

exit 100;;

* * *
例 ユーザが発した割り込みのキャプチャ。インタラクティブなプログラムではユー

ザが ctrl-C を押したときにメインループを抜けるようになっていることがあります。

sigint シグナルを受け取ったときに例外を出すようにすれば実装できます。

47

http://caml.inria.fr/pub/docs/manual-ocaml/libref/Sys.html#VALsignal
http://caml.inria.fr/pub/docs/manual-ocaml/libref/Sys.html#TYPEsignal_behavior
http://caml.inria.fr/pub/docs/manual-ocaml/libref/Unix.html#VALfork
http://caml.inria.fr/pub/docs/manual-ocaml/libref/Unix.html#VALexecve
http://caml.inria.fr/pub/docs/manual-ocaml/libref/Unix.html#VALexecvp


シグナルのマスク

exception Break;;

let break _ = raise Break;;

...

let main_loop () =

signal sigint (Signal_handle break);

while true do

try (* ユーザのコマンドを読み込んで評価する *)

with Break -> (* "stopped" と表示する *)

done;;

* * *
例 アニメーションなどのメインプログラムと切り離された周期的なタスクの実行。

例えば次のプログラムメインプログラムの動作 (計算、入出力) に関係なく 30 秒ごと
に “ビープ” 音を鳴らします。

let beep _ =

output_char stdout '\007'; flush stdout;

ignore (alarm 30);;

...

signal sigalrm (Signal_handle beep); ignore (alarm 30);;

* * *
チェックポイント

シグナルは非同期コミュニケーションに便利です — 実際それが存在理由です — が、

この非同期であるという特徴のせいでシグナルはシステムプログラミングにおける難

しい部分になっています。

シグナルハンドラは非同期に実行されるので、この関数とプロセスのメインプロ

グラムとは擬似的に並列な状態で実行されます。シグナルハンドラが値を返せないこ

とから、通常この関数は通常グローバル変数を変更します。したがってメインプログ

ラムも同時にその変数を変更しようとした場合、競合状態に陥ること可能性がありま

す。これに対する一つの解決法は次の節で説明するようにシグナルハンドラが変更す

る変数をメインプログラムが変更するときにはシグナルを一時的にブロックすること

です。

厳密にいうと、OCamlはシグナルを非同期に扱いません。OCamlは受け取ったシ
グナルを記録しますが、シグナルハンドラが実行されるのは特定の チェックポイント

においてだけです。チェックポイントはハンドラが非同期に実行されると考えても良

い程度に頻繁に設けられています。通常アロケーションやループ制御、システムとの

やり取り (システムコールを含む) のときにチェックポイントが設けられます。ルー
プが含まれず、アロケーションをせず、システムとのやり取りをしないプログラムに

ついて、OCamlはメインプログラムとシグナルハンドラの実行が互い違いにならない
ことを保証しています。そのため例えばアロケートされない値 (整数および真偽値な
ど。小数値は含まれません!) の参照セルは上記の状況でも競合状態に陥りません。

4.4 シグナルのマスク

シグナルはブロックできます。ブロックされたシグナルは無視されるわけではなく、

後で届けられるよう待機状態になります。システムコール sigprocmask を使うとシグ

ナルのマスクすることができます。

シグナルとシステムコール

val sigprocmask : sigprocmask_command -> int list -> int list

sigprocmask cmd sigs はブロックするシグナルを変更し、この呼び出しの前にブロッ

クされていたシグナルのリストを返します。返り値によってマスクを元の状態に戻す

ことが可能になります。引数 sigs はシグナルのリストであり、sigprocmask_command

型の値 cmd によって呼び出しの効果が異なります:

SIG_BLOCK sigs 内のシグナルがブロックするシグナルのリストに追加される。

SIG_UNBLOCK sigs 内のシグナルがブロックするシグナルのリストから削除される。

SIG_SETMASK sigs 内のシグナルがブロックするシグナルとなる。

典型的な sigprocmask の利用法はあるシグナルをマスクすることです。

let old_mask = sigprocmask cmd sigs in

(* 処理を行う *)

let _ = sigprocmask SIG_SETMASK old_mask

次のパターンを使うと起こりがちなエラーを防ぐことができます。

let old_mask = sigprocmask cmd sigs in

let treat () = (* 処理を行う *) in

let reset () = ignore (sigprocmask SIG_SETMASK old_mask) in

Misc.try_finalize treat () reset ()

4.5 シグナルとシステムコール

いくつかのシステムコールは無視されていないシグナルによって中断されます。この

ようなシステムコールは 遅い システムコールと呼ばれる、実行にいくらでも長い時間

がかかりうるもの (例えば端末との i/oや select、system など) です。割り込みが起
こった場合、これらのシステムコールは実行を完了しないまま EINTR 例外を出します。

一方ファイル i/oは割り込みされません。ファイル i/o中に実行中のプロセスを
中断して他のプロセスを実行することはありますが、ディスクが正常に機能している

ならばこの中断は常に短時間になります。ゆえにデータのスループットはシステムの

みに依存し、他のユーザのプロセスから影響を受けることはありません。

無視されたシグナルは届かず、マスクされたシグナルはマスクを解くまで届きま

せん。しかしそれ以外の場合はシステムコールをマスクしていないシステムコールか

ら守る必要があります。典型的な例は子プロセスの終了を待っている親プロセスです。

子のプロセス ID を wait とすると親は waitpid [] pid を実行しますが、waitpid は

ブロックするシステムコールなので 遅い システムコールであり、シグナルによって

中断される可能性があります。特に子プロセスが終了したときには sigchld シグナル

が親に送られます。

Miscモジュールの restart_on_EINTR関数はシステムコールが中断されたとき、す

なわち EINTR 例外が出たときにシステムコールをやり直します。

let rec restart_on_EINTR f x =

try f x with Unix_error (EINTR, _, _) -> restart_on_EINTR f x

子プロセスの終了をシグナルで中断されないように待つには、restart_on_EINTR (waitpid flags) pid

を呼びます。

シグナルとシステムコール

例 子プロセスの返り値が親プロセスにとって重要でない場合、sigchld のシグナル

ハンドラによって子プロセスを非同期に回収することも可能です。ただし短い時間に

何度も同じシグナルを受け取った場合シグナルハンドラが一度しか起動しないことが

あるので、sigchld を受け取ったときに子プロセスがいくつ終了したかを知ることは

できません。このことから、sigchld シグナルを処理するためには以下のライブラリ

関数 Misc.free_chidlren が必要になります。

let free_children _ =

try while fst (waitpid [ WNOHANG ] (-1)) > 0 do () done

with Unix_error (ECHILD, _, _) -> ()

free_children は waitpid をノンブロッキングモード (WNOHANG オプション) で実行し
て実行を終了した子プロセスを回収する、という処理を子プロセスが全て実行中とな

る (waitpid が子のプロセス ID ではなく 0 を返す) か子プロセスがなくなる (ECHILD
例外が出される) まで繰り返します。

waitpid が WNOHANG オプションでノンブロッキングとなっているので EINTR に対

する処理は必要ありません。

* * *
例 Unix モジュールの system 関数は次のように単純に定義されています:

let system cmd = match fork () with

| 0 -> begin try

execv "/bin/sh" [| "/bin/sh"; "-c"; cmd |]

with _ -> exit 127

end

| id -> snd (waitpid [] id);;

C 標準ライブラリの system 関数の規格では、親プロセスは sigint と sigquitを無

視しコマンドの実行中は sigchld をマスクするようになっています。これによって親

プロセスに影響を与えること無く子プロセスを中断したり終了させることが可能にな

ります。

ここでは system 関数をより一般的な exec_as_system 関数の特殊化として定義し

ます。この関数ではシェルを必ずしも起動されません。

1 let exec_as_system exec args =

2 let old_mask = sigprocmask SIG_BLOCK [ sigchld ] in

3 let old_int = signal sigint Signal_ignore in

4 let old_quit = signal sigquit Signal_ignore in

5 let reset () =

6 ignore (signal sigint old_int);

7 ignore (signal sigquit old_quit);

8 ignore (sigprocmask SIG_SETMASK old_mask) in

9 let system_call () = match fork () with

10 | 0 ->

11 reset ();

12 (try exec args with _ -> exit 127)

13 | k ->

14 snd (restart_on_EINTR (waitpid []) k) in

15 try_finalize system_call () reset ();;

16

17 let system cmd =

18 exec_as_system (execv "/bin/sh") [| "/bin/sh"; "-c"; cmd |];;

シグナルの変更は fork の実行の前に行う必要があることに注意してください。フォー

クを実行した後にシグナルを変更するようにすると、シグナルが変更しきる前にシグ

ナル (例えばすぐに終了した子プロセスからの sigchld) を受け取ってしまう可能性が
あるためです。シグナルの変更は子プロセスではコマンドの実行の前に 11 行目でリ
セットされます。fork と exec は無視するシグナルを保存し、fork はシグナルに対す

る動作を保存します。execは無視していないシグナルの動作をデフォルトに戻します。

最後に親プロセスはエラーが起きた場合でもシグナルの変更をリセットする必要

があります。このために 15 行目では try_finalize が使われます。

* * *

4.6 時間の経過

時刻に対するレガシーなアプローチ

Unix の初期のバージョンから、時間は秒で測られてきました。そのため互換性が重
要となるならば、常に秒単位で時間を計測するべきです。現在時刻は 1970 年 1 月 1
日 00:00:00 gmt からの経過秒数と定義されます。次の関数で取得できます:

val time : unit -> float

システムコール sleep は引数で指定した秒数だけプログラムの実行を止めます:

val time : unit -> float

しかしこの関数は原始的ではありません。さらに基本的なシステムコール alarm

(前の節を参照) と sigsuspend を使うことで sleep を実装することができます。

val sigsuspend : int list -> unit

sigsuspend l はリスト l 内のシグナルを一時的に差し止め、無視も差し止めもさ

れていないシグナルを受け取るまでプログラムの実行を停止します。値が返るときに

シグナルマスクは元の値に戻ります。

例 sleep は以下のように実装できます:

1 let sleep s =

2 let old_alarm = signal sigalrm (Signal_handle (fun s -> ())) in

3 let old_mask = sigprocmask SIG_UNBLOCK [ sigalrm ] in

4 let _ = alarm s in

5 let new_mask = List.filter (fun x -> x <> sigalrm) old_mask in

6 sigsuspend new_mask;

7 let _ = alarm 0 in

8 ignore (signal sigalrm old_alarm);

9 ignore (sigprocmask SIG_SETMASK old_mask) ;;

初期状態では何もしないことが sigalrm シグナルに対する動作です。“何もしない” と
はシグナルを無視することとは異なることに注意してください。sigalrm シグナルを

受け取ったときにプロセスが起動することを保証するために、このシグナルは無視さ

れないようにします。そのあと sigalrm 以外のマスクされているシグナルを全て差し

止めてからスタンバイ状態に移行します。この変更はアラームのシグナルの後で取り

消されます。sigsuspend がシグナルマスクを変更しないことから、9 行目は 2 行目の
直後に置くこともできます。

* * *
現代的な時刻の取り扱い

現代的なバージョンの Unix では時刻をマイクロ秒単位で測定することができます。
OCamlではマイクロ秒単位で測定した時刻は浮動小数点型で表されます。システム
コール gettimeofday は現代的なシステムにおいて time の代わりとなるものです。

val gettimeofday : unit -> float

タイマー

現在出回っている Unix では各プロセスはそれぞれ違う時間を測定する 3 つのタイ
マーを持ちます。タイマーは interval_timer 型の値として確認できます:

ITIMER_REAL 実時間 (sigalrm).

ITIMER_VIRTUAL ユーザ時間 (sigvtalrm).

ITIMER_PROF ユーザ時間とシステム時間 (sigprof).

タイマーの状態は interval_timer_status 型のレコードによって表され、各フィール

ドは以下の時間を表します (どちらも float です):

• it_interval フィールドはタイマーの周期を表す。

• it_valueフィールドはタイマーの現在の値を表す。この値が 0になった場合 sigvtalrm

シグナルが送られ、タイマーの値は it_interval にリセットされる。

二つのフィールドが 0 の場合タイマーはアクティブではありません。タイマーは次の

関数で取得および変更できます:

val getitimer : interval_timer -> interval_timer_status

val setitimer :

interval_timer -> interval_timer_status -> interval_timer_status

settimer の返り値は変更前の値です。

練習問題 11 複数のタイマーを管理するために、以下のインターフェースを持つモ

ジュールを書いてください:

module type Timer = sig

open Unix

type t

val new_timer : interval_timer -> (unit -> unit) -> t

val get_timer : t -> interval_timer_status

val set_timer : t -> interval_timer_status -> interval_timer_status

end

new_timer k f はタイプ k の新しいタイマーを作り f の実行を始めます。タイマー作

成時にはアクティブではありません。set_timer t はタイマーの値を t に設定し古い

値を返します。

シグナルの問題点

* * *
日付の計算

Unix の現代的なバージョンには日付を処理するための関数も含まれています。tm 構

造体はカレンダー (年、月など) によって表現でき、gmtime や localtime、mktime な

どによって他の単位と変換できます。

4.7 シグナルの問題点

シグナルは非同期なので、シグナルをプロセス間通信に使うと制限と困難がいくつか

あります:

• 送られる情報が少なすぎる — シグナルの種類のみ。

• シグナルプログラムの実行中いつでも起きるので、グローバル変数にアクセスするシグ
ナルハンドラは競合状態を避けるためにメインプログラムと協調しなければならない。

• シグナルをメインプログラムから使うと、シグナルに対する動作がデフォルトの動作
であったとしても、実行に時間のかかるシステムコールが中断される可能性が生まれ

ることになる。

• ライブラリ関数はシグナルの可能性を常に考慮しシステムコールの割り込みに対する
対処をしなければならない。

シグナルは制限された非同期通信であるにもかからわず、非同期通信にまつわる

困難や問題は全て含みます。そのため可能であれば利用を避けるべきです。たとえば

短い時間だけ待つためには select がアラームの代わりに使えます。ただしコマンド

ラインのインタープリターなどのシグナルを考えなければいけない状況もあります。

シグナルはおそらく Unix システムの中で最も有用でない概念です。古いバー
ジョンの Unix (System V など) ではシグナルの動作は受け取った後に自動的に
Signal_default にリセットされます。そのためシグナルハンドラは自分自身をもう一

度設定し直す必要があり、以下のように書く必要がありました:

let rec beep _ =

set_signal sigalrm (Signal_handle beep);

output_char stdout '\007'; flush stdout;

ignore (alarm 30);;

しかし問題はシグナルを受け取ってシグナルの動作が自動的に Signal_defaultに戻っ

てから set_signal が実行されるまでの短い間にシグナルを受け取った場合です。こ

の場合シグナルはハンドラを呼びだすことなく、種類に応じて無視されるかプロセス

を終了させます。

その他の Unix (bsd と Linux) はより良い動作をします。シグナルを受け取って
もその動作を置き換えず、シグナルを処理している間は他のシグナルは保留されます。

5 古典的なプロセス間通信: パイプ

これまでは見てきたのはプロセスを管理する方法とファイルを使って外部環境と通信

する方法です。コースの残りの部分では、並列に実行されるプロセス同士がコミュニ

ケーションを取りながら協調する方法を見ていきます。

5.1 パイプ

並列に実行されるプロセス同士がやり取りを行う手段として通常ファイルは十分では

ありません。例えば一つのファイルにあるプロセスが書き込み、その内容を別のプロ

セスが読み込む状況を考えてみてください。やり取りの手段としてファイルを使った

場合、読み込み側はファイルが終端に達したことを (read が 0 を返すことで)検出す
ることができますが、その理由が書き込みが終わったためなのかそれとも計算に時間

がかかっているためなのかを知ることはできません。さらにファイルはやり取りされ

るデータを全て保持する必要があるので、ディスクを不必要に圧迫します。

パイプはこのようなやり取りに向いた仕組みです。パイプは二つのファイルディ

スクリプタからなります。一つがパイプの出力を表し、もう一方がパイプの入力を表

します。パイプはシステムコール pipe で作成できます。

val pipe : unit -> file_descr * file_descr

pipe を呼ぶと (fd_in, fd_out) が返ります。fd_in は 読み込み専用 で開かれたパイ

プの出力を表すファイルディスクリプタで、fd_out は 書き込み専用 で開かれたパイ

プの入力を表すファイルディスクリプタです。パイプ自身はこれら 2 つのディスクリ
プタからのみアクセス可能なカーネルの内部オブジェクトです。またパイプはファイ

ルシステムにおいて名前を持ちません。

fd_out fd_in

パイプは先入れ先出しのキューのように振る舞います。最初にパイプに書き込ん

だものが最初にパイプから読み込まれます。書き込み (パイプの入力ディスクリプタ
への write) はパイプを満たし、パイプが満杯の場合はブロックします。ブロックは他
のプロセスがパイプの他方の端から十分な量のデータを読み込むまで続き、write に

渡された全てのデータが書き込まれるまで続きます。読み込み (パイプの出力ディス
クリプタへの read) はパイプの中身を消費します。パイプが空のとき read はパイプ

の他方の端に少なくとも 1 バイトが書き込まれるまでブロックします。read は要求

されたバイト数だけ読み込むまで待つこと無く返ります。

入出力が同じプロセスから起きるのならばパイプは役に立ちません。そのような

プロセスは大量の書き込みや空のパイプへの読み込みによって永遠にブロックしてし

まうからです。そのため普通パイプへの入出力は別々のプロセスが行います。パイプ

は名前を持たないことから、パイプを利用するプロセスの片方はパイプを作成したプ

ロセスからのフォークによって作られる必要があります。

例 次の短いコードは典型的なパイプの使用例です。

let (fd_in, fd_out) = pipe () in

match fork () with

完全な例: 並列エラトステネスのふるい

| 0 -> close fd_in; ... write fd_out buffer1 offset1 count1 ...

| pid -> close fd_out; ... read fd_in buffer2 offset2 count2 ...

fork のあとパイプの入力を指すディスクリプタは二つあり、親プロセスと子プロセス

が一つづつ持っています。出力用ディスクリプタについても同様です。

fd_out
親プロセス

fd_out
子プロセス

fd_in
親プロセス

fd_in
子プロセス

この例では子が書き込みを、親が読み取りを行います。子はパイプの出力 fd_in をク

ローズしますが、これはディスクリプタを節約しプログラミング上のミスを防ぐため

です。ディスクリプタはプロセスのメモリに確保され、フォークすると親子でメモリ

が別々になるので、子プロセスでディスクリプタをクローズしたとしても親プロセス

fd_in はその影響を受けません。パイプはシステムメモリに確保され、親プロセスの

fd_in がパイプの出力に読み込みモードで開いている限り生き続けます。同じ理由で

親プロセスはパイプの入力へのディスクリプタをクローズします。結果的に以下のよ

うな状態となります:

fd_out
子プロセス

fd_in
親プロセス

子プロセスが fd_out に書き込んだデータは親の fd_in に書き込まれます。

* * *
パイプの入力へのディスクリプタが全てクローズされかつパイプが空のとき、パイ

プの出力への read は 0 (ファイルの終端) を返します。パイプの出力へのディスクリ
プタが全てクローズされているとき、パイプの入力への write は書き込みプロセスを

終了させます。より正確に言うと、カーネルが write を呼んだプロセスに sigpipe シ

グナルを送り、このシグナルのデフォルトの動作がプロセスの終了です。sigpipe シ

グナルのハンドラが変更していた場合は、write は EPIPE エラーを出して失敗します。

5.2 完全な例: 並列エラトステネスのふるい

これは並列プログラミングの古典的な例です。ここで扱うのは素数を順に列挙しなが

ら見つかった順にインタラクティブに表示するという問題です。アルゴリズムのアイ

デアは次のようなものです。あるプロセスが 2 から順に整数を出力します。このプロ
セスに整数 p を一つ受け取って表示する別の フィルター プロセスをつなげます。

完全な例: 並列エラトステネスのふるい

ints read p
2, 3, 4, . . .

最初のフィルタープロセスは 2 を最初に受け取ることから p = 2 となります。このプ
ロセスは自分の出力と接続した新しいフィルタープロセスを作成します。そして新し

いプロセスに p の倍数でない整数を入力します。

ints filter
2n

read p
2, 3, 4, . . . 3, 5, 7, . . .

新しいフィルタープロセスは 3 を最初に受け取るので p = 3 となります。このプロセ
スは 3 の倍数をフィルターし、同じことが次のプロセスへと続きます。

ints filter
2n

filter
3n

. . . read p
2, 3, 4, . . . 3, 5, 7, . . . 5, 7, 11, . . .

このアルゴリズムは見つける素数の数よりひとつ多い数のプロセスが必要になります

が、これだとプロセスを多く作りすぎるのでUnixで直接実装できません。多くの Unix
システムではプロセス数を数十程度に制限しています。同時に実行されるプロセスが

多すぎる場合、プロセスが一つしか無い機械ではコンテキストスイッチによるコスト

によって性能が大幅に減少します。そのためこれからの実装ではプロセスは最初の n
個の素数 p1, . . . , pn を出力し、p1, . . . , pn の倍数でない整数を次のプロセスへと渡し

ます。n = 1000 程度にすればプロセス生成のコストが見えなくなります。
まず 2 から k までの整数を生成するプロセスを作ります。

open Unix;;

let input_int = input_binary_int

let output_int = output_binary_int

let generate k output =

let rec gen m =

output_int output m;

if m < k then gen (m+1)

in

gen 2;;

整数の入出力には次の関数が使われます:

val output_binary_int : out_channel -> int -> unit

val input_binary_int : in_channel -> int

標準ライブラリの output_binary_int関数は整数を表す 4バイトのバイナリをout_channel

に書き込みます。この整数は in_channel に対する input_binary_int で読み込むこと

ができます。これらの関数を使うことには二つのメリットがあります。まず、整数を

バイト表現に変換する関数を作る必要がありません1。さらに、これらの関数はバッ
ファされた i/oを使うのでシステムコールの回数が減り、パフォーマンスが上昇しま
す。次の関数は i/oをバッファするための in_channel または out_channel を作成し

ます。

1ここで紹介したライブラリ関数によって使われる内部表現は規定されていませんが、言語のバージョ

ンが同じならばプラットフォーム非依存であることが保証されています。

完全な例: 並列エラトステネスのふるい

val in_channel_of_descr : file_descr -> in_channel

val out_channel_of_descr : file_descr -> out_channel

これらの関数を使うと間接的に手に入れたディスクリプタ、あるいはファイルを開い

ていないディスクリプタに対してバッファされた i/oを行うことができます。バッ
ファされた i/oをバッファされていない i/oと混ぜることはしてはいけません。混ぜ
ることは不可能ではありませんが、とてもエラーを含みやすい — 特に入力に対して

は — ので決して行うべきではありません。また一つのディスクリプタに対して二つ

以上の in_channel あるいは out_channel を開くことも可能ですが、これも危険なの

で行うべきではありません。

フィルタープロセスの話を進めます。このプロセスは補助関数 read_first_primesを

使用します。read_first_primes input countは count個の素数を input (in_channel
型の値) から読み、すでに計算された素数は読み飛ばします。count 個の素数は読み

込まれた時点で出力され、計算された素数を保存するリストに入ります。

1 let print_prime n = print_int n; print_newline ()

2

3 let read_first_primes input count =

4 let rec read_primes first_primes count =

5 if count <= 0 then first_primes else

6 let n = input_int input in

7 if List.exists (fun m -> n mod m = 0) first_primes then

8 read_primes first_primes count

9 else begin

10 print_prime n;

11 read_primes (n :: first_primes) (count - 1)

12 end

13 in

14 read_primes [] count ;;

フィルター関数は以下のようになります:
1 let rec filter input =

2 try

3 let first_primes = read_first_primes input 1000 in

4 let (fd_in, fd_out) = pipe () in

5 match fork () with

6 | 0 ->

7 close fd_out;

8 filter (in_channel_of_descr fd_in)

9 | p ->

10 close fd_in;

11 let output = out_channel_of_descr fd_out in

12 while true do

13 let n = input_int input in

14 if List.exists (fun m -> n mod m = 0) first_primes then ()

15 else output_int output n

16 done

17 with End_of_file -> ();;

フィルタは read_first_primes を呼んで最初の 1000 個の素数を出力するところから
始まります (引数 input は in_channel 型です)。そのあとパイプを作ってから子プロ

完全な例: 並列エラトステネスのふるい

セスをフォークします、親プロセスは整数を入力 input から読み、それが最初に計算

した 1000 個の素数の倍数でない場合はそれをパイプに入力します。
最後に整数を出力するプロセスと最初のフィルタープロセスをつなぐ処理を書け

ばメインプログラムの完成です。プログラムを sieve k で起動すると k より小さい素

数が列挙されます。k が省略された場合には k は max_int となります。

1 let sieve () =

2 let len = try int_of_string Sys.argv.(1) with _ -> max_int in

3 let (fd_in, fd_out) = pipe () in

4 match fork () with

5 | 0 ->

6 close fd_out;

7 filter (in_channel_of_descr fd_in)

8 | p ->

9 close fd_in;

10 generate len (out_channel_of_descr fd_out);;

11

12 handle_unix_error sieve ();;

この例では親プロセスが終了するときに子プロセスの終了を待つことはしません。

この理由は親プロセスが子プロセスの ジェネレータ であるためです。

k が与えられたとき最初に実行を終えるのは親プロセスであり、終了するとき子プ

ロセスに繋がったパイプの入力へのディスクリプタをクローズします。OCamlはプロ
セスが終了するときに書き込みモードで開かれたディスクリプタのバッファを空にす

ることから、子プロセスは親プロセスが書き込んだ最後の整数まで読みきることがで

きます。その子プロセスが終了するときについても同様です。そのためこのプログラ

ムでは子プロセスは一時的に親を失って init の子となります

k が与えられていない場合はいずれかのプロセスが終了するまで全てのプロセス

がずっと動き続けます。プロセスの終了すると上記のように子プロセスも終了します。

またプロセスが終了すると親プロセスにつながっているパイプの出力をクローズする

ので、親プロセスは次の書き込みをしたときに終了します (正確には書き込みのとき
に sigpipe を受け取り、そのデフォルトの動作によって終了します)。

練習問題 12 親プロセスが子プロセスの終了を待つようにするにはどうすればよい

でしょうか ? 解答は 123 ページ。

* * *

練習問題 13 素数が見つかったとき、print_prime 関数は print_newline () を実

行します。この実行によって出力のバッファを空にするシステムコールが呼ばれるの

で、プログラムの実行速度が制限されます。print_newline () は print_char '\n' と

flush Pervasives.stdout を実行しますが、フラッシュを省略して print_newline ()

を print_char '\n' と置き換えた場合何が起こるでしょうか? この問題を解決するに
はどうすればよいでしょうか? 解答は 123 ページ。

* * *

5.3 名前付きパイプ

いくつかの Unix システム (System V, SunOS, Ultrix, Linux など) ではファイルシ
ステム内で名前を持つパイプを作成できます。この 名前付きパイプ (あるいは fifo) を
使うと親子関係にないプロセス同士がやり取りを行うことができます。パイプを作っ

たプロセスとその子でしかやり取りのできない通常のパイプとは対称的です。

システムコール mkfifo は名前付きパイプを作成します:
val mkfifo : string -> file_perm -> unit

第一引数がパイプの名前を、第二引数がアクセス権限を表します。

名前付きパイプは openfile を使って通常ファイルと同じように開くことができま

す。名前付きパイプの入出力も通常ファイルと同じです。名前付きパイプを読み込み

専用 (あるいは書き込み専用) モードで開くとそのパイプが別のプロセスによって書
き込みモード (あるいは読み込みモード) で開かれるまでブロックします。すでに開
かれていた場合にはブロックはありません。O_NONBLOCK フラグでパイプを開くことで

ブロックを避けることができますが、この場合パイプへの入出力もブロックしなくな

ります。clear_nonblock を使うとパイプを開いた後で入出力をブロックするように変

更でき、set_nonblock を使うとブロックしないようにすることができます。

val clear_nonblock : file_descr -> unit

val set_nonblock : file_descr -> unit

5.4 ディスクリプタのリダイレクト

ここまでの説明では、シェルが cmd1 | cmd2 のようなコマンドを実行するときのよう

に標準入出力をパイプに接続する方法はわかりません。pipe で作ったパイプの両端を

指すのは 新しい ディスクリプタであり、stdin や stdout、stderr ではないからです。

この問題に対処するために、Unix には dup2 というシステムコールがあります

(dup2 は「あるディスクリプタを別のディスクリプタに複製する (duplicate to)」と読
めます) 。dup2 はあるディスクリプタを別のディスクリプタとして振る舞うようにし

ます。これが可能なのはファイルディスクリプタ (file_descr 型のオブジェクト)と
ファイルテーブルエントリ と呼ばれるカーネル内部のオブジェクトとの間に間接参照

の仕組みが存在するからです。ファイルテーブルエントリ が開かれているファイルや

パイプ、現在の入出力位置などの情報を保持します。

val dup2 : file_descr -> file_descr -> unit

dup2 fd1 fd2 を呼ぶとディスクリプタ fd2 が fd1 の指すファイルテーブルエントリ

を指すようになります。この呼び出しの後には同じファイルまたはパイプを指し、同

じ入出力位置を持つファイルディスクリプタが二つあることになります。

dup2 fd1 fd2 の前

fd1

fd2

ファイル
テーブル
エントリ 1

ファイル
テーブル
エントリ 2

dup2 fd1 fd2 の後

fd1

fd2

ファイル
テーブル
エントリ 1

ファイル
テーブル
エントリ 2

例 標準入力のリダイレクト

let fd = openfile "foo" [O_RDONLY] 0 in

dup2 fd stdin;

close fd;

execvp "bar" [|"bar"|]

dup2 を呼ぶとディスクリプタ stdin はファイル foo を指すようになります。つまり

stdin を読み込むとファイル foo から読むことになります (fd への読み込みも同様で
すが、これは使用しないのですぐに閉じます) 。stdin の設定は execvp で保存される

ので、プログラム bar は標準入力がファイル foo に繋がった状態で実行されます。こ

れはシェルで bar < foo としたときと同じ動作です。

* * *
例 標準出力のリダイレクト

let fd = openfile "foo" [O_WRONLY; O_TRUNC; O_CREAT] 0o666 in

dup2 fd stdout;

close fd;

execvp "bar" [|"bar"|]

dup2 を呼ぶとディスクリプタ stdout はファイル foo を指すようになります。つま

stdout への書き込みはファイル foo への書き込みとなります (fd への書き込みも同
様ですが、これは使用しないのですぐに閉じます) 。stdout の設定は execvp で保存

されるので、プログラム bar は標準出力がファイル foo に繋がった状態で実行されま

す。これはシェルで bar > foo としたときと同じ動作です。

* * *
例 あるプログラムの出力を他のプログラムの入力にする

let (fd_in, fd_out) = pipe () in

match fork () with

| 0 ->

dup2 fd_in stdin;

close fd_out;

close fd_in;

execvp "cmd2" [|"cmd2"|]

| _ ->

dup2 fd_out stdout;

close fd_out;

close fd_in;

execvp "cmd1" [|"cmd1"|]

プログラム cmd2 は標準入力がパイプの出力となった状態で実行されます。それと並

行してプログラム cmd1 は標準出力がパイプの入力となった状態で実行されます。結

果として cmd1 が標準出力に書いたものはすべて cmd2 が標準入力から読みます。

cmd1 が cmd2 よりも前に終了すると何が起こるでしょうか? cmd1 が終了すると全

ての開かれているディスクリプタが閉じられるので、パイプの入力を指すディスクリ

プタがなくなります。結果として cmd2 がパイプにあるデータを全て読んだ次の読み

込みで EOF を読み込みます。cmd2 は標準入力が末尾に達したときの動作を行いま

す — 例えば終了するなどです。

反対に cmd2 が cmd1 よりも前に終了した場合、パイプの出力を指す最後のディス

クリプタが閉じられるので cmd1 は次に標準出力に書き込んだときに sigpipe シグナ

ルを受け取ります (このシグナルのデフォルトの動作はプロセスの終了です)。

* * *
練習問題 14 シェルが持つ他のリダイレクトを実装してください。具体的には以下

です:

>> 2> 2>> 2>1 <<

解答は 123 ページ。

* * *

二つのディスクリプタを交換する場合には注意が必要です。dup2 fd1 fd2; dup2 fd2 fd1

では上手く行きません。一つ目の dup2 によって fd1 と fd2 の両方のディスクリプタ

がファイルテーブルエントリ内の同じファイルを指すようになるので、最初に fd2 に

指されていた値は失われ、二つ目の dup2 は意味を持ちません。ここで行いたいのは 2
つの参照セルの交換なので、片方の値を保存しておく一時的な変数が必要になります。

システムコール dup によってディスクリプタをコピーして保存することができます。

val dup : file_descr -> file_descr

dup fd は fd が指すファイルテーブルエントリと同じファイルを指す新しいディスク

リプタを返します。これを使うことで、例えば stdout と stderr の交換は次のように

行なえます:

let tmp = dup stdout in

dup2 stderr stdout;

dup2 tmp stderr;

close tmp;;

ディスクリプタのリークを防ぐために交換の後で tmp をクローズします。

5.5 完全な例: N 個のコマンドの合成

以下のようなコマンドを作成します:

compose cmd1 cmd2 ... cmdn

このコマンドは次のシェルコマンドのように動作します:

cmd1 | cmd2 | ... | cmdn

1 open Sys;;

2 open Unix;;

3

4 let compose () =

5 let n = Array.length Sys.argv - 1 in

6 for i = 1 to n - 1 do

7 let (fd_in, fd_out) = pipe () in

8 match fork () with

9 | 0 ->

完全な例: N 個のコマンドの合成

10 dup2 fd_out stdout;

11 close fd_out;

12 close fd_in;

13 execv "/bin/sh" [| "/bin/sh"; "-c"; Sys.argv.(i) |]

14 | _ ->

15 dup2 fd_in stdin;

16 close fd_out;

17 close fd_in

18 done;

19 match fork () with

20 | 0 -> execv "/bin/sh" [|"/bin/sh"; "-c"; Sys.argv.(n) |]

21 | _ ->

22 let rec wait_for_children retcode =

23 try

24 match wait () with

25 | (pid, WEXITED n) -> wait_for_children (retcode lor n)

26 | (pid, _) -> wait_for_children 127

27 with

28 Unix_error(ECHILD, _, _) -> retcode in

29 exit (wait_for_children 0)

30 ;;

31 handle_unix_error compose ();;

処理の大部分は 6 行目から始まる for ループが行います。最後のコマンドを除いた各

コマンドについて、新しいパイプと子プロセスを作成します。子プロセスはパイプの

入力を標準出力に接続してからコマンドを実行します。fork のあと子プロセスは親プ

ロセスの標準入力を引き継ぎます。メインプロセス (親プロセス) はパイプの出力を
標準入力としてループを続けます。i 番目の反復において以下のような状況になると
(帰納法の仮定として) 仮定します:

stdin cmd1 cmd2 . . .

. . . cmdi−1 compose stdout

丸角の四角形がプロセスを表します。プロセスの標準入力を左に、標準出力を右に示

しています。楕円は compose プロセスの最初の標準入出力を表します。この状態から

pipe と fork を実行すると以下のような状況となります:

. . . cmdi−1
compose

(子プロセス)

compose

(親プロセス)
stdout

親プロセスが dup2 を実行すると、こうなります:

. . . cmdi−1
compose

(子プロセス)

compose

(親プロセス)
stdout

子プロセスが dup2 と execv を実行すると、こうなります:

. . . cmdi−1 cmdi compose stdout

これで次の反復への準備が整います。

最後のコマンドではパイプを作る必要がないのでループの外でフォークが実行さ

れます。compose プロセスの標準入力 (最後から二番目のコマンドの出力) と標準出力
(最初 compose コマンドに与えられたもの) が最初から正しいので、ただ fork と exec

呼ぶだけで十分です。親プロセスはそれから子プロセスの終了を待つために wait を

ECHILD エラー (終了待ちの子がいない) が出るまで繰り返し呼びます。子プロセスの
リターンコードはビットごとの “or” (lor 演算子) によってまとめられ、compose の

返り値となります。これによって全ての子プロセスが 0 を返した場合には 0 が、そう
でない場合には 0 以外が compose コマンドから返ります。

/bin/sh を使ってコマンドを実行していることに注意してください。次のようなコ

マンドを単語に分割する処理を /bin/sh に任せています:

compose "grep foo" "wc -l"

ミニシェルの例のようにこの処理を自分で書くこともできますが、コードを不必要に

複雑にしてしまうだけでしょう。

5.6 入出力の多重化

これまでの全ての例において、プロセス間の通信は 線形 でした。つまりそれぞれの

プロセスが読み込むのは多くとも一つのプロセスからのデータでした。この節では 複

数の プロセスからのデータを読み込む問題の解決方法を見ていきます。

複数のウィンドウを持つ端末エミュレータを例として考えます。あるコンピュー

タ (クライアントと呼びます) がシリアルポートで Unix マシンに接続しており、Unix
マシン上の異なるプロセスに接続するためにクライアント上で複数のウィンドウをエ

ミュレートする必要があるとします。例えばウィンドウの一つはシェルに、他のウィン

ドウはテキストエディタに使いたいという状況です。シェルからの出力は最初のウィ

ンドウに、テキストエディタからの出力は二番目のウィンドウに表示します。最初の

ウィンドウがアクティブならばクライアントのキーボードからの入力はシェルに送ら

れ、二番目のウィンドウがアクティブな場合はエディタに送られます。

Unix マシンとクライアントの間には物理的な接続が一つしか無いことから、デー
タの送受信を分割してウィンドウとプロセスの間に仮想的な接続を多重化する必要が

あります。シリアルポートには次のプロトコルを使用してメッセージを送ります:

• 受信者のプロセス番号またはウィンドウ番号を表す 1 バイト

• これから続くバイト数 N を表す 1 バイト

• 受信者に送られる N バイトのデータ

Unix マシンではユーザプロセス (シェルやエディタなど) はパイプによって一つ以上
の補助プロセスに接続され、補助プロセスはデータの多重化 (あるいは逆多重化) やシ
リアルポートへの入出力を行います。シリアルポート (例えば /dev/ttya) はスペシャ
ルファイルであり、補助プロセスがクライアントとやり取りを行うために使われます。

逆多重化 (クライアントからユーザプロセスへのデータ転送)は難しくありません。
シリアルポートからデータを読み、抽出したデータを目的のユーザプロセスの標準入

力に接続されたパイプに書き込めばよいです。

/dev/ttya 逆多重化

shell

emacs

多重化 (ユーザプロセスからクライアントへのデータ転送) はもっと複雑になります。
まず逆多重化と同じようなことを試してみましょう。ユーザプロセスの標準出力に接

続されたパイプからの出力を読み込み、受信者のウィンドウ番号とデータの長さをつ

けてシリアルポートに送るプロセスです。

/dev/ttya 逆多重化

shell

emacs

多重化 /dev/ttya

パイプからの読み込みがブロックする可能性があるので、これは上手く行きません。

例えばシェルからの出力を読み込みを行ない、その時点では表示するものがなかった

場合、多重化プロセスはブロックし、エディターからの文字列は無視されます。表示

されるべきデータがあるのはどちらのプロセスのパイプなのかを事前に知る方法はあ

りません (並列アルゴリズムにおいては、あるプロセスが共有リソースへのアクセス
をずっと拒否される状況を 飢餓 と言います)。
次に別のアプローチを示します。各ユーザプロセスに リピータ プロセスを結びつ

けます。このリピータはユーザプロセスの標準出力に接続したパイプの出力を読み、

データをメッセージに変換し、メッセージをシリアルポートに直接書き込みます。各

リピータプロセスは /dev/ttya を書き込みモードで開きます。

/dev/ttya 逆多重化

shell

emacs

リピータ

リピータ

/dev/ttya

各ユーザプロセスの出力は独立して転送されることから、ブロッキングの問題は

解決されます。しかしこの方法を使うとプロトコルが破られる場合があります。二つ

のリピータが同時にメッセージを書き込んだ場合、Unix カーネルはそれらの書き込み
の原始性、すなわち書き込みが割り込まれることなく行われなければいけないことを

考慮することができません。そのため書き込みは最初にメッセージの一部を送り、そ

の次に別のメッセージを送ってから、最初のメッセージの残りの部分を送るかもしれ

ません。クライアントの多重化プロセスはこれに対処できず、二番目のメッセージを

最初のメッセージの後半部分と解釈し、それ以降のデータを次のデータのヘッダと解

釈するでしょう。

これを避けるためには、シリアルポートに書き込むプロセスがどんなときでも多

くとも一つになるようにリピータプロセス同士が同期する必要があります (並列アル
ゴリズムではこのことを「リピータのシリアル接続へのアクセスは排他制御される必

要がある」と言います)。技術的には、これまでに見た概念でこれを行うことができま
す。リピータはメッセージを送る前に特定のファイル (“ロック”) を O_EXCL フラグで

作成し、シリアルポートへの書き込みが終わった後にそのファイルを削除するように

すればよいです。しかしロックの作成と削除にコストが掛かり過ぎるので、あまり効

率的とは言えません。

より良い解決法は最初のアプローチ (一つの多重化プロセス) をとり、ユーザプロ
セスの標準出力に接続されたパイプからの出力を set_nonblock によってノンブロッ

キングに設定することです。空のパイプからの読み込みはブロックせずすぐに EAGAIN

または EWOULDBLOCK エラーを出して返るので、このエラーは無視して次のプロセスの

出力を読みにいくことができます。こうすることで飢餓と排他制御の問題を回避でき

ます。しかしこれはとても非効率的な方法でもあります。多重化プロセスがデータを

送っているプロセスがなくても実行を止めず、“ビジーウェイト” と呼ばれることを行
うためです。この非効率性は読み込みループに sleep を入れることで緩和できますが、

残念ながら sleep する正しい時間を見つけるのはとても難しいです。sleep が短いと

データが少ないときにプロセッサに無駄な負荷がかかり、sleep が短いとデータが多

いときに知覚可能なレベルの遅延を引き起こすためです。

これは深刻な問題であり、bsd の設計者たちはこの問題を解決するために select

という新しいシステムコールを Unix に追加しました。このシステムコールはほとん
どの Unix で利用可能です。select を呼ぶと一つ以上の入出力のイベントを (受動的
に) 待つことができます。ここでイベントとは以下のことです:

• 読み込みイベント: ディスクリプタからデータを読み込む準備が整った。

• 書き込みイベント: ディスクリプタにブロックせずにデータを書き込む準備が整った。

• 例外的なイベント: ディスクリプタについて例外的な条件が真になった。例えばネッ
トワーク接続において優先度の高いデータ (アウトオブバンドデータ) が送信待ちの
通常データを追い抜いて送信された。この種の優先度の高いデータは例外的な条件と

なる。

システムコール select は次のシグネチャを持ちます:

val select :

file_descr list -> file_descr list -> file_descr list ->

float -> file_descr list * file_descr list * file_descr list

最初の三つの引数がディスクリプタの集合をリストで表します。最初の引数が読み込

みイベントを監視するディスクリプタの集合で、二つ目が書き込みイベントを監視す

るディスクリプタの集合、三つ目が例外的なイベントを監視するディスクリプタの集

合です。第四引数はタイムアウトまでの秒数です。第四引数が 0 以上のとき、select

はイベントが何も起こらなかったとしてもその時間が経てば返ります。この値が負の

とき、select は監視を要求されたイベントの一つが起こるまでブロックします。

select はディスクリプタのリストの三つ組を返します。最初の要素が読み込みの

準備ができたディスクリプタを、二つ目の要素が書き込みの準備ができたディスクリ

プタを、三つ目の要素が例外的な条件が真となったディスクリプタをそれぞれ表しま

す。イベントが起こる前にタイムアウトした場合、これら三つのリストは全て空です。

例 以下のコードはディスクリプタ fd1 と fd2 の読み込みイベントを監視し、0.5 秒
後に返ります。

match select [fd1; fd2] [] [] 0.5 with

| [], [], [] -> (* 0.5 秒でタイムアウトした *)

| fdl, [], [] ->

if List.mem fd1 fdl then

(* fd1 から読む *)

if List.mem fd2 fdl then

(* fd2 から読む *)

* * *
例 以下の multiplex 関数は説の最初で説明したマルチウィンドウに対応したターミ

ナルエミュレータの多重化/逆多重化の中心となります。
単純のために、多重化関数はメッセージの受信者番号を送信者番号と同じように

ラベル付けし、逆多重化関数は受け取ったデータを受信者番号に直接リダイレクトす

ることにします。つまり、ユーザプロセスとクライアントのどちらについても、送信

者は必ず同じ番号の受信者とやり取りをすると仮定します。あるいは同じことですが、

送信者と受信者の対応付けはシリアル接続の内部で受信者番号を書き換えることで行

われると仮定します。

multiplex 関数はシリアルポートへ開いたディスクリプタと同じサイズのディス

クリプタの配列二つを受け取ります。配列の一つにはユーザプロセスの標準入力につ

ながったパイプが、もう一方にはユーザプロセスの標準出力に繋がったパイプが含ま

れています。

open Unix;;

let rec really_read fd buff start length =

if length <= 0 then () else

match read fd buff start length with

| 0 -> raise End_of_file

| n -> really_read fd buff (start+n) (length-n);;

let buffer = String.create 258;;

let multiplex channel inputs outputs =

let input_fds = channel :: Array.to_list inputs in

try

while true do

let (ready_fds, _, _) = select input_fds [] [] (-1.0) in

for i = 0 to Array.length inputs - 1 do

if List.mem inputs.(i) ready_fds then begin

let n = read inputs.(i) buffer 2 255 in

buffer.[0] <- char_of_int i;

buffer.[1] <- char_of_int n;

ignore (write channel buffer 0 (n+2));

()

end

done;

if List.mem channel ready_fds then begin

really_read channel buffer 0 2;

let i = int_of_char(buffer.[0])

and n = int_of_char(buffer.[1]) in

if n = 0 then close outputs.(i) else

begin

really_read channel buffer 0 n;

ignore (write outputs.(i) buffer 0 n);

()

end

end

done

with End_of_file -> () ;;

multiplex 関数は読み込みに使うディスクリプタの集合 input_fds を作るところ

から始まります。この集合にはユーザプロセスの標準出力につながっているディスク

リプタとシリアルポートのディスクリプタが含まれます。一番外側の while の反復で

は select を呼んで input_fds 内に書き込み待ちがないか確認します。ここでは例外

的なイベントを待つことはせず、待ち時間に制限を設けることもありません。select

が返ったならば、ユーザプロセスからの入力ディスクリプタまたはシリアルポートの

ディスクリプタが入力を待っているかを調べます。

ユーザプロセスからの入力ディスクリプタの準備ができているならば read でデー

タをバッファに読み、メッセージのヘッダをつけてシリアルポートに書き込みます。

read が 0 を返した場合対応するユーザプロセスのパイプが閉じられたことを意味し
ます。クライアントのターミナルエミュレータは 0 バイトのメッセージによってウィ
ンドウに対応する番号のユーザプロセスが終了したことを通知されるので、そのウィ

ンドウを閉じることができます。

シリアルポートにデータがある場合、まず 2 バイトのヘッダから受信者番号 i と

読み込むデータのバイト数 n を読み込みます。その後シリアルポートのチャンネルか

ら n バイトを読み、読み込んだデータを i 番目のユーザプロセスの標準入力に書き込

みます。このとき n が 0 ならば、i 番目の出力ディスクリプタを閉じます。接続先の
ターミナルエミュレータが発する n = 0 というメッセージはそのエミュレータのプロ

セスの標準入力が終端に達したことを意味するからです。

really_read が End_of_file 例外を出したら一番外側のループを抜けます。この

例外はシリアルポートが EOF に達したことを意味します。

* * *

5.7 その他: write

Unix モジュールの write 関数は要求されたデータがバイト数まで全て書き込まれ終

わるまでシステムコール write を呼び続けます。

val write : file_descr -> byte -> int -> int -> int

ディスクリプタがパイプ (あるいはソケット、6 参照) だった場合、書き込みはブロッ
クしシステムコール write はシグナルによって中断する可能性があります。このよう

な場合には OCamlの Unix.write は中断され EINTR エラーが出ます。ここで問題と

なるのはシステムコール write によってデータの一部が書き込まれたにも関わらずそ

のサイズがわからないまま失われてしまうということです。このためシグナルがある

状況では Unix モジュールの write は役に立たなくなってしまいます。

この問題を解決するために、Unix モジュールには single_write という名前の “生
の” システムコールがあります。
val single_write : file_descr -> byte -> int -> int -> int

single_write を使えば、エラーが起こったときにデータが書き込まれていないことを

保証できます。

この説の残りではこの関数をどうやって実装するかを示します。基本的にこれは

OCamlと C の間のインターフェースの問題です (OCamlのマニュアルにはこの話題
に関する情報が載っています) 。次のコードは single_write.c に保存されます:

1 #include <errno.h>

2 #include <string.h>

3 #include <caml/mlvalues.h>

4 #include <caml/memory.h>

5 #include <caml/signals.h>

6 #include <caml/unixsupport.h>

7

8 CAMLprim value caml_single_write

9 (value fd, value buf, value vofs, value vlen) {

10 CAMLparam4(fd, buf, vofs, vlen);

11 long ofs, len;

12 int numbytes, ret;

13 char iobuf[UNIX_BUFFER_SIZE];

14 ofs = Long_val(vofs)

15 len = Long_val(vlen)

16 ret = 0;

17 if (len > 0) {

18 numbytes = len > UNIX_BUFFER_SIZE ? UNIX_BUFFER_SIZE : len;

19 memmove (iobuf, &Byte(buf, ofs), numbytes);

20 caml_enter_blocking_section ();

21 ret = write(Int_val(fd), iobuf, numbytes);

22 caml_leave_blocking_section ();

23 if (ret == -1) uerror("single_write", Nothing);

24 }

25 CAMLreturn (Val_int(ret));

26 }

最初の二行は標準の C ヘッダをインクルードします。その次の四行はディストリ
ビューションと共にインストールされる OCaml特有の C ヘッダをインクルードしま
す。unixsupport.h ヘッダは OCamlの Unix ライブラリが持つ再利用可能な C の関
数を定義しています。

一番重要な行は write を呼ぶ行です。この関数は (ディスクリプタがパイプかソ
ケットの場合に) ブロックすることがあるので、呼び出す前に OCamlランタイムの
グローバルのロックを開放し (20 行目) 、呼び出した後にロックを取得します (22 行
目)。これによってこの関数は Thread モジュール (7 章参照) と同時に利用できます。
ブロックしている間に他のスレッドが実行できるからです。

システムコールの間に OCamlはガベージコレクションが実行され、OCamlの文
字列 buf がメモリ内を移動する可能性があります。この問題を解決するために buf を

C の文字列 iobuf にコピーします。これには追加のコストがかかりますが、10% 前後
のオーダーです (50% 程度と考えるかもしれませんが、そうではありません)。関数全
体のコストはシステムコールがほとんどを占めるためです。バッファに使う C の文字
列の大きさは unix_support.h に定義されています。システムコール中にエラーが起

こった場合 (負の返り値で示されます) そのエラーは uerror 関数によって OCamlに
運ばれます。uerror 関数は OCamlの Unix ライブラリで定義されています。
この関数を OCamlから利用するには、unix.mli を作って以下の宣言を書きます:

external unsafe_single_write :

file_descr -> string -> int -> int -> int = "caml_single_write"

実際に使うときには関数を呼ぶ前に引数を確認します:

let single_write fd buf ofs len =

if ofs < 0 || len < 0 || ofs > String.length buf - len

then invalid_arg "Unix.write"

else unsafe_single_write fd buf ofs len

single_write関数はバージョン 3.08から Unixモジュールで利用可能ですが、上記の

ように自分で書いた場合には以下のようにしてコンパイルする必要があります (OCaml
のコードは write.mli と write.ml にあると仮定します)。

ocamlc -c single_write.c write.ml

ocamlc -custom -o prog unix.cma single_write.o write.cmo mod1.ml mod2.ml

通常は以下のようにして C と OCamlのコードを両方含むライブラリ write.cma を

使ったほうが実用的です。

ocamlc -custom -a -o write.cma single_write.o write.cmo

ライブラリ write.cma は unix.cma のように使えます。

ocamlc -o main.byte unix.cma write.cma main.ml

single_write の動作はシステムコール write によく似ていますが、書き込む文字

列がとても長い場合の動作だけ異なります。文字列が UNIX_BUFFER_SIZE よりも長い

場合一度のシステムコールで全てのデータを書き込むことができないので、システム

コールは繰り返されます。そのためシステムコール write が持つ通常ファイルに対す

る書き込みの原子性は single_write では長い文字列に対しては成り立ちません。こ

の違いは通常重要ではありませんが意識しておくべきでしょう。

この関数を元にして更に高レベルの関数 really_write を実装することができま

す。要求されたデータを (アトミックでなく) 全て書き込むので、multiplexer 関数で

使った readlly_read 関数に似ています。

let rec really_write fd buffer offset len =

let n = restart_on_EINTR (single_write fd buffer offset) len in

if n < len then really_write fd buffer (offset + n) (len - n);;

6 モダンなプロセス間通信: ソケット

パイプを使った通信にはいくつか制限があります。まず、パイプを使うとローカルの

マシンとしか通信できません。名前付きパイプではやり取りをするプロセスが同じマ

シンで実行される必要があり、名前無しパイプでは二つのプロセスは親子関係になく

てはいけません。次に、パイプでは クライアント-サーバ モデルという便利な通信モ
デルを使うことが難しいです。このモデルではちょうど一つの サーバ プログラムだ

けが共有リソースにアクセスできます。そして他の クライアント はサーバにアクセ

スすることで共有リソースにアクセスします。サーバが共有リソースへのアクセスの

制御と管理を行うということです。例えば x-window システムでは共有リソースとは
スクリーンとキーボード、そしてマウスです。

クライアント-サーバモデルをパイプで実装するのは難しいです。一番の困難はク
ライアントとサーバの間の接続を確立する部分です。名前無しパイプではこれは不可

能です。サーバとクライアントに共通の親が必要であり、その親は任意の数のパイプ

を前もって保持する必要があるからです。名前付きパイプを使えばサーバがクライア

ントからの接続を読むことは不可能ではありません。しかしこの場合サーバが受け取

るリクエストにはクライアントがサーバとやり取りするのに使う他の名前付きパイプ

の名前も含まれるので、複数のクライアントからの接続リクエストをどう排他制御す

るかが問題となります。

ソケットはこれらの問題を解決するために一般化されたパイプです。ソケットの

説明のために、まずクライアント-サーバモデルを図 3 に示します。

1. サーバ U がソケット s をポート p に作成し、接続を待つ (1)。クライアントはポー
ト番号 p を知っている。

2. クライアント A がソケットを作成し、ポート p に接続する (2)。サーバ上では、シ
ステムがクライアント A とプライベートにやり取りするための新しいソケットを確
保する (3)。この例ではサーバは補助サーバ V をフォークし、A との接続を閉じる
(4)。閉じられた接続は破線で示されている。以降はこの子プロセス V が A との通信
を行う (5)。

サーバ V

サーバ U
(1)

サーバ W

クライアント A

クライアント B

(2)

(3)

(5)

(6)

(4)

Figure 3 — クライアント-サーバモデル

3. サーバは新しいクライアント B を受け付けることができる。新しい接続が作成され、
クローンしたプロセス W によって通信が行われる (6)。以降も同様となる。

4. サーバはソケット s に結び付けられたファイルディスクリプタを閉じることでサービ
スを終了できる。システムはポート p を開放して再利用する。このポートは例えば他
のサービスをインストールするのに使われる。

上記のモデルでは、サーバ U とクライアント A はプライベートな接続 (3) を確
立することで他のクライアントの影響を受けずにやり取りを行います。このことから、

この通信方式は コネクション型通信 と呼ばれます。通信が短ければサーバが (フォー
クせずに) 直接リクエストを処理することもできます。しかしこうするとサーバが接
続 (3) を処理するのを待つため、あるいは複数の接続が多重化によって管理されてい
るために、クライアントはサーバが利用可能になるまで待たなければいけません。

ソケットを使うと コネクションレス型通信 を利用することもできます。この方式

ではサーバはクライアントとプライベートな接続を確立することなく直接通信します。

コネクション型通信と比べるとこの方式が使われることは少ないです。6.10 節で少し
コメントしますが、この章では主にコネクション型通信を扱います。

6.1 ソケット

ソケットはパイプの拡張であり、bsd 4.2 で初めて実装されました。現在ではネット
ワークに接続するすべての Unix がソケットの機能を持ちます。クライアント-サーバ
モデルに沿った通信を行うための専用のシステムコールが存在します。このシステム

コールによって、プロセス間のローカル通信およびリモート通信を (ほぼ) 透過的に
行うことができます。

ソケットの 通信ドメイン は通信できるプロセス (およびそのアドレスのフォー
マット) を制限します。異なる通信ドメインが利用可能です。例えば:

• Unix ドメイン: アドレスはマシンのファイルシステム上の名前。通信は同じマシンで
実行されるプロセス間に限られる (パイプを使った通信など)。

• インターネットドメイン: アドレスはインターネットネットワークにおけるマシンのア
ドレス (例えば 129.199.129.1 のような形をしたもの) とマシンのポート番号。通信は
インターネットに接続している任意の二つのマシンで実行されるプロセス間で可能1。

ソケットの 通信方式 は通信が信頼できるか (データの消失や重複があるか)、そし
てデータの送受信の方法 (バイトストリーム、パケット列 — バイトの小さなブロッ

ク) を示します。通信方式はデータの送受信に使われるプロトコルを規定します。い
くつかの通信方式が利用可能です。三つの通信方式をその特徴と共に示します:

方式 信頼性 データ表現

ストリーム 信頼できる バイトストリーム

データグラム 信頼できない パケット

セグメント化されたパケット 信頼できる パケット

“ストリーム”方式はパイプを使った通信にとても良く似ており、一番良く使われます。
例えば構造化されていないバイトシーケンスの転送 (例えば rsh) などに使われます。

1インターネットは専用のリンクで接続されたローカルネットワーク (通常はイーサネットネットワー
ク) で構成され、世界中の何百万台のマシンが接続しています。インターネットドメインにおいては、同
じイーサネットケーブルで接続されている隣のマシンとの通信と、いくつものルータと通信衛星を使っ

て接続された地球の反対側にあるマシンとの通信にはなんの違いもありません。

“セグメント化されたパケット” 方式はデータをパケットとして転送します。すべての
書き込みはパケットごとに区切られ、すべての読み込みは多くとも一つのパケットし

か読み込めません。この方式はメッセージ指向の通信に適しています。“データグラム”
方式はイーサネットネットワークのハードウェアの特徴に一番近いです。データはパ

ケットで転送されますが、それぞれのパッケットが目的地に届く保証はありません。

ネットワークのリソースという観点では、この方式が一番無駄が少ないです。この方

式は致命的な重要性を持たないデータの転送を行うプログラム (例えば biff) によっ
て使われます。またデータの消失を手動で管理することでネットワークのパフォーマ

ンスを向上させることもできます。

6.2 ソケットの作成

システムコール socket は新しいソケットを作成します:

val socket : socket_domain -> socket_type -> int -> file_descr

返り値は新しいソケットを表すファイルディスクリプタです。初期状態ではこのディ

スクリプタは “切断” 状態であり、read や write を受け付ける準備が整っていません。

第一引数は socket_domain 型の値で、通信ドメインを指定します:

PF_UNIX ユニックスドメイン

PF_INET インターネットドメイン

第二引数は socket_type 型の値で、通信方式を指定します:

SOCK_STREAM 信頼できるバイトストリーム

SOCK_DGRAM 信頼できないパケット

SOCK_RAW ネットワークの下層への直接のアクセス

SOCK_SEQPACKET 信頼できるパケット

第三引数は通信で使うプロトコルです。通信ドメインに対するデフォルトのプロトコル

(例えば SOCK_DGRAM に対しては udp、SOCK_STREAM に対しては tcp など) を選択する
0が通常使われます。第三引数に他の値を設定すると特殊なプロトコルを使うことがで

きます。例えば pingコマンドは icmp (Internet Control Message Protocol)を使って、
自動的に送信者に送り返すパケットを送ります。特殊なプロトコルに対する引数の値

は /etc/protocols ファイルまたは nis (Network Information Service) データベース
の protocols テーブルが存在すればそこにあります。システムコール getprotobyname

はプロトコルに関する情報をポータブルな形で返します。

val getprotobyname : string -> protocol_entry

引数はプロトコルの名前で、返り値は protocol_entry 型のレコードです。このレコー

ドの p_proto フィールドがプロトコルの番号を表します。

6.3 アドレス

ソケットに関する操作はソケットアドレスを利用するものがあります。ソケットアド

レスはヴァリアント型 sockaddr によって表されます:

サーバへの接続

type sockaddr =

| ADDR_UNIX of string

| ADDR_INET of inet_addr * int

ADDR_UNIX f は Unix ドメインにおけるアドレスです。f は対応するファイルのマシ

ンのファイルシステムにおける名前です。ADDR_INET (a,p) はインターネットドメイ

ンにおけるアドレスです。a がマシンのインターネットアドレスで、p がそのマシン

のポート番号です。

インターネットアドレスは抽象型 inet_addrを表します。次の二つの関数は 128.93.8.2

のような形をした文字列と inet_addr 型の値の間の変換を行います。

val inet_addr_of_string : string -> inet_addr

val string_of_inet_addr : inet_addr -> string

インターネットアドレスを調べるもう一つの方法はホストの名前を使って /etc/hosts

にあるテーブル、nis データベース、あるいはドメインネームサーバから引くことで
す。システムコール gethostbyname がこれを行います。現代的なマシンではドメイン

ネームサーバへの問い合わせが最初に行われ、/etc/hosts はフォールバックとしてし

か使われませんが、一般的にはこの順番はマシンの設定によります。

val gethostbyname : string -> host_entry

引数はホストの名前で、返り値は host_entry 型のレコードです。このレコードの

h_addr_list フィールドが目的のマシンのインターネットアドレスを表します (同じ
マシンが複数のネットワークに違うアドレスで接続することは可能です)。
最もよく使われるサービスが使うポート番号は /etc/services のテーブルにリス

トされており、getservbyname 関数でポ− タブルに取得できます。

val getservbyname : string -> string -> service_entry

第一引数はサービスの名前 (ftp サーバには "ftp"、E メールには "smtp"、ニュース

サーバには "nntp"、talkと ntalkにはコマンドと同じ文字列、など)です。第二引数は
プロトコルの名前です。プロトコルの名前にはサービスの通信方式がストリーム方式の

場合は "tcp" が、データグラム方式の場合は"udp" が通常使われます。getservbyname

の返り値は service_entry 型の値で、s_port フィールドがサービスが用いるポート

番号を表します。

例 pauillac.inria.frの ftpサーバのアドレスを入手するには以下のようにします:

ADDR_INET((gethostbyname "pauillac.inria.fr").h_addr_list.(0),

(getservbyname "ftp" "tcp").s_port)

* * *

6.4 サーバへの接続

システムコール connect はソケットに関連づいたサーバとの通信を確立します。

val connect : file_descr -> sockaddr -> unit

第一引数はソケットのディスクリプタで、第二引数はサーバのアドレスです。

接続が確立されると、ソケットのディスクリプタに対する write はサーバにデー

タを送信し、read はサーバからのデータを受信します。ソケットは入出力の操作に対

してパイプのように振る舞います。まず、read はデータが利用可能でない場合にはブ

ロックし、要求よりも少ないバイト数を返すことがあります。次に、サーバが接続を

閉じると read は 0 を返し、write は読んだプロセスに sigpipe シグナルを送ります。

connect 関数はソケットをシステムによって選ばれたローカルアドレスにバイン

ドします。このアドレスを手動で選んだほうが良い場合もあります。そのような場合

は connect 関数を呼ぶ前に bind 関数 (6.7 節を参照) を呼ぶことでアドレスを選ぶこ
とができます。

Unix コマンド netstat は現在のマシン上の接続と状況をリストします。

6.5 ソケットの切断

ソケットを切断する方法は二つあります。最初の方法はソケット close を呼んで入出

力用の接続が閉じることです。しかしこれでは大雑把すぎることがあります。例えば

ファイルの終端を伝えるためにクライアントからサーバの接続を閉じつつもサーバか

らのデータを受け取るために反対側の接続は保ちたい場合があります。システムコー

ル shutdown を使うと接続の一部を切断できます。

val shutdown : file_descr -> shutdown_command -> unit

第一引数は閉じるソケットのディスクリプタで、第二引数は shutdown_command 型の

値で、どちらの方向の節毒を閉じるのかを指定します。

SHUTDOWN_RECEIVE 読み込み用のソケットを閉じる。接続のもう一端における

write は sigpipe シグナルを呼び出しプロセスに送るよう

になる。

SHUTDOWN_SEND 書き込み用のソケットを閉じる。接続のもう一端における

read は EOF を返すようになる。

SHUTDOWN_ALL 読み込みと書き込み用のソケットを両方閉じる。close と違

い、閉じられたソケットのディスクリプタは開放されない。

ソケットの切断には close と shutdown のどちらを使っても多少の時間がかかること

があることに注意してください。

6.6 完全な例: ユニバーサルクライアント

client host port が host マシンの port ポートへの接続を確立するような clientコ

マンドを作成します。さらにこのコマンドは標準入力からのデータを作成したソケッ

トから送信し、受信したデータを標準出力へ書き込みます。例えば、以下のコマンド

echo -e 'GET /~remy/ HTTP/1.0\r\n\r\n' | ./client pauillac.inria.fr 80

は pauillac.inria.frの 80版ポートに接続し、ウェブページ /~remy/に対する http
リクエストを送ります。

client が行うのはクライアントとの接続を確立する部分だけであり、特定のプロ

トコルの実装は client を呼んだプログラムに任されます。この意味で、client コマ

ンドは “ユニバーサルな” クライアントアプリケーションです。
ライブラリ関数 Misc.retransmit fdin fdout はディスクリプタ fdin から読んだ

データを fdout に書き込みます。入力ディスクリプタの終端に達するとディスクリプ

タを閉じずに関数は終了します。retransmit はシグナルで中断されることがあること

に注意してください。

完全な例: ユニバーサルクライアント

let retransmit fdin fdout =

let buffer_size = 4096 in

let buffer = String.create buffer_size in

let rec copy () = match read fdin buffer 0 buffer_size with

| 0 -> ()

| n -> ignore (write fdout buffer 0 n); copy ()

in

copy ();;

本当の問題はここからです。

1 open Sys;;

2 open Unix;;

3

4 let client () =

5 if Array.length Sys.argv < 3 then begin

6 prerr_endline "Usage: client <host> <port>";

7 exit 2;

8 end;

9 let server_name = Sys.argv.(1)

10 and port_number = int_of_string Sys.argv.(2) in

11 let server_addr =

12 try (gethostbyname server_name).h_addr_list.(0)

13 with Not_found ->

14 prerr_endline (server_name ^ ": Host not found");

15 exit 2 in

16 let sock = socket PF_INET SOCK_STREAM 0 in

17 connect sock (ADDR_INET(server_addr, port_number));

18 match fork () with

19 | 0 ->

20 Misc.retransmit stdin sock;

21 shutdown sock SHUTDOWN_SEND;

22 exit 0

23 | _ ->

24 Misc.retransmit sock stdout;

25 close stdout;

26 wait ();;

27

28 handle_unix_error client ();;

接続するインターネットアドレスを決めることから処理が始まります。アドレスは

ホストネームまたは数字の形を使うことができます。gethostbyname がどちらの場合

にも対応しているためです。それからインターネットドメインでストリームタイプの

ソケットをデフォルトプロトコルで作成し、決定したアドレスのマシンに接続します。

プロセスは fork を使ってクローンします。子プロセスは標準入力からのデータを

ソケットにコピーします。標準入力の終端に達すると、子プロセスは送信方向の接続

を閉じて終了します。親プロセスはソケットから呼んだデータを標準出力にコピーし

ます。ファイルの終端に達すると標準出力を閉じ、子プロセスの終了を待ってから終

了します。

このプログラムにおいて、接続が切れる原因は以下の三つです:

• 子プロセスが標準入力から EOF を受け取った場合。子プロセスはクライアントから
サーバ方向の接続を閉じる。これによってサーバはソケットから EOF を受け取るの
で、おそらくはいくらかの処理をした後に反対方向の接続を閉じる。そして親プロセ

スはソケットから EOF を受け取り通常終了する。

• サーバが接続を閉じた場合。親プロセスはソケットから EOF を受け取り子プロセス
の終了を待つ。子プロセスは次にソケットに対する write を実行したときに sigpipe

シグナルを受け取って終了する。接続が切れたことは呼び出し側に報告されないが、

もし必要ならば以下のコードを 19 行目の後に入れることで sigpipe を無視すること

ができる。

ignore (signal sigpipe Signal_ignore)

これによって代わりに EPIPE エラーが出るようになる。

• 親または子プロセスが実行の途中で終了した場合、入力または出力用のソケットは閉
じられる。ソケットが閉じられたことをサーバが検知した場合、サーバはその方向の

接続を閉じる。これによってもう一方向の接続もしばらくしてから閉じられることに

なる。

6.7 サービスの確立

前の節でクライアントがサーバに接続する方法を見たので、今度はサーバーがクライ

アントにサービスを提供する方法について見ていきます。まずソケットに特定のアド

レスを関連付けネットワークから到達可能にします。システムコール bind がこれを

行います。

val bind : file_descr -> sockaddr -> unit

第一引数はソケットのディスクリプタで、第二引数はバインドするアドレスです。イ

ンターネットアドレスとして定数 inet_addr_any を使うと、マシンが持つすべてのイ

ンターネットアドレス (複数のサブネットワーク内におけるアドレス) をバインドす
ることができます。

システムコール listen を使ってソケットが接続を受け入れられる状態にします:

val listen : file_descr -> int -> unit

第一引数はソケットのディスクリプタで、第二引数はサーバーがビジー状態のときに

保持できるリクエストの数です (数十から大きなサーバーでは数百程度になります) 。
接続を待っているクライアントが第二引数で指定した値よりも大きくなった場合、そ

れ以降の接続リクエストは失敗します。

最後に、ソケットのディスクリプタに対するへの接続リクエストはシステムコー

ル accept を通じて受け入れられます。

val accept : file_descr -> file_descr * sockaddr

accept の呼び出しが返った場合、引数として与えられたソケットは影響を受けない

ので続けて接続リクエストを受け入れることができます。返り値の第一要素はクライ

アントに接続した新しいディスクリプタです。サーバがこのディスクリプタに書き込

んだデータは、クライアントが connect に渡したディスクリプタから読み込むことが

できます。またクライアントが connect に渡したディスクリプタに書き込んだデータ

は、accept の返り値の第一要素のディスクリプタから読み込むことができます。

accept の返り値の第一要素はクライアントのアドレスです。接続したクライアン

トが接続する権限を持っているかを確認したり (例えば x サーバはこの処理を行いま

す。xhost で新しい許可されたユーザを追加できます)、サーバーからクライアントへ
の二つめの接続を作成する (ftp はすべてのファイル転送リクエストに対してこの処
理を行います) ことに使うことができます。

tcp サーバの一般的な構造は以下のようになっています:

let install_tcp_server_socket addr =

let s = socket PF_INET SOCK_STREAM 0 in

try

bind s addr;

listen s 10;

s

with z -> close s; raise z;;

ライブラリ関数 Misc.install_tcp_server addr は bind と listen を使ってストリー

ムタイプでインターネットドメインのソケットをデフォルトプロトコルで作成し、ア

ドレス addr からの接続を受けられるように準備します。この関数はライブラリ関数

なので、エラーの場合にはソケットを閉じます。

let tcp_server treat_connection addr =

ignore (signal sigpipe Signal_ignore);

let server_sock = install_tcp_server_socket addr in

while true do

let client = restart_on_EINTR accept server_sock in

treat_connection server_sock client

done;;

ライブラリ関数 Misc.tcp_server は Misc.install_tcp_server でソケットを作成し

た後無限ループに入ります。ループでは accept を使って接続要求を待ち、受け入れ

た接続は treat_connection 関数を使って処理します。treat_connection はライブラ

リ関数なことから、accept が途中で終了した場合にはやり直します。また予期しない

接続の切断が起こったときにサーバプログラムを終了させるのではなく、EPIPE 例外

が treat_connection によって捕捉されるように sigpipe シグナルを無効化しておき

ます。接続を閉じるときにディスクリプタ client を閉じるのは例外の有無にかかわ

らず treat_connection の責任です。

treat_connection 関数はアドレスの他にサーバのディスクリプタも受け取ります

が、これは treat_connection 関数が fork や double_fork をした場合でもサーバを

閉じられるようにするためです。

以下のアプリケーション特有の service 関数を考えます。この service 関数はク

ライアントと通信を行ってから接続を閉じます。

let service (client_sock, client_addr) =

(* ディスクリプタが client_sock のクライアントと処理を行う *)

(* 処理が終わったならば: *)

close client_sock;;

この場合サーバは接続を順番に処理することができます。次の Misc モジュールに

あるライブラリ関数はこのパターンに対処します:

let sequential_treatment server service client = service client

sequential_treatment 関数を使った場合サーバがあるクライアントを処理している間

は他の接続リクエストを処理することができないので、この関数が使えるのは service

関数が決まった短い時間で終わるサービスのときに限られます。

ほとんどのサーバプログラムは accept が返った直後に fork を呼ぶことでサービ

スの実行を子プロセスに移譲します。子プロセスが接続を処理し、親プロセスはもう

一度 accept を行います。Misc モジュールの以下の関数がこの操作を行います。

let fork_treatment server service (client_sock, _ as client) =

let treat () = match fork () with

| 0 -> close server; service client; exit 0

| k -> ()

in

try_finalize treat () close client_sock;;

親プロセスがディスクリプタ client_socket を閉じることが重要であることに注意し

てください。ここで閉じることを忘れると子プロセスが client_socket を閉じても接

続を終了しなくなり、すぐに親プロセスのディスクリプタが枯渇します。このディスク

リプタはフォークが失敗したときにも閉じられます。エラーが致命的でなければサー

バの実行が続くためです。

同様に、子プロセスは接続リクエストを受け取った server ディスクリプタをすぐ

に閉じます。一つ目の理由はこのディスクリプタが必要がないことです。二つ目の理

由は子プロセスが終了する前にサーバが新しい接続の受け付けを終了する場合がある

ことです。ml+exit 0+ の呼び出しは子プロセスがサービスの実行を終えた後に終了
するためおよびサーバループを実行し始めないことを確実にするためにあるので重要

です。

今までのライブラリ関数では子プロセスがいずれゾンビプロセスになり回収する

必要が生じるというのを無視してきました。この問題の解決方法は二つ考えられます。

一つ目のシンプルな方法はダブルフォーク ( 41 ページ参照)を使って孫プロセスに接
続を処理させる方法です。この方法では Misc モジュールの以下のライブラリ関数を

使います:

let double_fork_treatment server service (client_descr, _ as client) =

let treat () = match fork () with

| 0 ->

if fork () <> 0 then exit 0;

close server; service client; exit 0

| k ->

ignore (restart_on_EINTR (waitpid []) k)

in

try_finalize treat () close client_descr;;

しかしこの方法でサーバプロセスが孫プロセスを管理する方法が一切無くなってしま

います。サービスを終了するときにサーバプロセスとサービスを実行している複数の

プロセス全てが終了するように、サービスを同じプロセスグループで管理するのが望

ましいです。この理由からサーバはフォークを行いながらも子プロセスを回収できる

ようにしておくことが多いです。例えば sigchld シグナルのハンドラを使うことがで

きます (50 ページの Misc.free_children 関数を参照)。

6.8 ソケットの設定

ソケットには設定できる内部変数がたくさんあります。例えば転送バッファのサイズ

や転送の最小サイズ、接続を閉じるときの動作などです。

これらの変数は異なった型を持ちます。そのため OCamlには型の数だけgetsockopt

関数と setsockopt 関数があります。getsockopt 関数とその仲間の詳細なリストは

完全な例: ユニバーサルなサーバ

OCamlのドキュメントを、getsockopt と setsockopt の厳密な意味は posix のリ
ファレンスをそれぞれ参照してください。

例 この例で紹介する二つの変数はストリームタイプでインターネットドメインのソ

ケットにしか適用することができません。

tcp プロトコルではソケットの切断にやり取りが必要なために少し時間がかかり
ます。通常は close の呼び出しはすぐに返って切断に必要なやり取りはシステムが行

います。次のコードを実行すると sock に対する close がブロックするようになりま

す。この例では全てのデータが送られるか 5 秒が経過するまでブロックします。

setsockopt_optint sock SO_LINGER (Some 5);;

SO_REUSEADDR オプションを使うとローカルアドレスに対するソケットを閉じた直

後に システムコール bind で再利用することが可能になります (前の接続のデータを
受け取ってしまう可能性はあります)。このオプションを使うとサーバを停止してすぐ
に再起動できるのでテスト用途にとても便利です。

setsockopt sock SO_REUSEADDR;;

* * *

6.9 完全な例: ユニバーサルなサーバ

次の server コマンドを作成します:

./server port cmd arg1 ... argn

このコマンドはポート port からの接続リクエストを受け取り、それぞれの接続につ

いて cmd を引数 arg1 ... argn として、標準入出力をソケットの接続として実行しま

す。例えば以下のコマンド:

./server 8500 grep foo

をマシン pomerol で実行し、別のマシンでユニバーサルクライアント (6.6 ページ参
照) を以下のように実行します:

./client pomerol 8500 < /etc/passwd

クライアントは以下のコマンドを実行したときと同じ出力を表示します:

grep foo < /etc/passwd

grep がローカルのマシンではなく pomerol で実行される点が異なります。

server コマンドは多くのサーバが行うコードをまとめていて、特定のサービスや

コミニケーションプロトコルの実装は起動されるプログラム cmd に任されています。

この意味で service コマンドは ユニバーサルな サーバです。

1 open Sys;;

2 open Unix;;

3

4 let server () =

5 if Array.length Sys.argv < 2 then begin

6 prerr_endline "Usage: client <port> <command> [arg1 ... argn]";

7 exit 2;

8 end;

9 let port = int_of_string Sys.argv.(1) in

10 let args = Array.sub Sys.argv 2 (Array.length Sys.argv - 2) in

11 let host = (gethostbyname(gethostname ())).h_addr_list.(0) in

12 let addr = ADDR_INET (host, port) in

13 let treat sock (client_sock, client_addr as client) =

14 (* ログを出力 *)

15 begin match client_addr with

16 | ADDR_INET(caller, _) ->

17 prerr_endline ("Connection from " ^ string_of_inet_addr caller);

18 | ADDR_UNIX _ ->

19 prerr_endline "Connection from the Unix domain (???)";

20 end;

21 (* 接続を処理 *)

22 let service (s, _) =

23 dup2 s stdin; dup2 s stdout; dup2 s stderr; close s;

24 execvp args.(0) args

25 in

26 Misc.double_fork_treatment sock service client in

27 Misc.tcp_server treat addr;;

28

29 handle_unix_error server ();;

tcp_server に渡されるアドレスにはプログラムを実行しているマシンのインターネッ

トアドレスが含まれます。このアドレスを取得するには普通 gethostname を使います

(11 行目)。しかし一般的には複数のアドレスが一つのマシンを指していることがあり
ます。例えばマシン pauillac のアドレスが 128.93.11.35 だったとしても、ローカル

ならば (すでに pauillac の中にいるならば)このマシンには 127.0.0.1 を使ってアク

セスすることができます。定数インターネットアドレス inet_addr_any を使えば、マ

シンを指す全てのアドレスにサービスを行うことができます。

サービスは “ダブルフォーク” で処理されます。孫プロセスによって実行される
service は標準入力と標準出力、そして標準エラー出力をソケットにリダイレクトし

てからコマンドを実行します。サービスの処理をブロックしながら順番に行うことは

できないことに注意してください。

クライアントからの接続は server の親プロセスからの介入なしに閉じられます。

以下の三つのケースが考えられます。

• クライアントがサーバ方向の接続を閉じた。サーバによって起動されたコマンドは標準
入力に EOF を受け取る。行うべきを行った後、コマンドを実行するプロセスは exit

を実行して終了する。これによって標準出力が閉じられ、このディスクリプタはクラ

イアントにつながるソケットの出力を指す最後のディスクリプタであるから、クライ

アントは次の読み込みで EOF を受け取る。

• クライアントが途中で終了し、サーバからクライアント方向の接続が閉じた。サーバ
によって起動されたコマンドは次の書き込みで sigpipe シグナルを受け取る場合があ

る。これによってプロセスが終了するが、誰もコマンドの出力を読んでいないことか

ら、これは問題ない。

• サーバによって起動されたコマンドが入力から EOF を読む前に終了した。クライア
ントは次に書き込みで sigpipe シグナル (あるいは EPIPE 例外) を受け取る。

注意点

サーバを書くのはクライアントを書くよりも労力が要ります。クライアントは接続す

るサーバのことを普通知っているのに対し、サーバはクライアントのことを何も知ら

ないからです。とくにサーバがパブリックな場合、クライアントは “敵” である可能
性があります。そのため例外的なケースに対する防御をしなければいけません。

典型的な攻撃は接続を開けた後リクエストを送ることなく開いたままにするという

ものです。接続を受け入れた後サーバはクライアントが接続している間ずっとブロッ

クします。攻撃者は何もしない接続を大量に開くことでサービスを飽和させることが

できます。そのためサーバは同時接続数を制限してリソースを枯渇を防ぐと共に、長

い間何もしていない接続を終了させる必要があります。

サーバがフォークを行わず、接続を来た順番に処理する場合、このブロッキング

の問題が生じることになります。処理している接続は何もしていないにもかかわらず

他の接続に応答することができなくなります。順番に処理するサーバに対する解決法

として通信の多重化がありますが、これは実装するのが難しくなります。並列サーバ

はよりエレガントですが、それでもタイムアウトは必要になります。タイムアウトに

は例えばアラームを使うことができます (4.2 節参照)。

6.10 コネクションレス型通信

SOCK_STREAM タイプの接続のほとんどで使われる tcp プロトコルではコネクション型
通信だけを使うことができます。逆に、SOCK_DGRAM タイプの接続のほとんどで使われ

る udp プロトコルでは常にコネクションレス型通信が使われ、二つのマシンの間に
は確立された接続がありません。このタイプのソケットではデータの転送はシステム

コール recvfrom と sendto を使って行われます。

val recvfrom :

file_descr -> string -> int -> int -> msg_flag list -> int * sockaddr

val sendto :

file_descr -> string -> int -> int -> msg_flag list -> sockaddr -> int

転送されたデータのサイズを返すという点で、インタフェースは read と write に似

ています。recvfrom は読み込んだバイト数の他にデータを送っているマシンのアドレ

スも返します。

SOCK_DGRAM タイプのソケットに connect を呼ぶと疑似接続が取得できます。この

疑似接続は本当は存在せず、connect の引数に渡されたアドレスがソケットによって

記憶され、データの送受信で使われるようになるというだけです (他のアドレスから
のデータは無視されます)。connect を複数回読んでアドレスを変更したり、0 などの
無効なアドレスを渡して疑似接続を切断することは可能です。反対にストリームタイ

プのソケットにこのようなことを行うとエラーが出ます。

6.11 低レベル入出力

システムコール recv と send はそれぞれ read と write に対応しますが、ソケットの

ディスクリプタ専用です。

val recv : file_descr -> bytes -> int -> int -> msg_flag list -> int

val send : file_descr -> bytes -> int -> int -> msg_flag list -> int

高レベルプリミティブ

recv と send のインタフェースは read と write に似ていますが、msg_flag 型のフラ

グのリストが引数に追加されています。この型の値と意味を次に示します:

MSG_OOB アウトオブバンドデータを処理する。

MSG_DONTROUTE デフォルトのルーティングテーブルを使用しない。

MSG_PEEK データを読むことなく確認だけ行う。

これらのプリミティブ (原始的な関数) はコネクション型通信通信でも read と write

の代わりに利用できるほか、疑似通信モードでも recvfrom と sendto の代わりに利用

できます。

6.12 高レベルプリミティブ

ユニバーサルクライアント/サーバの例はよく利用されるので、Unix モジュールには

ネットワークサービスを確立したり利用するための高レベルな関数があります。

val open_connection : sockaddr -> in_channel * out_channel

val shutdown_connection : Pervasives.in_channel -> unit

open_connection は引数のアドレスに対する接続を開き、そのソケットに対する入出

力用の Pervasives チャンネルの組を作成します。返り値のチャンネルに対する入出

力はサーバとの通信となりますが、チャンネルはバッファされるために、リクエスト

を本当に送られたことを保証するにはチャンネルをフラッシュする必要があります。

クライアントはチャンネルを閉じる (この操作はソケットも閉じます) ことで任意の
タイミングで接続を閉じることができるほか、shutdown_connection を使って “きれ
いに” 接続を閉じることもできます。サーバが接続を閉じた場合、クライアントは入
力チャンネルから EOF を受け取ります。

establish_server 関数でサービスを確立することができます。

val establish_server :

(in_channel -> out_channel -> unit) -> sockaddr -> unit

establish_server f addr はアドレス addr にサービスを確立し、リクエストを関数 f

で処理します。コネクションがあるたびにサーバは新しいソケットを作成してフォー

クします。子プロセスはクライアントと通信に利用する入出力用の Pervasives チャ

ンネルを作成し fに渡します。f が返ると子プロセスはソケットを閉じて終了します。

クライアントが接続をきれいに閉じたならば、子プロセスは入力チャンネルに EOF
を受け取ります。そうでなくクライアントが接続を突然閉じた場合には f が書き込み

を行おうとしたときに sigpipe を受け取ります。いずれの場合でもこのとき親プロセ

スは他のリクエストを処理しているはずです。establish_server はエラー (サービス
の確立時の OCamlランタイムやシステムのエラーなど) の場合を除いて返りません。

6.13 プロトコルの例

クライアントとサーバの間でやり取りされるデータは単純なプロトコル (rsh, rlogin,
. . . ) ではクライアントからサーバへ、およびサーバからクライアントへの二つのバイ
ト列として自然に表現できます。その他のプロトコルではやり取りされるデータはもっ

と複雑であり、バイト列からデータへ、あるいはデータからバイト列へのデコード/エ
ンコードが必要になります。クライアントとサーバはリクエストの形式と接続で交わ

されるレスポンスについて取り決めた転送プロトコルについて合意しておく必要があ

プロトコルの例

ります。Unix コマンドが利用するプロトコルの多くは “rfc” (request for comments)
と呼ばれるドキュメントに記載されています。これらのドキュメントは議論のための

提案として始まりましたが、時が経ちユーザがこのプロトコルに慣れるにしたがって

標準となりました2。

“バイナリ” プロトコル

バイナリプロトコルの多くはデータをメモリ内の表現にできるだけ近いコンパクトな

形式で転送します。エンコード/デコードに必要な処理を最小化しネットワーク帯域を
節約するためです。このタイプのプロトコルの典型的な例は x サーバと x アプリケー
ションの間の通信で利用される x-window プロトコルと nfs プロトコル (rfc 1094)
です。

バイナリプロトコルはデータを次のようにエンコードすることが多いです。整数

と浮動小数点数には同じ 1, 2, 4, あるいは 8 バイトのバイナリ表現が使われます。文
字列には長さを示す整数とその後に続く内容で表現されます。構造化されたオブジェ

クト (タプル、レコード) はフィールドを順番通りに並べて表現されます。可変長のオ
ブジェクト (配列、リスト) は長さを表す整数とそれに続く要素で表現されます。転
送されるデータの型が正確に分かっているならば、受け取ったプロセスは簡単にメモ

リ上に復元できます。そうでなく異なるタイプのデータがソケットでやり取りされた

場合、データのエンコードはタイプを表すデータ先頭の整数を読み取ってから行われ

ます。

例 x ライブラリの XFillPolygon 関数は多角形を塗りつぶす関数であり、以下の形

のメッセージを送信します:

• バイト 69 (FillPoly コマンドの番号)

• パディングバイト

• 多角形の頂点数 n を表す 16 ビット整数

• 描画するウィンドウを表す 32 ビット整数

• “グラフィクスコンテキスト” を表す 32 ビット整数

• “形” を表すバイト。多角形が凸であるか、など。

• 頂点の座標が絶対か相対かを表すバイト

• 頂点の座標を 16 ビット整数でエンコードした 4n バイト

* * *
バイナリプロトコルでは接続するマシンのアーキテクチャに注意する必要があり

ます。例えば複数のバイトからなる整数は ビッグエンディアン のマシンでは最上位

バイトを先に (メモリの若いアドレスに) 格納しますが、リトルエンディアンのマシン
では最下位バイトが先になります。16 ビット整数 12345 = 48× 256 + 57はビッグエ
ンディアンのマシンではアドレスが n の場所に 48 が、n + 1 の場所に 57 が格納さ
れますが、リトルエンディアンのマシンではアドレスが n の場所に 57 が、n + 1 の
場所に 48 が格納されます。このためプロトコルは複数のバイトからなる整数を転送
する場合にどちらの方法を使うのかを指定する必要があります。あるいは転送される

メッセージのヘッダにどちらの方法を使うか書くことでどちらも使えるようにするこ

ともできます。

2rfc はたくさんのサイトから匿名 ftp で入手できます。フランスでは ftp.inria.fr のディレクト

リ rfc から入手できます。リファレンスサイトは http://www.faqs.org/rfcs/ です。

OCamlシステムにはデータ構造をエンコード/デコードする (文献によってはマー
シャリング とか シリアライゼーション とか ピックリング などと呼ばれる処理です)
ための関数が二つあります。これらの関数を使うと OCamlの値とバイト列の間の変
換を行うことができます。

val output_value : out_channel -> 'a -> unit

val input_value : in_channel -> 'a

output_values と input_values は値をディスクに保存してあとで読み出せるように

するための関数ですが、任意の値をパイプやソケットへ送ることにも利用できます。

関数を除く全ての OCamlの値を処理でき、値の間の共有や循環を保存し、エンディ
アンの異なるマシン間でも利用できます。これ以上の情報は Marshal モジュールで確

認できます。

セマンティクス上は input_valueの型が間違っていることを指摘しておきます。全

ての型 'a に対して input_value の返り値の型が'a になることはないので、この関数

の型は一般的すぎます。input_value の返り値は明確な型であり、全ての可能な型で

はありませんが、これをコンパイル時にチェックすることはできません。返り値の型

は実行時に入力チャンネルから読み込むデータに依存しているからです。input_value

に正しく型検査を行うにはダイナミックオブジェクトと呼ばれる ML 言語の拡張が必
要になります。ダイナミックオブジェクトでは値が型と組にされ、実行時の型検査が

可能になります。詳細は [15] を参照してください。

例 x-ウィンドウプロトコルが OCamlで書かれていたならば、サーバへのクエスト
のためのヴァリアント型 request とサーバからのレスポンスのためのヴァリアント型

reply が定義されていたでしょう。

type request =

| FillPolyReq of (int * int) array * drawable * graphic_context

* poly_shape * coord_mode

| GetAtomNameReq of atom

| ...

and reply =

| GetAtomNameReply of string

| ...

サーバの主な処理はリクエストをデコードして返事を書くループとなるはずです:
(* 接続リクエストをディスクリプタ s に受け取る *)

let requests = in_channel_of_descr s

and replies = out_channel_of_descr s in

try

while true do

match input_value requests with

...

| FillPoly(vertices, drawable, gc, shape, mode) ->

fill_poly vertices drawable gc shape mode

| GetAtomNameReq atom ->

output_value replies (GetAtomNameReply(get_atom_name atom))

| ...

done

with End_of_file -> (* 接続の終了 *)

アプリケーションとリンクされる xライブラリの関数は次のような構造になるでしょう:

システム

RPC サーバ

ユーザ 1

ユーザ 2

マシン A

RPC サーバ

f の呼び出し

g の呼び出し

マシン B

Figure 4 — 遠隔手続き呼び出し

(* まずサーバとの接続を確立してディスクリプタを s とする *)

...

let requests = out_channel_of_descr s

and replies = in_channel_of_descr s;;

let fill_poly vertices drawable gc shape mode =

output_value requests

(FillPolyReq(vertices, drawable, gc, shape, mode));;

let get_atom_name atom =

output_value requests (GetAtomNameReq atom);

match input_value replies with

| GetAtomNameReply name -> name

| _ -> fatal_protocol_error "get_atom_name";;

* * *
遠隔手続き呼び出し バイナリプロトコルのもう一つの使用例は遠隔手続き呼び出し

です。遠隔手続き呼び出しとは、マシン A のユーザがマシン B で関数を f を実行す

るというものです。関数を直接呼び出すことは当然できません。ですが、マシン B に
接続して関数を実行し、結果をマシン A に送り返すという作業を呼び出しごとにプ
ログラムすることは可能です。

遠隔手続き呼び出しは一般的な状況であるために、これを行う rpc サービスが存
在します (図 4)。rpc サーバはマシン A と B の両方で実行され、以下のように動作
します。まずマシン A 上のユーザがマシン A 上の rpc サーバに遠隔地のマシン B
上における関数の実行をリクエストします。次にマシン A 上の rpc サーバはリクエ
ストをマシン B 上の rpc サーバにリレーし、マシン B は f を実行して結果をマシン

A 上の rpc サーバに送り返します。最後にマシン A 上の rpc サーバが結果をユー
ザに渡して終わりです。

マシン B 上で実行される他の関数に対する遠隔手続き呼び出しが、マシン A 上
の同じ rpc サーバで処理されることになります。マシン A と B にインストールされ
た rpc サービスの間で接続処理は共有され、ユーザからはこれらの遠隔手続き呼び
出しの処理が単純な関数呼び出しに見えます (図の破線)。

プロトコルの例

“テキスト” プロトコル

プロトコルの効率が重要でないネットワークサービスでは “テキスト” プロトコルが
よく利用されます。“テキスト” プロトコルは実際のところ小さなコマンド言語です。
リクエストは複数のコマンドであり、最初の語がリクエストのタイプを、残りの語が

コマンドの引数を表します。応答も一つ以上のテキストからなり、大抵は応答の種類

を表す数字コードから始まります。“テキスト” プロトコルの例をいくつか示します:

名前 説明 目的

smtp (Simple Mail Transfer Protocol) rfc 821 電子メール

ftp (File Transfer Protocol) rfc 959 ファイル転送

nntp (Network News Transfer Protocol) rfc 977 ニュースの閲覧

http/1.0 (HyperText Transfer Protocol) rfc 1945 ウェブの巡回

http/1.1 (HyperText Transfer Protocol) rfc 2068 ウェブの巡回

これらのプロトコルの大きな利点はクライアントとサーバ間のやり取りが人間にも読

めることです。例えば telnet コマンドを使ってサーバと直接対話することができま

す。host を接続するホストの名前、service をサーバで実行されたサービスの名前

(http や smtp、nntp など) として telnet host service を起動すると、クライアント

としてのリクエストを打ち込ことができ、それに対するサーバからの応答は標準出力

に出力されます。これによってプロトコルは理解しやすくなりますが、コーディング

とリクエストと応答のデコードはバイナリプロトコルよりも複雑になり、メッセージ

はより大きく、したがって非効率になる傾向にあります。

例 シェルから smtp サーバにメールを送ったときの対話ログの例を示します。→ か
ら始まる行はクライアントからサーバへ向かうテキストであり、← から始まる行は
サーバからクライアントへ向かうテキストです。

telnet margaux smtp

Trying 128.93.8.2 ...

Connected to margaux.inria.fr.

Escape character is '^]'.

← 220 margaux.inria.fr Sendmail 5.64+/AFUU-3 ready at Wed, 15 Apr 92 17:40:59

→ HELO pomerol.inria.fr

← 250 Hello pomerol.inria.fr, pleased to meet you

→ MAIL From:<god@heavens.sky.com>

← 250 <god@heavens.sky.com>... Sender ok

→ RCPT To:<xleroy@margaux.inria.fr>

← 250 <xleroy@margaux.inria.fr>... Recipient ok

→ DATA

← 354 Enter mail, end with "." on a line by itself

→ From: god@heavens.sky.com (Himself)

→ To: xleroy@margaux.inria.fr

→ Subject: Hello!

→
→ Is everything ok down there?

→ .

← 250 Ok

完全な例: http リクエスト

→ QUIT

← 221 margaux.inria.fr closing connection

Connection closed by foreign host.

コマンド HELO、MAIL そして RCPT はそれぞれクライアントのマシン名、送信者の

アドレス、受信者のアドレスをサーバに送信します。DATA コマンドは電子メールの本

文を送信してよいかをサーバに訪ねます。その後にメールの本文を入力し、'.' だけ

からなる行で終了します (この文字だけからなる行をメール本文に含めたい場合には、
'.' を二つ書いておけばサーバは一つ目のピリオドを読み飛ばします) 。
サーバからの応答は全てコメントのついた 3 桁の数字コードです。5xx の形をし

た応答はエラーを表し、2xx は全てが上手くいっていることを表します。クライアン

トが実際のプログラムであれば応答コードだけを解釈します。コメントはメールシス

テムの開発者を助けるために存在します。

* * *

6.14 完全な例: http リクエスト

http(HyperText Transfer Protocol)プロトコルは有名な “ワールドワイドウェブ”で
ドキュメントを読むのに主に使われます。この分野はクライアント-サーバの例のニッ
チな領域です。ページを読み込むクライアントとページを書き込むサーバの間には無数

の中継リレーが存在し、それらは実際のクライアントに対して仮想サーバとして振る

舞ったり、物理サーバに対して移譲されたクライアントとして振る舞ったりします。こ

れらのリレーではキャッシングやフィルタリングなどのサービスがよく実行されます。

httpプロトコルにはいくつかのバージョンがあります。本質的な部分、つまりク
ライアントとリレーのアーキテクチャ、に集中するために、一番最初のバージョンの

httpプロトコルを改変したシンプルなプロトコルを使います。このプロトコルはホ
コリをかぶっていますが、それでもほとんどのサーバによって理解されます。節の最

後にはウェブを探索する本物のツールを作るために必要となる、より現代的でより複

雑なバージョンのプロトコルを示します。ただし例をこの新しいバージョンに書き直

す部分は練習問題として残しておきます。

httpプロトコルのバージョン 1.0 は rfc 1945 で規定され、以下の形のリクエス
トを定義しています:

GET sp uri crlf

ここで sp はスペースを、crlf は文字列"\r\n" (“リターン” と改行) を表します。こ
の単純なリクエストに対する応答もシンプルであり、応答はヘッダの何もない urlの
要素そのものです。

リクエストの終端は EOF によって合図され、そこで接続が閉じられます。バー
ジョン 0.9 のプロトコルから受け継いだこの形のリクエストは一つの接続に対するリ
クエストを一つだけに制限します。

url の取得

引数 urlを一つ取り、それが表すウェブ上のリソースを取得して表示するプログラム
geturl を作ります。

最初のタスクは urlをパースしてプロトコルの名前 (ここでは "http" でなくて

はいけません)、サーバのアドレス、ポート番号 (無くても良い)、そしてサーバ上のド
キュメントの絶対パスを取り出すことです。この処理には OCamlの正規表現ライブ
ラリ Str を使います。

完全な例: http リクエスト

open Unix;;

exception Error of string

let error err mes = raise (Error (err ^ ": " ^ mes));;

let handle_error f x = try f x with Error err -> prerr_endline err; exit 2

let default_port = "80";;

type regexp = { regexp : Str.regexp; fields : (int * string option) list; }

let regexp_match r string =

let get (pos, default) =

try Str.matched_group pos string

with Not_found ->

match default with Some s -> s | _ -> raise Not_found in

try

if Str.string_match r.regexp string 0 then

Some (List.map get r.fields)

else None

with Not_found -> None;;

let host_regexp =

{ regexp = Str.regexp "\\([^/:]*\\)\\(:\\([0-9]+\\)\\)?";

fields = [ 1, None; 3, Some default_port; ] };;

let url_regexp =

{ regexp = Str.regexp "http://\\([^/:]*\\(:[0-9]+\\)?\\)\\(/.*\\)";

fields = [ 1, None; 3, None ] };;

let parse_host host = match regexp_match host_regexp host with

| Some (host :: port :: _) -> host, int_of_string port

| _ -> error host "Ill formed host";;

let parse_url url = match regexp_match url_regexp url with

| Some (host :: path :: _) -> parse_host host, path

| _ -> error url "Ill formed url";;

次の関数が示すように、シンプルなリクエストの送信は簡単な処理です。

let send_get url sock =

let s = Printf.sprintf "GET %s\r\n" url in

ignore (write sock s 0 (String.length s));;

urlはサーバのアドレスとポート番号を含む完全なものであるか、そうでなければア
ドレスとサーバ上のパスだけが含まれるものであることに注意してください。

応答はドキュメントの内容だけを含み追加の情報を含まないので、応答を読み込む

のはより簡単です。リクエストにエラーがあった場合、サーバからはエラーメッセー

ジが html ドキュメントとして返ります。このことから、応答がエラーかどうかを確
認することなく Misc.retransmit を使って出力すればすむことがわかります。プログ

ラムの残りの部分ではサーバとの接続を確立します。

完全な例: http リクエスト

let get_url proxy url fdout =

let (hostname, port), path = match proxy with

| None -> parse_url url

| Some host -> parse_host host, url

in

let hostaddr =

try inet_addr_of_string hostname

with Failure _ ->

try (gethostbyname hostname).h_addr_list.(0)

with Not_found -> error hostname "Host not found"

in

let sock = socket PF_INET SOCK_STREAM 0 in

Misc.try_finalize

begin function () ->

connect sock (ADDR_INET (hostaddr, port));

send_get path sock;

Misc.retransmit sock fdout

end ()

close sock;;

いつもどおり、コマンドライン引数のパースする処理を追加して完成です。

let geturl () =

let len = Array.length Sys.argv in

if len < 2 then

error "Usage:" (Sys.argv.(0) ^ " [ proxy [:<port>] ] <url>")

else

let proxy, url =

if len > 2 then Some Sys.argv.(1), Sys.argv.(2) else

None, Sys.argv.(1)

in

get_url proxy url stdout;;

handle_unix_error (handle_error geturl) ();;

http リレー

この節では httpリレー (あるいは プロキシ) プログラムを作成します。このプログ
ラムはクライアントからの httpリクエストを他のサーバにリダイレクト (あるいは
リレー) し、応答をクライアントに転送します。
リレーの役割を図 5 に示します。クライアントがリレーを使った場合、そのリクエ

ストは世界中に存在する個々の httpサーバではなくリレーサーバに送信します。リ
レーにはいくつもの利点があります。まず、リレーサーバは最後の、あるいは頻繁に

送られる応答を記録してリモートサーバに問い合わせることなくリクエストに答える

ことができます。これによってネットワークの過負荷を避け、サーバがダウンしてい

るときでも応答が得られるようになります。次に、広告や画像の削除といった応答の

フィルターを行うことができます。また、ワールドワイドウェブ全体を一つのサーバ

を通して見るようにすることでプログラムの開発が容易になります。

コマンド proxy port が port 番ポート (省略された場合は httpプロトコルのデ
フォルトのポート番号) にリレーサーバを作成するようなコマンド proxy を作成しま

pauillac.inria.fr

www.inria.fr

kelen.polytechnique.fr

リレー

ブラウザ

他のクライアント

Figure 5 — http リレー

す。get_url 関数を作るときに使った関数のコードを Url モジュールに入れて再利用

します。書くべきなのはリクエストの解析とサーバの立ち上げの部分だけです。

open Unix

open Url

let get_regexp =

{ regexp = Str.regexp "^[Gg][Ee][Tt][ \t]+\\(.*[^ \t]\\)[ \t]*\r";

fields = [ 1, None ] }

let parse_request line = match regexp_match get_regexp line with

| Some (url :: _) -> url

| _ -> error line "Ill formed request"

サービスを確立には establish_server 関数を使うので、定義する必要があるのは接

続リクエストを処理する関数だけです:

let proxy_service (client_sock, _) =

let service () =

try

let in_chan = in_channel_of_descr client_sock in

let line = input_line in_chan in

let url = parse_request line in

get_url None url client_sock

with End_of_file ->

error "Ill formed request" "End_of_file encountered"

in

Misc.try_finalize

(handle_error service) ()

close client_sock

プログラムの残りの部分ではサービスを確立するだけです:

let proxy () =

let http_port =

if Array.length Sys.argv > 1 then

try int_of_string Sys.argv.(1)

with Failure _ -> error Sys.argv.(1) "Incorrect port"

else

try (getservbyname "http" "tcp").s_port

with Not_found -> error "http" "Unknown service"

in

let treat_connection s = Misc.double_fork_treatment s proxy_service in

let addr = ADDR_INET(inet_addr_any, http_port) in

Misc.tcp_server treat_connection addr;;

handle_unix_error (handle_error proxy) ();;

The http/1.1 プロトコル

単純な httpリクエストはリクエストごとに新しい接続を必要としますが、これは非
効率です。なぜならリクエストには通常他のリクエストが続くからです (例えばクラ
イアントが画像つきのウェブページを受け取った場合、クライアントは続けて画像を

リクエストします)。さらに接続を確立する時間はリクエストを処理する時間を簡単に
上回ります (7 章ではプロセスではなくスレッドでリクエストを処理することでこの
時間を削減する方法を見ます)。rfc 2068 で規定される httpバージョン 1.1 では一
つの接続で複数のリクエストを処理することが可能になる複雑なリクエストが利用さ

れます3。
http/1.1 で使われる複雑なリクエストではサーバは全ての応答の先頭に応答の

フォーマットと転送されるドキュメントのサイズを表すヘッダを付けます。これによっ

てサイズが既知となってドキュメントの終端は EOF ではなくなり、更に多くのリク
エストを処理できるようになります。複雑なリクエストは以下の形をしています:
GET sp uri sp HTTP/1.1 crlf

header crlf

header の部分はキーと値からなるフィールドのリストであり、以下の構文を持ちます:
field : value crlf

':' の周りに余分なスペースを置くことが許されており、スペースはタブまたは複数

のスペースと取り替えることができます。ヘッダフィールドは複数行からなっていて

も構いません。このとき、およびこのときに限り、行末 crlf の次にスペース sp が続

きます。最後に、大文字と小文字はフィールドのキーワードと特定のフィールドの値

において無視されます。

フィールドが必須かどうかはリクエストの種類によります。例えば GET リクエス

トでは目的地となるマシンを示すフィールドが必須です。

Host : hostname crlf

この種類のリクエストでは必要に応じて If-Modified フィールドを指定することで、

ドキュメントが指定した日時から改変されている場合に限ってドキュメントを返すよ

うにすることができます。

If-Modified : date crlf

header 内のフィールドの数は前もって定まっていませんが、ヘッダの終わりは crlf

だけからなる行で示されます。

以下は完全な例です (各行の最後の \r の次には \n が続きます):
3http/1.0 プロトコルでも複雑なリクエストを使って複数のリクエストを一つの接続で処理するこ

とは可能ですが、ここではそれとは関係のない http/1.1 で使われるリクエストを説明します。

GET /~remy/ HTTP/1.1\r

Host:pauillac.inria.fr\r

\r

複雑なリクエストに対する応答も複雑になります。応答はステータスを表す行、ヘッ

ダ、そして (もしあれば)応答の本体からなります。

HTTP/1.1 sp status sp message crlf

header crlf

body

応答のヘッダのフィールドはリクエストのものと似た構文を持ちますが、必須、ある

いは必須でないフィールドが異なっています (リクエストの種類と応答のステータス
によって異なります — プロトコルのドキュメントを見てください)。
応答の body は空でなければ一つのブロックまたは複数のチャンクで送られます。

• body が一つのブロックの場合、ブロックのバイト数を 10 進で表した ascii 文字列を
指定する Content-Length フィールドがヘッダに含まれる。

• bodyがチャンクで転送される場合、ヘッダは “chunked”という値を持つ Transfer-Encoding

フィールドを持つ。データ本体は空のチャンクで終わる複数のチャンクの列であり、

各チャンクは以下の形を持つ:

size [ ; arg ] crlf

chunk crlf

ここで size はチャンクのサイズを 16 進で表し、chunk はそのサイズのデータ本体で
ある ([ と ] の間の部分は必須ではないので安全に無視できる)。最後の空のチャンク
は必ず以下の形式となる:

0 crlf

header crlf crlf

• 応答が Content-Length フィールドを持っておらず、チャンクされてもいない場合、

データ本体は空である (例えば HEAD リクエストに対する応答はヘッダしか持たない)。

一つのブロックからなる応答の例を示します:

HTTP/1.1 200 OK\r

Date: Sun, 10 Nov 2002 09:14:09 GMT\r

Server: Apache/1.2.6\r

Last-Modified: Mon, 21 Oct 2002 13:06:21 GMT\r

ETag: "359-e0d-3db3fbcd"\r

Content-Length: 3597\r

Accept-Ranges: bytes\r

Content-Type: text/html\r

\r

<html>

...

</html>

ステータス 200はリクエストが成功したことを示します。301は urlが応答の Location

フィールドで示される 他の urlにリダイレクトされたことを示します。4XX の形のス
テータスはクライアントにおけるエラーを、5XX はサーバにおけるエラーを示します。

練習問題 15 http/1.1 プロトコルで動くリレーを書いてください。

* * *
練習問題 16 リレーにキャッシュを追加してください。ページはハードドライブに

保存され、リクエストされたページがキャッシュに存在していてキャッシュが古すぎ

ない場合にはそれを返します。そうでない場合にはサーバは問い合わせを行いキャッ

シュを更新します。

* * *
練習問題 17 次の wget コマンドを作ってください。wget u1 u2 ... un は u1, u2,
. . . , un へのリクエストを送り、応答を ./m1/p1, ./m2/p2, . . . , ./mn/pn に保存します。
ここで mi と pi はそれぞれサーバの名前とリクエスト ui の絶対パスを表します。連

続するリクエストが同じマシン m に対するものである場合、プロトコルの利点を活か

してコマンドが開く接続はひとつだけにするべきです。パーマネント urlによるリ
ダイレクトは追うようにしてください。以下のオプションを追加することもできるで

しょう:

-N ファイル ./mi/ui が存在しない場合、あるいは urlよりも古い場合
にはダウンロードしない

-r htmlフォーマットの応答に含まれる urlを再帰的に全てダウンロー
ドする。

* * *

7 スレッド

スレッド とは一つのプログラムの中で互いに並列に実行できる制御の流れであり、軽

量プロセス とも言われます。

この章ではスレッドを作成し (Thread モジュール) 同期を行うプログラムを説明
します。同期にはロック (Mutex モジュール)、条件変数 (Condition モジュール)、そ
して同期イベント (Event モジュール) が使われます。

7.1 イントロダクション

fork は現在のプロセスのコピーを作成します (したがってプログラムのコピーも作成
します) が、スレッドの作成はこれとは大きく異なります。フォークを行った場合親
プロセスと子プロセスのアドレス空間は完全に分離し、二つのプロセスは (ファイル
やパイプに対する入出力といった) システムコールを通じてでしかやり取りをするこ
とができません。

これに対して、あるプログラム内の全てのスレッドは同じアドレス空間を共有し

ます。スレッド間で共有されず、スレッド同士で異なる唯一の情報はスレッドの ID
と実行スタック (シグナルマスクやロックと条件変数の状態などのシステム情報を含
む) です。この観点では、スレッドはコルーチンに似ています。プログラム内のスレッ
ドは全て同じ方法で扱われますが、プログラムが始まって最初に作られるスレッドは

別です。このスレッドが終了すると、そのプログラムが持つ全てのスレッドが終了し

したがってプログラム全体が終了します (これから複数のスレッドといった場合には
考えているプログラム内の複数のスレッドを意味し、別のプログラム内で動いている

スレッドのことは意味しないことにします)。
制御を明示的にに受け渡し並列に実行できないコルーチンと違い、スレッドは並

列に実行でき、(他のスレッドやプロセスではなく) システムによってプリエンプティ
ブにスケジュールされます。この観点ではスレッドはプロセスに似ています。

アドレス空間が共有されることで、スレッド同士は共有メモリを使って直接通信す

ることができます。スレッドは並列に実行されるために、全てのスレッドは共有デー

タに対するアクセスを同期しなければいけません。あるスレッドが書き込み終わって

から別のスレッドが読み込むようにするためです。原理上は必要なわけではありませ

んが、実際にはこれを行うためにオペレーティングシステムを介する必要があります。

スレッドの同期はロックや条件変数を使うか、あるいはイベントを使って高レベルに

行われます。

プロセスに対するスレッドの利点は低い作成コストと大きなデータを受け渡すと

きにデータ全体をコピーするのではなくポインタを渡すだけですむことです。

一方、スレッドを使うとスレッド同士の同期を管理しなければならず、スレッドが

致命的なエラーを出す場合にも対処する必要があります。例えばスレッドは停止する

前にロックを開放し不変量が保存されるようにする必要があります。上記の利点によ

る恩恵があまりないのであればスレッドよりもプロセスのほうが望ましいでしょう。

OCaml における実装 ネイティブスレッドを使うアプリケーションをコンパイルす

るには次のようにします:

ocamlc -thread unix.cma threads.cma -o prog mod1.ml mod2.ml mod3.ml

ocamlopt -thread unix.cmxa threads.cmxa -o prog mod1.ml mod2.ml mod3.ml

ocamlbuild ツールを使うのであれば、_tags ファイルに以下の内容を追加するだけで

すみます:

<mod{1,2,3}.ml> : thread

<prog.{native,byte}> : use_unix, thread

インストールされた OCamlがネイティブスレッドをサポートしない場合、7.8 節およ
びマニュアルにあるシミュレートされた “vm レベル” スレッドを使う方法を参照し
てください。この章のこれ以降の内容はネイティブスレッドを仮定し、vm レベルス
レッドには一般的では成り立ちません。

7.2 スレッドの作成と終了

この章で説明される関数は Thread モジュールに定義されています。

システムコール create f v は関数適用 f v を実行する新しいスレッドを作成し、

スレッド ID を返します。呼び出し側はこの ID を使って新しく作られたスレッドを
制御することができます。

val create : ('a -> 'b) -> 'a -> t

新しいスレッドで実行される f v という関数適用はプログラム内の他のスレッドと並

行に実行されます。関数適用が値を返すとスレッドは終了し、返り値は無視されます。

スレッドが捕捉されないエラーによって終了したとしても、そのエラーが他のスレッ

ドに及ぶことはありません。そのエラーに関するメッセージが標準エラー出力に出力

されるか、エラーは完全に無視されます。他のスレッドは独立して実行されるので、

このエラーを受け取ることはできません。

スレッドはこの他にも Thread モジュールのシステムコール exit を呼ぶことに

よってでも終了できます。Pervasives.exit と混同しないようにしてください。これ

はプログラム全体、つまり全てのスレッドを終了します。

val exit : unit -> unit

プログラムの最初のスレッドは終了するときに暗黙のうちに Pervasives.exit を呼び

ます。

最初のスレッドが終了する前にそれ以外のスレッドが終了すると、そのスレッド

は OCamlのランタイムライブラリによってすぐに開放されます。fork で作成された

Unix プロセスのようにゾンビになることはありません。
システムコール self は呼び出したスレッドのスレッド ID を返します。

val self : unit -> t

前の章で作成した “フォーク”と “ダブルフォーク”を使った並行サーバを、子プロセ
スではなくスレッドを使うように書き換えるための準備が整いました。そのようなサー

バを作るためには、6.7節で定義した Misc.fork_treatmentに似た Misc.co_treatment

を作成します。

let co_treatment server_sock service (client_descr, _ as client) =

try ignore (Thread.create service client)

with exn -> close client_descr; raise exn;;

スレッドの作成に成功した場合、サービスと client_descrを閉じる処理は service関数

によって処理されます。スレッドの作成に失敗した場合、ディスクリプタ client_descr

を閉じてクライアントを放棄し、エラーに対する処理をメインプログラムに任せます。

スレッドを使ったサーバでは難しい部分が service 関数に隠れていることに注意

してください。この関数は接続を切断するまでロバストに処理しなければいけません。

プロセスを使った並行サーバの例ではサービスが他のプロセスによって実行されてお

り、エラーが起こってサービスが途中で終了したとしても望みどおりの処理 — 接続

の処理 — が行われます。これはプロセスが終了するとシステムがファイルディスク

リプタを閉じるためです。しかしサービスがスレッドによって実行されている場合、

デフォルトの設定ではディスクリプタはスレッド間で共有され、スレッドの終了時に

閉じられることがありません。そのためスレッドは終了する前にディスクリプタを閉

じる必要があります。加えて、サービスは致命的なエラーのときに Pervasives.exit

を呼ぶこともできません。なぜならこの呼び出しはサービスだけではなくサーバ全体

を終了させるからです。致命的なエラーのときに Thread.exit を使っても解決になら

ないことが多いです。なぜならスレッドの持つ開いたままのリソース、つまり接続が

閉じられていない可能性があるからです。

一つの解決法は終了処理を実行するコードがハンドラとして登録されている例外

(例えば Exit 例外) を使うことです。また同様の理由でサービスの実行中に sigpipe

シグナルをブロックし、シグナルに対する動作を動作をスレッドの終了から EPIPE 例

外に変えておくことが重要です。

7.3 スレッドの終了待ち

この節で説明される関数は Thread モジュールに定義されています。

システムコール join を使うとあるスレッドから他のスレッドの終了を待つことが

できます。

val join : t -> unit

この関数を呼んだスレッドの実行は指定された ID のスレッドの実行が終了するまで
中断されます。主スレッドがこの関数を使えば終了する前に他の全てのスレッドが終

了するのを待つことができます (この関数を使わないと主スレッドが終了した時点で
他の全てのスレッドも終了します)。

join はブロックするので “遅い” システムコールですが、この関数はシグナルを受
け取ったときに自動的に再開します。正確にはシグナルによって実行が中断され、ハン

ドラが呼び出されてから、もう一度関数が呼ばれます。そのためこの関数は指定した

スレッドが本当に終了するまで返らず、EINTR 例外を出すことはありません。OCaml
プログラマの観点からは、join はシグナルは関数から返った瞬間に実行されるように

振る舞います。

スレッドは非同期に実行されるので、値を返すことがありません。しかしその動

作は — 幸運にも! — 副作用によって観測することができます。例えばスレッドは計

算の結果を参照値に保存し、他のスレッドは計算が終わったことを確認してからその

値を読み込むことができます。次の例がこのことを説明します。

exception Exited

type 'a result = Value of 'a | Exception of exn

let eval f x = try Value (f x) with z -> Exception z

let coexec (f : 'a -> 'b) (x : 'a) : unit -> 'b =

let result = ref (Exception Exited) in

let p = Thread.create (fun x -> result := eval f x) x in

function () -> match (join p; !result) with

| Value v -> v

| Exception exn -> raise exn;;

let v1 = coexec succ 4 and v2 = coexec succ 5 in v1 () + v2 ();;

システムは他のスレッドに制御を移すため、あるいはスレッドが他のスレッドや

プロセスが利用中のリソース (例えばロックや条件変数、ファイルディスクリプタ) を
待つために、スレッドを停止することができます。yeild 関数を使うとシステムによ

るプリエンプションを待つことなく明示的に実行を停止することができます。

val yield : unit -> unit

この関数はスレッドスケジューラに対してヒントを出しますが、なんの効果も持たな

い場合もあります。例えば yield が呼ばれた時点で他に実行できるスレッドがない場

合は、システムは同じスレッドの実行を続けます。

また他のスレッドが実行されるためには yeild を呼ばなくてはいけないわけでは

ありません。システムは yield コマンドを任意のタイミングで実行する権限を持って

いるからです。実際にはシステムはこの権限を頻繁に使って他のスレッドを実行し、

プロセスが一つしか無いマシンでもスレッドが並列に実行されているように見せてい

ます。

例 3.3 節の例をプロセスではなくスレッドを使って書き直します。

1 let rec psearch k cond v =

2 let n = Array.length v in

3 let slice i = Array.sub v (i * k) (min k (n - i * k)) in

4 let slices = Array.init (n/k) slice in

5 let found = ref false in

6 let pcond v = if !found then (Thread.exit (); cond v) else cond v in

7 let search v = if simple_search pcond v then found := true in

8 let proc_list = Array.map (Thread.create search) slices in

9 Array.iter Thread.join proc_list;

10 !found;;

psearch k f v は k このスレッドを使って配列の要素のうち f を満たすものを並列に

探索します。pcond は探索がすでに成功している場合にスレッドを終了させることで

探索を打ち切ります。全てのスレッドは参照 found を共有し、アクセスは並列に起こ

ります。この参照への書き込みは必ず値を true にするので、クリティカルセクショ

ンは存在しません。探索対象が見つからなかったときに全体の探索結果をセットして

はいけません。例えば 7 行目を

let search v = found := !found && simple_search pcond v

あるいは:

let search v = let r = simple_search pcond v in found := !found && r

に入れ替えると正しく動かなくなります。

* * *
要素の比較が (デイスクアクセスやネットワークへの接続で) 一時的なブロックを

引き起こすならば、並列探索はプロセスが一つしか無いマシンでも意味があります。

この場合リソースを待っているスレッドが他のスレッドに実行を譲ることで配列の他

の部分の計算を続けることができ、ブロックされたスレッドの実行はリソースが利用

可能になってから再開されます。

特定の要素へのアクセスがとても大きなレイテンシを持つという場合があり、と

くにネットワーク越しにデータを取得しなければいけない場合にはレイテンシが他の

要素に比べて数百倍となります。このようなケースではシーケンシャルな探索と並列

な探索の差が明確になります。

練習問題 18 配列に対するクイックソートを並列化してください。解答は 124ページ。

* * *

これから紹介するスレッドを停止させる別の方法は、オペレーティングシステム

のリソースに関係しています。スレッドは delay s を呼ぶことで任意の時間だけ実行

を停止することができます。実行は s 秒後に再開されます。

val delay : float -> unit

この関数は vm-レベルのスレッドでもポータブルに使うことができますが、delay s

は ignore (Unix.select [] [] [] s) の省略形に過ぎません。join と違ってこの関

数はシグナルで中断されても自動的に再開しません。

スレッドを外部の操作と同期させるために select を使うことができます。この関

数がブロックするのはプログラム全体ではなくこの関数を読んだスレッドだけである

ことに注意してください。シミュレートされたスレッドが Unix モジュールの select

を呼ぶとプログラム全体すなわち全てのスレッドをブロックしてしまうために、Thread

モジュールはこの関数を再定義しています。ネイティブスレッドを利用するプログラ

ムでは Unix.select と Thread.select の動作に違いはありませんが、この理由から

ネイティブスレッドを利用するプログラムでも Unix モジュールではなく Thread モ

ジュールの select 関数を使うことが必要になります。

例 5.2 節のエラトステネスのふるいの例を Unix プロセスではなくスレッドを使っ
て動かすためには、filter 関数の 5–16 行目を

let p = Thread.create filter (in_channel_of_descr fd_in) in

let output = out_channel_of_descr fd_out in

try

while true do

let n = input_int input in

if List.exists (fun m -> n mod m = 0) first_primes then ()

else output_int output n

done;

with End_of_file ->

close_out output;

Thread.join p

に変え、さらに sieve 関数の 4–10 行目を
let k = Thread.create filter (in_channel_of_descr fd_in) in

let output = out_channel_of_descr fd_out in

generate len output;

close_out output;

Thread.join k;;

に変えれば十分です。ただし、計算時間に比べてプロセスをあまり使わないこの例で

は実行速度の高速化は望めません。

Figure 6 — 共有リソースへのアクセスの競合

* * *

7.4 スレッドの同期: ロック

この節で説明される関数は Mutex (Mutual exclusion) モジュールに定義されています。
前の節で変更可能なリソースに対する並行アクセスの問題に触れました。次のシ

ナリオは共有リソースへのアクセスの問題を説明します。二つのスレッド p と q がカ
ウンタ c の値を進めようとしているとします。
図 6にあるシナリオを考えます。スレッド p がカウンタ c の値を読み、実行が q

へ移ります。この次に q は c の値を読み、値 k + 1 を c に書き込みます。次に実行は
p へと戻りますが、このスレッドは k + 1 を c に書き込みます。この結果、最小的な
c の値は k + 2 ではなく k + 1 となってしまいます。
この古典的な問題はロックを使って p と q が交互に実行されることをなくすこと

で解決できます。

ロックとは共有オブジェクトであり、プログラムの各時点でロックを所有できる

のは多くとも一つのスレッドだけです。ロックは create 関数で作成します。

val create : unit -> t

この関数はどのスレッドにも所有されていない新しいロックを返します。作成したロッ

クを取得するには、そのロックに対してシステムコール lock を呼ぶ必要があります。

目的のロックが他のスレッドによって所有されていた場合、lock を読んだスレッドは

ロックが開放されるまでブロックします。ロックを所有しているスレッドはシステム

コール unlock によってロックを開放しなければいけません。

val lock : t -> unit

val unlock : t -> unit

シグナルへの動作について、lock は Thread.join に似ています。もし lock の実行中

にシグナルを受け取った場合、そのシグナルは記録されOCamlランタイムがシグナ
ルの到着を通知しますが、関数を呼び出したスレッドは lock がロックが取得できる

までブロックされ、lock から ENITR 例外が出ることはありません。OCamlによるシ
グナルの処理は lock が返るタイミングで行われます。

システムコール trylock を使うとブロックせずにロックの取得を試みることがで

きます。

val try_lock : t -> bool

この関数はロックが取得できた場合に true を返し、それ以外のときには false を返

します。後者の場合、ロックは取得されないのでスレッドの実行が停止することはあ

りません。そのためスレッドは他の処理を行ってからもう一度ロックの取得を試みる

ことができます。

例 複数のスレッドによって利用されるグローバルなカウンタを進める処理には同期

の問題があります。カウンタの値を読んでから新しい値を書き込むまでの間はクリティ

カルセクション、つまり二つ以上のスレッドが同時に実行してはいけない区間になり

ます。ロックを使えば同期を簡単に管理できます。

type counter = { lock : Mutex.t; mutable counter : int }

let newcounter () = { lock = Mutex.create (); counter = 0 }

let addtocounter c k =

Mutex.lock c.lock;

c.counter <- c.counter + k;

Mutex.unlock c.lock;;

カウンタの値を読むだけであれば同期の問題は起こりません。カウンタの変更と同時

に読み込みが起こることはありえます。しかしその場合書き込みの直前または直後の

値を読み込むだけであり、いずれにせよ矛盾が起きることはありません。

* * *
ある関数を実行する間ロックを保持するというパターンがよく必要になります。こ

のパターンでは関数が成功するか失敗するかにかかわらず最後にロックを開放するこ

とが当然必要になります。この処理を次のライブラリ関数にまとめます:

let run_with_lock l f x =

Mutex.lock l; try_finalize f x Mutex.unlock l

一つ前の例は次のように書けます:

let addtocounter c =

Misc.run_with_lock c.lock (fun k -> c.counter <- c.counter + k)

例 スレッドを使ったサーバのモデルの代わりに、リクエストを並列に処理するスレッ

ドをいくつか事前に作っておくことができます。

val tcp_farm_server :

int -> (file_descr -> file_descr * sockaddr -> 'a) -> sockaddr -> unit

tcp_farm_server 関数は tcp_server と同じように動作しますが、開始するスレッド

の数を表す引数を追加で取ります。スレッドは同じアドレスに対するサーバになりま

す。スレッドプールを使う利点はスレッドを一括して事前に作っておくことで接続を

処理するときにスレッドを作る処理の分だけ時間を節約できる点です。

let tcp_farm_server n treat_connection addr =

let server_sock = Misc.install_tcp_server_socket addr in

let mutex = Mutex.create () in

let rec serve () =

let client =

Misc.run_with_lock mutex

(Misc.restart_on_EINTR accept) server_sock in

treat_connection server_sock client;

serve () in

for i = 1 to n-1 do ignore (Thread.create serve ()) done;

serve ();;

唯一注意するべき点は accept 関数を排他的に実行することです。これによって接続

を受け付ける関数が常に多くとも一つとなります。ここでは treat_connection 関数

100



完全な例: http リレー

がシーケンシャルな処理を行うことを想定していますが、これが必須なわけではあり

ません。スレッドプールとスレッドの作成を同時に行って使用率に応じてスレッドの

数を調整することもできます。

* * *
ロックの取得がブロックせずに成功した場合、処理はすぐに終了します。ロックの

取得は全ての現代的なプロセッサで提供される “test-and-set” 命令 (とキャシュの更
新などのコストの小さないくつかの処理) で一般的に実装されているためです。一方
ロックが利用可能でない場合は、スレッドは実行を停止しもう一度スケジュールされ

るまで待たなくてはならず、大量のコストがかさみます。そのためこのコストを取るの

は他のスレッドに実行を受け渡すために本当に実行を中断するときだけにするべきで

あり、ロックを取得しようとした結果実行がブロックされたというのはできる限り避

けるべきです。したがってほとんど常にロックはなるべく早く開放して必要になった

らもう一度取得するようにするべきで、ロックを保持したままでいることは避けるべ

きです。ロックの開放を行わないとクリティカルセクションが大きくなり、他のスレッ

ドのロックの取得が競合し、実行を停止しなければならなくなる頻度が上がります。

ロックによってスレッドが交互に実行されなくなりますが、そのかわりデッドロッ

クの危険性があります。例えばスレッド p がロック v の開放を待っていて、v を持っ
ているスレッド q は p が持っているロック u を待っている場合です (最悪な場合に
は、スレッドが自分が持っているロックを待つこともあります) 。並行プログラミン
グは難しく、デッドロックの排除は常に簡単なわけではありません。デッドロックを

避ける単純で実行可能なことが多い方法は、ロックに階層を定義してロックの取得が

階層をたどるようにするというものです。スレッドがあるロックを取得できるのはそ

のロックよりも下のロックを全て取得しているときだけです。

7.5 完全な例: http リレー

6.14 節で作成した httpリレーを改変し、リクエストをスレッドを使って処理するよ
うにします。

直感的には、プロセスのクローンを作成する establish_server 関数をスレッドを

作るようにすればすみます。しかし、いくつか注意点があります。複数のスレッドを使

用する上での困難はメモリ空間を全て共有することです。そのためあるスレッド行っ

たことを他のスレッドが打ち消してしまうような、“互いのつま先を踏み合う” ことの
ないようにしなければなりません。これは二つのスレッドが変更可能な同じ構造体を

並列に変更するときによく起こります。

httpサーバにはいくつかの変更が必要になります。まずリソースへのアクセスの
問題から始めます。6.14 節で説明した proxy_service 関数は接続への対応を処理しま

す。parse_host と parse_url そして parse_request を途中で使って、proxy_service

は Str ライブラリの regexp_match 関数を呼びます。しかし、このライブラリは再入

可能ではありません (最後の探索の結果がグローバル変数にか格納されます)。一見害
の無いように見える関数呼び出しも内部にデータの衝突を隠しているかもしれないの

で注意しなければいけないことをこの例は示しています。今回は Str ライブラリを書

き直すのではなく単純にその利用をシーケンシャルにします。ライブラリ関数の呼び

出しをロックで防御するだけで十分です (他の選択肢があるわけでもありません)。例
外によって関数が異常終了した場合にもロックを開放するように注意する必要があり

ます。

現在のコードへの変更を最小限にするために、Url モジュールの regexp_match 関

数の名前を unsafe_regexp_match に変更し、新たに regexp_match 関数を防御された

unsafe_regexp_match として作成します。

let strlock = Mutex.create ();;

let regexp_match r string =

Misc.run_with_lock strlock (unsafe_regexp_match r) string;;

変更は最小限だと言えるでしょう。regexp_match関数が正規表現に対するマッチン

グとマッチしてグループの取得を両方を含むことを指摘しておきます。Str.string_match

と Str.matched_group を個別にロックで防御するのは完全な間違いです。

もう一つの解法は Str ライブラリを使わずに解析関数を書き直すことです。しか

しライブラリ関数を同期して実行するのは簡単であり、実行を遅くしないことが実行

すれば分かるので、この方法を取るべき理由は何もありません。ただ Str が最初から

再入可能であればそのほうが良かったでしょう。

他の関数はすでに再入可能です。例えば Misc.retransmit 関数は呼び出しごとに

違うバッファを確保します。

これ以外にエラー処理に関連して気をつけなければいけないことがあります。前に

書いたように、スレッドによる接続の処理はロバストでなければいけません。特にエ

ラーが起こった場合に他のスレッドが影響を受けてはいけません。つまりそのスレッ

ドは受け持った接続を適切に閉じて他の待っている接続を受け付ける状態に戻るとい

う “通常” 終了をする必要があります。
このためにまず handle_error 関数中の exit の呼び出しを Thread.exit に変更し

ます。これは exit がプロセス全体を終了させるためです。しかしただ Thread.exit

を読ぶだけでは正しくなりません。なぜならプロセスの終了と違ってスレッドの終了

は (共有された) ディスクリプタを閉じないからです。接続処理中にエラーが起きて
も接続は開いたままなので、Exit 例外を出して最終処理を行うようにする必要があり

ます。そして treat_connection 関数は EPIIE だけでなく Exit を含んだ全ての例外

を補足しなければいけません。例外を捕捉するように関数を呼び出す関数を使ってこ

れを行います:

let allow_connection_errors f s =

try f s with Exit | Unix_error(EPIPE,_,_) -> ()

let treat_connection s =

Misc.co_treatment s (allow_connection_errors proxy_service) in

練習問題 19 http/1.1 プロトコルのプロキシをスレッドを使って書き直してくだ
さい。

* * *
練習問題 20 コルーチンは別のスレッドへの実行の受け渡しを常に明示的に行う必

要がある非常に特殊なスレッドと見ることができます。スレッドを使ってコルーチン

を実装してください。

* * *

7.6 条件変数

この節で説明される関数は Condition モジュールに定義されています。

ロックを使った同期はとてもシンプルですが、機能が十分ではありません。ロッ

クを使うと共有データが開放されるのを待つことができますが、データがある状態に

なるのを待つことはできません。

カウンタの例をスレッド感で共有される (先入れ先出しの) キューを使って書き直
してみましょう。キューに要素を追加することはこれまで説明してきたロックによっ

102



条件変数

て行えます。これはキューの状態にかかわらず要素を追加することは常に可能なため

です。

しかしキューから要素を削除する場合はどうでしょうか? キューが空のときは何を
すればいいでしょうか? キューが空の場合にロックを持ちながら新しい要素を待つこ
とはできません。なぜならキューに要素を追加するにはロックが必要なために、ロッ

クを持っている限りキューに新しい要素が足されることはないからです。だとすれば、

キューが空でないことをどうやって知ることができるでしょうか?キューの状態を定
期的にチェックする方法は “ビジーウェイト” と呼ばれ、望ましい解決法ではありま
せん。(周期が短ければ) 計算機のサイクルを無駄遣いし、(周期が長ければ) 反応時間
が長くなるからです。

条件変数 がこの問題を解決します。ロックを持つスレッドは条件変数に対して別

のスレッドがシグナルを送るまで待つことができます。ロックと同じように、条件変

数は同期関数によって生成される中身の見えないオブジェクトです。create 関数で条

件変数を作ることができます。

val create : unit -> t

システムコール wait を使うと、ロック v を すでに持つ プロセス p は条件変数 c と

ロック vに対して待つことができます。この呼び出しによって プロセス p はシステ
ムに条件変数 c とロック v に対して待っていると伝え、ロック v を開放して実行を

中断します。他のスレッド q が条件変数 c の変更をシグナルし、かつロック v が利

用可能になると v を持った状態で実行が再開されます。

val wait : t -> Mutex.t -> unit

ロック v を取得していない状態で wait c v を呼ぶとエラーとなります。またシグナ

ルに対する wait c v の動作は Mutex.lock と同じです。

スレッドが条件変数が変更されたシグナルを送ると、その条件変数に対して待っ

ている全てのスレッドを起動する (システムコール broadcast) か、それらのうち一つ
を起動する (システムコール signal) かのどちらかが行われます。

val signal : t -> unit

val broadcast : t -> unit

条件変数のシグナルの送信またはブロードキャストは (wait と違って) ロックを必要
としません。そのためシステムのエラーは起きませんが、それでもプログラミング上

のエラーは起きます。

スレッドを一つだけ起動するのか全て起動するかは問題によります。キューの例

をもう一度考えると、スレッドが空のキューに要素を入れたときに待っているスレッ

ドを全て起動する必要はありません。キューから要素を取り出して処理を行えるのは

一つのスレッドだけだからです。一方、もしスレッドが複数の要素をキューに入れ、

その要素の数が静的にわからない場合には待っている全てのスレッドを起動する必要

があります。

空でないキューに要素を追加するときにシグナルを出さない場合には、空のキュー

への要素の追加は常にブロードキャストを行う必要があることに注意してください。

空のキューへの要素の追加されてからシグナルが発せられるまでに他の要素が追加さ

れることがあり、このときは要素が複数追加されたとして対処しなければならないた

めです。

通常スレッドは他のスレッドが条件変数に対して待っている理由を近似的にしか

知りません。そのため他のスレッドが条件変数を待っているかもしれない状況では常

にシグナルすることが求められます。また起動されたスレッドは期待した状況が整っ

ていると仮定してはいけません。一般的には、起動されたスレッドは共有状態のデータ

を読み取って所望の条件が成り立っていることを確認し、必要ならばもう一度待つ必

要があります。この処理は他のスレッドがシグナルしたときにだけ起こるのでビジー

ウェイトとは違います。

このアプローチを正当化する理由がもう一つあります。スレッドが大量のリソー

スを生み出して broadcast を使って全てのスレッドを起動したときに、最初に実行さ

れるスレッドは貪欲に全てのリソースを消費することが可能な点です。このとき二番

目以降に起動されるスレッドは次回はもっとラッキーになるように願いながらもう一

度実行を中断することになります。

今までの説明でキューに関する完全な回答の準備が整いました。Queue に定義され

ているキューの構造がロックと条件変数 non_empty で拡張されます。

1 type 'a t =

2 { queue : 'a Queue.t; lock : Mutex.t; non_empty : Condition.t }

3 let create () =

4 { queue = Queue.create ();

5 lock = Mutex.create (); non_empty = Condition.create () }

6

7 let add e q =

8 Mutex.lock q.lock;

9 if Queue.length q.queue = 0 then Condition.broadcast q.non_empty;

10 Queue.add e q.queue;

11 Mutex.unlock q.lock;;

12

13 let take q =

14 Mutex.lock q.lock;

15 while Queue.length q.queue = 0

16 do Condition.wait q.non_empty q.lock done;

17 let x = Queue.take q.queue in

18 Mutex.unlock q.lock; x;;

要素の追加ではスレッドはブロックしませんが、追加の前にキューが空だった場合に

条件変数 non_empty をシグナルすることを忘れてはいけません。この条件変数に対し

て待っているスレッドが存在する可能性があるからです。

要素の削除はもう少し複雑です。スレッドはロックを取得してから削除を試みま

すが、キューが空だった場合には条件変数 non_emptyに対して待ち起動されたあとに

もう一度同じことを行います。同じことができるのは起動されたスレッドがロックを

持っているからです。

上で説明したように、broadcast q.non_empty によるシグナル (9 行目) はすでに
ロック q.lock を持っているスレッド p によって実行されます。これは読み込むを行
うために take を実行するスレッド q が 15 行目と 16 行目の間の、キューが空であ
ることは確認したがまだ実行を中断していない状態にないことを意味します。もしス

レッド q がこの状態になったとすると、p によって送られるシグナルは無視され効果
を持ちません。その後 q は実行を中断しますが、p がすでにシグナルしているために
q が起動されることはありません。スレッド p によるロックの取得は q がすでに実行
を中断したかキューの状態を調べていないかのどちらかであり、キューの状態を調べ

たが実行を中断していないという状況にならないことを保証します。

練習問題 21 有界なキューを実装してください。キューへ追加されるのはキューが

ある長さに達したときにブロックする機能です (並行プログラミングで限りなく生産

する生産者と実行がブロックする消費者が存在する場合にこのキューが必要になりま

す)。解答は 124 ページ。

* * *

7.7 イベントベースの同期通信

この節で説明される関数は Event モジュールに定義されています。

ロックと条件変数を使えばどんな形の同期も行うことができますが、必ずしも簡単

に行えるというわけではありません。最初は単純だったキューでも同期のためのコー

ドが緻密になることは前節の例で見ました。

イベントベースの同期通信は高レベルの通信プリミティブを集めたものであり、並

行プログラミングを容易にすることを意図しています。Eventモジュールのプリミティ

ブはまず John Reppy によって Standard ML 言語の拡張 Concurrent ML [16] とし
て開発されました。OCamlではこれらの通信プリミティブはロックと条件変数などの
基本的な同期よりも上のレベルに位置しています。

通信はイベントをチャンネルを通じて送ることで行います。チャンネルとは “軽
量パイプ” のようなものです。これを使うと同じプログラム内のスレッド間で通信を
行い、生産者と消費者の間で同期を行うことができます。'a 型の値を運ぶチャンネル

は 'a channel 型を持ちます。チャンネルは途中で型を変えることなく常に同じ型の

値を運びます。チャンネルは new_channel 関数で作成されます。

val new_channel : unit -> 'a channel

メッセージの送受信は直接的にではなくイベントという中間体を通して行われます。

基本的なイベントは “メッセージの送信” と “メッセージの受信” であり、以下のプリ
ミティブによって作成できます。

val send : 'a channel -> 'a -> unit event

val receive : 'a channel -> 'a event

メッセージの作成してもすぐに何かが起こるわけではありません。行われるべき動作

を示すデータ構造が作られるだけです。イベントを実際に起こすには、スレッドはそ

のイベントを実行すべきスレッドと同期しなければいけません。sync プリミティブを

使うと引数として渡すイベントが実際に実行されるのを待つことができます。

val sync : 'a event -> 'a

まとめると、値 v をチャンネル c に送るには sync (send c v) を実行します。sync

を読んだスレッドはイベントが起こるまで、つまり他のスレッドがチャンネル cから

値を受信可能になるまでブロックされます。対称的に、スレッドがチャンネル c に対

するメッセージを待つには sync (recieve c) を使います。

全ての生産者と全ての消費者は競合します。例えばいくつかのスレッドが一つの

チャンネルにメッセージを送りメッセージを受信できるスレッドが一つしか無かった

場合、自分の送ったイベントが起きる生産者は一つだけなのは明らかです。他の生産

者は実行を中断されたままであり、先に他のスレッドがイベントを受け取ったことに

気がつくことはありません。

競合は同じスレッド内でも起こりえます。複数のイベントは choose プリミティブ

を使うとまとめることができます。

val choose : 'a event list -> 'a event

この関数は引数に渡されたイベントを並列にオファーし、これらのイベントのうちちょ

うど一つが実行されます。ここではオファーと実行を区別していることに注意してく

ださい。sync (choose [e1; e2]) は二つのイベント e1 と e2 の選択をオファーし、同

期したときには二つのうち一つのイベントしか実行されません (実行と同時に同時に
もう一つのイベントはキャンセルされます)。wrap_abort プリミティブを使うとイベ

ントがキャンセルされたときの処理を設定できます。

val wrap_abort : 'a event -> (unit -> unit) -> 'a event

wrap_abort e f は e と同じイベントを作成しますが、同期したときにこのイベント

が実行されなかった場合には f が実行されます (このイベントが複雑なイベントの一
部であるときにだけ意味を持ちます)。

poll を使うと (Mutex.try_lock のように) ブロックせずにイベントの同期を試み
ることができます。

val poll : 'a event -> 'a option

poll e はイベント e をオファーしますが、このイベントがすぐに実行されない場合

にはブロックせずにオファーをキャンセルします。このとき関数呼び出しはなんの効

果も持ちません (より正確には、poll e という式が None で置き換えられたかのよう

な動作をします)。これに対して、もしイベントがすぐに実行できるならば sync e が

実行されたときの動作をします。ただしその時の返り値は v ではなく Some v となり

ます。

例 7.3 節で示したエラトステネスのふるいの例では、異なるスレッド感の通信には
元のプログラムと同じパイプが使われ、システムメモリ (パイプ) が中間体でした。プ
ロセスのメモリを使って直接通信を行ったほうが効率的な可能性もあります。単純な

解法はパイプを整数が送られるチャンネルで置き換えることです。

整数をチャンネルに送るだけでは十分ではなく、ストリームの終端を検出する必

要もあります。最も単純な解法は整数を Some n の形で送り、None を送ることで終了

することです。変更を最小にするために、5.2 節の例で使ったコードを再利用します。
パイプとその入出力関数をチャンネルとその入出力関数でシミュレートします。

前のバージョンのプログラムを取ってきて入出力関数を Pervasives ライブラリの

入出力バッファを使ったものからチャンネルを使ったものに入れ替えればすみます。

例えば次のコードを open Unix 命令の後に追加します。

let pipe () = let c = Event.new_channel () in c, c

let out_channel_of_descr x = x

let in_channel_of_descr x = x

let input_int chan =

match Event.sync (Event.receive chan) with

| Some v -> v

| None -> raise End_of_file

let output_int chan x = Event.sync (Event.send chan (Some x))

let close_out chan = Event.sync (Event.send chan None);;

しかしチャンネルを使ったバージョンの実行速度を前のバージョンと比べると、

チャンネルを使うほうが二倍程度遅いことがわかります。整数を通信するときに毎回

二つのスレッドの同期が必要となり、ロックの取得と開放に何回かのシステムコール

が必要となるためです。一方で、バッファされた i/oを使うパイプを用いた通信では
一回のシステムコールで数千個の整数がまとめて送られます。

公平な比較のためには、チャンネルを使った通信にバッファを用意していくつかの

整数をまとめてチャンネルを利用するようにするべきです。子はプライベートなキュー

に計算結果を他のスレッドと同期することなく蓄積します。キューが満杯になるか明

示的な要求が届くと、チャンネルが同期されてキューが空になります。親は自分用の

キューを持ち、このキューはチャンネル同期によって満たされ計算が進むに連れて一

つづつ消費されます。

このときのコードは以下のようになります:

type 'a buffered =

{ c : 'a Queue.t Event.channel;

mutable q : 'a Queue.t;

size : int }

let pipe () = let c = Event.new_channel () in c, c;;

let size = 1024;;

let out_channel_of_descr chan =

{ c = chan; q = Queue.create (); size = size };;

let in_channel_of_descr = out_channel_of_descr;;

let input_int chan =

if Queue.length chan.q = 0 then begin

let q = Event.sync (Event.receive chan.c) in

if Queue.length q > 0 then chan.q <- q

else raise End_of_file

end;

Queue.take chan.q;;

let flush_out chan =

if Queue.length chan.q > 0 then Event.sync (Event.send chan.c chan.q);

chan.q <- Queue.create ();;

let output_int chan x =

if Queue.length chan.q = size then flush_out chan;

Queue.add x chan.q

let close_out chan =

flush_out chan;

Event.sync (Event.send chan.c chan.q);;

このバージョンではパイプを使ったバージョンと比較可能な程度の効率を取り戻すこ

とができます。

プロセスとパイプを使ったオリジナルのバージョンと比べると、チャンネルと使っ

た方法には利点が二つあります。一つ目はスレッドの起動はプロセスのものと比べてコ

ストがかからないことです。二つ目はチャンネルを使った通信がデータ全体のコピー

ではなくポインタの受け渡しのみを行うことです。今回の例ではスレッドの数とやり

取りされるデータのサイズがシステムコールと計算のコストに比べて小さいのでこの

利点はわかりにくなっています。

まとめると、スレッド間の通信は (プロセスが中断される場合) システムコールと
同程度のコストがかかりますが、このコストは通信をバッファして大きな構造体をや

り取りすることで大きく削減できるということになります。

* * *
練習問題 22 httpサーバには高く激しい負荷がかかることがあります。応答にか
かる時間を減らすために、新しい接続を処理するためのスレッドを常に数十個保持す

るように httpサーバのアーキテクチャを変更することができます。この変更は各ス
レッドが一つのリクエストを処理するのではなく、キューから読んだリクエストを際

限なく処理することを意味します。

マシンへの負荷を避けるために、スレッドの数をタスクを管理する時間がリクエ

ストに対応する時間 (ディスクのデータ待ちの時間など) よりも小さくなる数に制限
する必要があります。さらに接続が増えた場合は接続を処理待ち状態にしておき、そ

れ以上に増えた場合は接続を拒否します。負荷が減りスレッドの数が “理想的な” 数
よりも大きくなればスレッドのいくつかを削除します。

7.5 節の例をこのアーキテクチャに変更してください。

* * *

7.8 実装の詳細

Unix におけるスレッドの実装 もともと Unix システムはスレッドをサポートする
ようにデザインされていませんでしたが、ほとんどの現代的な Unix 実装はスレッド
をサポートします。ただしスレッドが後から追加されたものであるということが明白

になることがあります。例えば exec を直後に実行するのでない限り、スレッドを使っ

ているときに fork を使うことは強く非推奨とされています。fork は現在のスレッド

をコピーし、そのスレッドは他のスレッドが存在すると信じて実行を続けますが、実

際には他のスレッドは存在せず、正しく動作することができません。

fork を実行した親は通常通り実行を続けます。また fork の直後に子プロセスが

他のプログラムを起動する特殊ケースについては親プロセスに問題は起こりません。

この方法で問題が起きないことはラッキーです。なぜならこれが他のプログラムを起

動する唯一の方法だからです。

順番を逆にして、fork の後に親または子プロセスがスレッドを作成することは問

題ありません。

OCaml におけるネイティブおよびシミュレート実装 実行されているオペレーティ

ングシステムがスレッドの機能を持つ場合には、OCamlはスレッドのネイティブ実装
を使います。このとき OCamlはオペレーティングシステムにスレッドの管理をでき
るだけ任せます。スレッドは同じアドレス空間を共有する異なる Unix プロセスとな
ります。

システムがスレッドをサポートしない場合には、OCamlはスレッドをエミュレート
することができます。全てのスレッドは同じ Unixプロセスで実行されそのスケジュー
リングを含んだ管理はOCamlのランタイムシステムによって行われます。ただし、こ
の実装はバイトコードにコンパイルするときにのみ利用可能です。

OCamlシステムはネイティブとシミュレートされたバージョンのスレッドに同じ
インターフェースを提供しています。そのためスレッドの実装は二つに分かれていま

す。一つはエミュレートされたバージョンのためのもので、タスクコントローラを含

みます。もう一つは posix (1003.1c) に基づくスレッドの実装で、他の言語で書かれ
たライブラリ関数を OCamlへとつなぐものです。OCamlは単純な管理タスクを行い
インターフェースがエミュレートされたバージョンと同じことを確認します。これに

よってある Unix アーキテクチャでコンパイルできたプログラムがほかの Unix アー
キテクチャでもコンパイルできることが保証されます。

ただしネイティブのスレッドとエミュレートされたスレッドでは同期のための C
ライブラリへの呼び出しが異なるために、プログラムの意味全体が異なってしまいま

す。そのためプログラムが二つのバージョンで同じように動くと信じる前にいくつか

対処をしなくてはいけません。この節では主に二つの実装の違いについて議論します

が、通常はネイティブ実装について語ることを覚えておいてください。

エミュレートされたスレッドを使うには ocamlc コンパイラに -thread ではなく

-vmthread オプションを渡します。このオプションを ocamlopt コンパイラに渡すこ

とはできません。

OCaml コードの逐次化 OCamlによるスレッドの実装は自動メモリ管理やアロケー
トされたデータの頻繁な利用といった OCamlの特色を持っている必要があります。採
用されている解法は一番シンプルで最も効率的なもので、OCamlの全てのスレッドの
実行を逐次化するというものです。つまりランタイムシステムが持つロックによって

二つ以上の OCamlコードを同時に実行できなくします。これはスレッドという考え方
と矛盾するように聞こえますが、そうではありません。ロックはブロックするシステ

ムコールの直前に開放され、システムコールから返るともう一度取得されるからです。

この間に他のスレッドが実行されます。このようなシステムコールの特殊なケースは

実行中のスレッドで定期的に実行され、他のスレッドに実行を受け渡す sched_yield

の呼び出しです。

マルチプロセッサのマシンでは、本当の並列性は C のコードとシステムコールを
実行することでしか得ることができません。プロセッサが一つしか無いマシンでは、

OCamlコードが逐次化されたということに気づくことはできません。
スレッドの実行はほぼ任意のタイミングで他のスレッドに移るので、プログラマ

はこの逐次化を頼ることはできません。ただし例外として、逐次化されたコードでは

メモリの一貫性が保証されます。つまり、C のコードを実行するときなどを除いて二
つのスレッドは同じメモリを目にするということです。その結果として、ロックの受

け渡しがメモリの同期を意味するようになります。あるスレッドが書き込んだアドレ

スを別のスレッドが読み込んだ場合常に新しい値が読み込まれ、同期処理は必要あり

ません。

スレッドとシグナル 一般的に言って、非同期という特性を持つシグナルを使うこと

はシングルスレッドのプログラムでさえ注意を要する難しいものでした。シグナルは

マルチスレッドのプログラムではさらに難しくなります。どのスレッドにシグナルが

送られるべきでしょうか? 全てのスレッド、主スレッド、それとも現在実行している
スレッド? あるスレッドが別のスレッドにシグナルを送りたい場合どうすればいいで
しょうか? 実際のところ、スレッドはこのような問題に答えることなく実装されまし
た。そのため実装によってシグナルへの動作は異なります。

Thread.join と Mutex.lock そして Condition.wait は遅いシステムコールにもか

かわらずシグナルによって中断されることがありません (したがって EINTR エラーを

出して失敗することもありません)。待っているときにシグナルが送られた場合、その
シグナルはシステムコールから返ってから受け取られて処理されます。

posix 規格はシグナルハンドラがスレッド間で共有されることとシグナルマスク
はスレッドの作成時にもとのスレッドから継承された後はスレッドにプライベートと

なることを定めています。しかしシグナルに対するスレッドの動作は主に定められて

おらず、ポータブルにはなっていません。

したがってスレッドではできる限り非同期シグナル (sigalrmや sigvtalrm、sigchld

など) を避けることが望ましいです。シグナルは Thread.wait_signal によってブロッ

クと調査をすることができます。シグナルの処理だけを行うスレッドを作ることもで

きます。このスレッドはシグナルを待ち、適切な処理を行い、他のスレッドにシグナ

ルの情報を伝えます。

加えて、(バージョン 3.08以降の) OCamlのスレッドはスレッドのプリエンプショ
ンに内部で sigvtalarm を使います。そのためこのシグナルを使うとシステムと干渉

する可能性があるために、sigvtalarm シグナルはプログラムの中で使うことができま

せん。

さらに先へ

これまで見てきたのはオペレーティングシステムとやり取りをするアプリケーション

を作るために OCamlの Sys、Unix そして Threads モジュールを使う方法です。

これらのモジュールを使うと重要な Unixシステムコールのほとんどを OCamlか
ら直接呼び出すことができます。プログラミングを容易にするため、あるいは OCaml
のランタイムシステムが必要とする不変量を保つために、いくつかのシステムコール

は高レベルな関数に置き換えられています。いずれの場合でも、この Unix システム
への高レベルなアクセスはアプリケーションの開発を円滑にします。

全ての Unix の機能がこれらのモジュールから利用可能なわけではありません。し
かし C のバインディングを書けば利用できない機能にアクセスすることが可能です。
別の便利なライブラリは OCaml から直接スクリプトを書くことを目的とした

Cash [3] です。このライブラリは Unix を二つの方法で補完します。一つ目はスク

リプトを書くためのヘルパー関数に加えて、Unix モジュールの上にプロセスとパイ

プの管理を助けるためのシステムコールの亜種を用意していることです。もう一つは

Unix システムに追加のエントリーポイントを追加することです。

練習問題の回答

練習問題 1 (16 ページ)
let main () =

let action p infos =

let b = not (infos.st_kind = S_DIR || Filename.basename p = "CVS") in

if b then print_endline p; b in

let errors = ref false in

let error (e,c,b) =

errors:= true; prerr_endline (b ^ ": " ^ error_message e) in

Findlib.find error action false max_int [ "." ];;

handle_unix_error main ()

練習問題 2 (16 ページ) ヒントを示します。現在の場所からルートに向かって登っ

て行くことで、答えとなるパスを作ることができます。ルートは親がそれ自身と等し

い唯一のディレクトリノードとして検出できます (ルートからの相対パス . と .. は

等しいです)。ディレクトリ r の名前を見つけるには、親ディレクトリの要素を一覧

で表示し r に対応するものを見つければよいです。

練習問題 3 (21 ページ) -a オプションが与えられた場合、行うべき処理が

openfile output_name [O_WRONLY; O_CREAT; O_APPEND] 0o666

ではなく

openfile output_name [O_WRONLY; O_CREAT; O_TRUNC] 0o666

となります。新しいオプションをコマンドラインからパースする部分は読者に残し

ます。

練習問題 4 (26 ページ) 書き込む文字列をバッファにコピーするのが基本的なアイ

デアです。バッファに十分な空きが無く途中でバッファを空にする必要がある場合と

そうでなく文字列全てをバッファに直接書き込める場合とを考える必要があります。

解答例を以下に示します。

open Unix;;

let output_string chan s =

let avail = Bytes.length chan.out_buffer - chan.out_pos in

if Bytes.length s <= avail then begin

Bytes.blit s 0 chan.out_buffer chan.out_pos (Bytes.length s);

chan.out_pos <- chan.out_pos + Bytes.length s

end

else if chan.out_pos = 0 then begin

ignore (write chan.out_fd s 0 (Bytes.length s))

end

else begin

Bytes.blit s 0 chan.out_buffer chan.out_pos avail;

let out_buffer_size = Bytes.length chan.out_buffer in

ignore (write chan.out_fd chan.out_buffer 0 out_buffer_size);

let remaining = Bytes.length s - avail in

if remaining < out_buffer_size then begin

Bytes.blit s avail chan.out_buffer 0 remaining;

chan.out_pos <- remaining

end else begin

ignore (write chan.out_fd s avail remaining);

chan.out_pos <- 0

end

end;;

let ex2 () =

if Array.length Sys.argv < 3 then begin

prerr_string "Usage: test <sources> <dest>";

exit 2;

end;

let fdin = open_in Sys.argv.(1) in

let fdout = open_out Sys.argv.(2) in

prerr_endline "copying";

try while true do output_char fdout (input_char fdin) done

with End_of_file ->

prerr_endline "Done";

output_string fdout "The end.\n";

prerr_endline "Closing";

close_out fdout;;

handle_unix_error ex2 ();;

./ex2.byte ex2.ml ex2.out

(cat ex2.ml; echo "C'est la fin.") | diff --brief - ex2.out

rm ex2.out

練習問題 5 (27 ページ) tail のナイーブな実装は最後に呼んだ n 行を巡回バッファ
に記録しながらファイルを最初から読み、ファイルの末尾に達したところでバッファ

を表示するというものです。lseek の実装されないパイプやスペシャルファイルから

データが来るならば、これが最善の方法になります。

もしデータが通常ファイルから来るならば、ファイルの末尾から読み込むほうが

効率が良くなります。まず lseek を使って最後の 4096 文字を読み込み、その中に改
行がないか調べます。n 個以上の改行があるなら対応する行を出力しそうでなければ
次の 4096 バイトを読み、同じことをします。

-f オプションを追加するにはまずオプション無しの場合の動作を行い、そのあと

ファイルの末尾に移動してから read します。read が返ったときはその内容をすぐに

画面に表示しもう一度 read します。もし read が 0 を返したならプログラムは一定

時間待ってから (sleep 1) もう一度試します。

練習問題 6 (31ページ) すでにコピーされたファイルについて、識別子 (st_dev, st_ino)

からコピー先へのマップを記録するようにします。各コピーを実行する前にマップを

探索し、同じ識別子を持ったファイルがすでにコピーされていないかをチェックしま

す。もしすでにコピーされたファイルが見つかったらファイルをコピーする代わりに

コピー元ファイルと同じ名前のハードリンクを作成します。マップの大きさを最小化

するために一つ以上の名前を持つファイル、つまり st_nlink > 1 を満たすファイル

のみについてマップを持つようにします。

let copied_files = (Hashtbl.create 53 : ((int * int), string) Hashtbl.t)

let rec copy source dest =

let infos = lstat source in

match infos.st_kind with

S_REG ->

if infos.st_nlink > 1 then begin

try

let dest' =

Hashtbl.find copied_files (infos.st_dev, infos.st_ino)

in link dest' dest

with Not_found ->

Hashtbl.add copied_files (infos.st_dev, infos.st_ino) dest;

file_copy source dest;

set_infos dest infos

end else begin

file_copy source dest;

set_infos dest infos

end

| S_LNK -> ...

練習問題 7 (37 ページ) 問題は一見単純なように見えますが、シンボリックリンク

はハードリンクと違ってディレクトリを含んだ任意のパスを表すことができ、そのパ

スがアーカイブの中に存在するとは限らないことから、簡単ではありません。

シンプルな解決法はアーカイブに含まれるファイル階層をメモリ上に再構築する

ことです。

type info = File | Link of string list | Dir of (string * inode) list

and inode = { mutable record : record option; mutable info : info;}

メモリ上に構築されるファイルシステムのノードは inode型で表されます。infoフィー

ルドが通常ファイルとシンボリックリンクとディレクトリに制限されたファイルの種

類を表します。パスは文字列のリストとして、ディレクトリは含まれるファイルの名

前とノードの組のリストとして表されます。record フィールドはノードに対応する

tar レコードを保持します。中間のディレクトリはアーカイブ内にレコードを持たな

い場合があるので、このフィールドはオプショナルです。ファイルはアーカイブ内に

複数回現れることがあり、最後に現れたものが優先されるので info フィールドは変

更可能になっています。

let root () =

let rec i =

{ record = None; info = Dir [ Filename.current_dir_name, i ] }

in i

let link inode name nod = match inode.info with

| File | Link _ -> error name "Not a directory"

| Dir list ->

try let _ = List.assoc name list in error name "Already exists"

with Not_found -> inode.info <- Dir ((name, nod) :: list)

let mkfile inode name r =

let f = { record = r; info = File } in

link inode name f; f

let symlink inode name r path =

let s = { record = r; info = Link path } in

link inode name s; s

let mkdir inode name r =

let d = mkfile inode name r in

d.info <-

Dir [ Filename.current_dir_name, d; Filename.parent_dir_name, inode ];

d

Unix と同様に各ディレクトリは自分自身と親に対するリンクを持ちます。ただしルー
トディレクトリの親が自分自身となる Unixとは違い、いま考えているファイルシステ
ムではルートディレクトリは親を持ちません。これによってアーカイブの外側のファ

イルに対するアクセスを検出しやめさせることができます。

let rec find link inode path = match inode.info, path with

| _, [] -> inode

| Dir list, name :: rest ->

let subnode = List.assoc name list in

let subnode =

match subnode.info with

Link q ->

if link && rest = [] then subnode else find false inode q

| _ -> subnode in

find link subnode rest

| _, _ -> raise Not_found;;

find関数はアーカイブから pathに対応するノードを inodeを始点として探索します。

引数 link は探索がリンクを返した場合にリンクそのものを返すべきなのか (true) そ
れともリンクが指すファイルを返すべきなのか (false) を決めます。
let rec mkpath inode path =

match inode.info, path with

| _, [] -> inode

| Dir list, name :: rest ->

let subnode =

try List.assoc name list

with Not_found -> mkdir inode name None in

mkpath subnode rest

| _, _ -> raise Not_found;;

mkpath 関数はパス path をたどってパス上の存在しないノードを作成します。

let explode f =

let rec dec f p =

if f = Filename.current_dir_name then p

else dec (Filename.dirname f) (Filename.basename f :: p) in

dec (if Filename.basename f = "" then Filename.dirname f else f) [];;

ml 関数は Unix のパスをパースして文字列のリストにします。このときアーカイブで
は許されているディレクトリ末尾の “/” を削除します。

let add archive r =

match r.header.kind with

| CHR (_,_) | BLK (_,_) | FIFO -> ()

| kind ->

match List.rev (explode r.header.name) with

| [] -> ()

| name :: parent_rev ->

let inode = mkpath archive (List.rev parent_rev) in

match kind with

| DIR -> ignore (mkdir inode name (Some r))

| REG | CONT -> ignore (mkfile inode name (Some r))

| LNK f -> ignore (symlink inode name (Some r) (explode f))

| LINK f -> link inode name (find true archive (explode f))

| _ -> assert false;;

add 関数はレコード r をアーカイブに追加します。ルートノードで表されるアーカイ

ブは副作用で変更されます。

let find_and_copy tarfile filename =

let fd = openfile tarfile [ O_RDONLY ] 0 in

let records = List.rev (fold (fun x y -> x :: y) [] fd) in

let archive = root () in

List.iter (add archive) records;

let inode =

try find false archive (explode filename)

with Not_found -> error filename "File not found" in

begin match inode.record with

| Some ({ header = { kind = (REG | CONT) }} as r) -> copy_file r stdout

| Some _ -> error filename "Not a regular file"

| None -> error filename "Not found"

end;

close fd;;

最後はこれまでと同じです。

let readtar () =

let nargs = Array.length Sys.argv in

if nargs = 2 then list Sys.argv.(1)

else if nargs = 3 then find_and_copy Sys.argv.(1) Sys.argv.(2)

else prerr_endline ("Usage: " ^Sys.argv.(0)^ " <tarfile> [ <source> ]");;

Printexc.print (handle_unix_error (handle_error readtar)) ();;

練習問題 8 (37 ページ) この練習問題では一つ前の練習問題 2.16と再帰的なファイ
ルのコピーについての練習問題 2.15 を組み合わせます。
すこし難しい部分は権限の管理です。アーカイブのディレクトリを書き込み権限

で作成して全てのファイルを抽出し終わってから本来の権限に設定しなければいけま

せん。

最初に補助関数 mkpath p m を書きます。この関数はパス p に含まれるディレク

トリでまだ作成されていないものを、m の権限で作成します。p の最後には余分な “/”
が含まれていても構いません。

let warning mes = prerr_string mes;prerr_newline ();;

open Filename

let mkpath p perm =

let normal_path =

if basename p = "" then dirname p else p in

let path_to_dir = dirname normal_path in

let rec make p =

try ignore (stat p)

with Unix_error (ENOENT, _, _) ->

if p = current_dir_name then ()

else if p = parent_dir_name then

warning "Ill formed archive: path contains \"..\""

else begin

make (dirname p);

mkdir p perm

end in

make path_to_dir;;

2.15 節でファイルをコピーするときに利用したものと似た set_infos を定義します。

let set_infos header =

chmod header.name header.perm;

let mtime = float header.mtime in

utimes header.name mtime mtime;

begin match header.kind with

| LNK f -> ()

| _ -> chmod header.name header.perm

end;

try chown header.name header.uid header.gid

with Unix_error(EPERM,_,_) -> ();;

untar_file_collect_dirs は作成されるディレクトリを記録しながら一つのレコード

をコピーします。

let verbose = ref true;;

let default_dir_perm = 0o777;;

let default_file_perm = 0o666;;

let protect f x g y = try f x; g y with z -> g y; raise z

let file_exists f = try ignore (stat f); true with _ -> false;;

let untar_file_collect_dirs file dirs =

let fh = file.header in

if !verbose then begin print_string fh.name; print_newline () end;

match fh.kind with

| CHR (_,_) | BLK(_,_) | FIFO ->

warning (fh.name ^ "Ignoring special files");

dirs

| DIR ->

mkpath fh.name default_dir_perm;

if file_exists fh.name then dirs

else begin mkdir fh.name default_dir_perm; fh :: dirs end

| x ->

mkpath fh.name default_dir_perm;

begin match x with

| REG | CONT ->

let flags = [ O_WRONLY; O_TRUNC; O_CREAT; ] in

let out = openfile fh.name flags default_file_perm in

protect (copy_file file) out close out

| LNK f ->

symlink f fh.name

| LINK f ->

begin

try if (stat fh.name).st_kind = S_REG then unlink fh.name

with Unix_error(_,_,_) -> ();

end;

Unix.link f fh.name;

| _ -> assert false

end;

set_infos fh;

dirs;;

メインプログラムは untar_file_collect_dirs を全てのレコードに適用し、最後に

ディレクトリのアクセス権限を修正するだけです。

let extract tarfile =

let fd = openfile tarfile [ O_RDONLY ] 0 in

let new_directories =

fold untar_file_collect_dirs [] fd in

List.iter set_infos new_directories;

close fd;;

let untar () =

let nargs = Array.length Sys.argv in

if nargs = 2 then extract Sys.argv.(1)

else prerr_endline ("Usage: " ^ Sys.argv.(0) ^ " <tarfile>");;

handle_unix_error untar ();;

練習問題 9 (37 ページ) 一つ前の練習問題で定義したデータ構造を再利用するため

に Tarlib モジュールに集めておきます。プログラムを停止したりプログラムの返り

値を変更したりせずに警告を表示する関数を定義します。

open Sys

open Unix

open Tarlib

let warning path message = prerr_endline (path ^ ": " ^ message)

レコードのヘッダをバッファに書き込む関数から始めます。処理は退屈ですが、ヘッ

ダがおかしいとアーカイブ全体が読み込めなくなるので慎重になる必要があります。

特にファイルのフォーマットによる制限には注意が必要です。例えばパスの長さは 99
バイトに制限されています (これより長いパスを扱えるようにするフォーマットの拡
張もありますが、この練習問題では取り扱いません)。
let write_header_to_buffer source infos kind =

let size = if kind = REG then infos.st_size else 0 in

Bytes.fill buffer 0 block_size '\000';

let put len string offset =

Bytes.blit string 0 buffer offset (min (Bytes.length string) len) in

let put_int8 x = put 7 (Printf.sprintf "%07o" x) in

let put_int12 x = put 11 (Printf.sprintf "%011o" x) in

let put_char c offset = buffer.[offset] <- c in

let put_path s offset =

if Bytes.length s <= 99 then put 99 s offset

else raise (Error ("path too long", s)) in

put_path (if kind = DIR then source ^ "/" else source) 0;

put_int8 infos.st_perm 100;

put_int8 infos.st_uid 108;

put_int8 infos.st_gid 116;

put_int12 size 124;

put_int12 (int_of_float infos.st_mtime) 136;

put 7 "ustar " 257;

put 31 (getpwuid infos.st_uid).pw_name 265;

put 31 (getgrgid infos.st_gid).gr_name 297;

(* Fields dev and rdev are only used for special files, which we omit *)

put_char

begin match kind with

| REG -> '0'

| LINK s -> put_path s 157; '1'

| LNK s -> put_path s 157; '2'

| DIR -> '5'

| _ -> failwith "Special files not implemented"

end 156;

let rec sum s i =

if i < 0 then s else sum (s + Char.code buffer.[i]) (pred i) in

let checksum = sum (Char.code ' ' * 8) (block_size - 1) in

put 8 (Printf.sprintf "%06o\000 " checksum) 148;;

次の関数はファイルに対するレコードのバッファを作成します。source はファイルの

名前、infos は stats で取得できるファイルの情報、kind はファイルの種類です。

let header source infos kind = {

name = source;

size = if kind = REG then infos.st_size else 0;

perm = infos.st_perm;

mtime = int_of_float infos.st_mtime;

uid = infos.st_uid;

gid = infos.st_gid;

user = (getpwuid infos.st_uid).pw_name;

group = (getgrgid infos.st_gid).gr_name;

kind = kind }

アーカイブにファイルの内容を書き込むために、file_copy に似た関数を定義します。

この関数は引数としてコピーされるバイト数を受け取り、ファイルの末尾がそのサイ

ズと対応していることを確認します。受け取ったバイト数だけ書き込んだ結果ファイ

ルの末尾に到達しなかった場合、エラーを出します。これによってアーカイブ中にファ

イルが変更されるケースに対応できます。アーカイブの欠損を一つのファイルに抑え

るために、引数で受け取ったバイト数を超えて書き込むことはしません。

let write_file len source fdout =

let fdin = openfile source [O_RDONLY] 0 in

let error () = raise (Error ("File changed size", source)) in

let rec copy_loop len =

match read fdin buffer 0 buffer_size with

0 ->

close fdin; if len > 0 then error ()

| r ->

let len = len - r in

if len < 0 then (close fdin; error ());

ignore (write fdout buffer 0 r); copy_loop len in

copy_loop len;;

let padding fd len =

if len > 0 then ignore (write fd (Bytes.make len '\000') 0 len);;

ここからがアーカイブの作成になります。アーカイブに書き込まれたファイルはその

パスと共にハッシュテーブルに保存され、同じファイルが何度もコピーされることが

無いように利用されます。またファイルだけではなくディレクトリについてもすでに

書き込んだパスをハッシュテーブルを保存します。アーカイブのルートが他のディレ

クトリに含まれることがありえますが、そのような場合にコピーを行わないようにす

るためです (ただしコピーをしても問題はありません)。
アーカイブを書き込むのに必要となるのは書き込みファイルを指すファイルディ

スクリプタとファイルとディレクトリのキャッシュ (一つ前の練習問題参照)、そして
現在のアーカイブのサイズを記録する変数 (必要な場合に最小サイズにパディングす
るため) です。archive 型がこれらの情報を保持するレコードです:

type archive =

{ regfiles : (int * int, string) Hashtbl.t;

dirfiles : (int * int, bool) Hashtbl.t;

fd : file_descr; st : stats; mutable size : int }

let try_new_dir archive dir =

try Hashtbl.find archive.dirfiles dir

with Not_found -> Hashtbl.add archive.dirfiles dir false; true

コマンドラインで与えられるパス file から始まるファイル階層全体を書き込む関数

は以下のようになります。この関数は難しくありませんが、いくつか例外的なケースが

あります。例えばファイルがアーカイブ中に変更されたことを検出する方法について

は前に示しました。このケースの特別な場合は、アーカイブがアーカイブ自身をアー

カイブしようとしている場合です。

let verbose = ref true;;

let write_from archive file =

if not (Filename.is_relative file) then

raise (Error ("absolute path", file));

let rec write_rec archive file =

let source =

if Filename.basename file = "" then Filename.dirname file else file in

if !verbose then begin prerr_endline source end;

let st = lstat source in

if st.st_ino = archive.st.st_ino && st.st_dev = archive.st.st_dev

then warning source "Skipping archive itself!"

else

let write_header kind =

write_header_to_buffer source st kind;

ignore (write archive.fd buffer 0 block_size) in

match st.st_kind with

S_REG ->

begin try

if st.st_nlink = 1 then raise Not_found;

let path =

Hashtbl.find archive.regfiles (st.st_ino, st.st_dev) in

write_header (LINK path);

with Not_found ->

if st.st_nlink > 1 then

Hashtbl.add archive.regfiles (st.st_ino, st.st_dev) source;

write_header REG;

write_file st.st_size source archive.fd;

let t =

(block_size-1 + st.st_size) / block_size * block_size in

padding archive.fd (t - st.st_size);

archive.size <- archive.size + t + block_size;

end

| S_LNK ->

write_header (LNK (readlink source));

| S_DIR when try_new_dir archive (st.st_ino, st.st_dev) ->

write_header DIR;

Misc.iter_dir

begin

fun file ->

if file = Filename.current_dir_name then ()

else if file = Filename.parent_dir_name then ()

else write_rec archive (source ^ "/" ^ file)

end

source

| S_DIR ->

warning source "Ignoring directory already in archive."

| _ ->

prerr_endline ("Can't cope with special file " ^ source) in

write_rec archive file;;

ハードリンクを持つ可能性がある通常ファイルをハッシュテーブル regfile に記録し

ます。リンクを一つしか持たないファイルは記録する必要がありません。

メイン関数は以下のようになります。エラーが出た場合には、アーカイブを削除

したほうが良いでしょう。

let min_archive_size = 20 * block_size;;

let build tarfile files =

let fd, remove =

if tarfile = "-" then stdout, ignore

else openfile tarfile [ O_WRONLY; O_CREAT; O_TRUNC ] 0o666, unlink in

try

let arch =

{ regfiles = Hashtbl.create 13; dirfiles = Hashtbl.create 13;

st = fstat fd; fd = fd; size =0 } in

Array.iter (write_from arch) files;

padding fd (min_archive_size - arch.size);

close fd

with z ->

remove tarfile; close fd; raise z;;

最後にコマンドライン引数をパースする処理を書いて練習問題を終わります。

let usage () =

prerr_endline "Usage: tar -cvf tarfile file1 [ file2 ... ] ";

exit 2;;

let tar () =

let argn = Array.length Sys.argv in

if argn > 3 && Sys.argv.(1) = "-cvf" then

build Sys.argv.(2) (Array.sub Sys.argv 3 (argn-3))

else usage ();;

let _ =

try handle_unix_error tar ()

with Error (mes, s) ->

prerr_endline ("Error: " ^ mes ^ ": " ^ s); exit 1;;

練習問題 10 (44 ページ) コマンドが & で終わっていた場合には親プロセスが wait

を呼ばずに次のループにすぐに移るようにすればよいです。しかし一つ難しいところ

があります:親プロセスは同時に実行される複数の子プロセスを持つことになるので、
wait がそのうちのどれとも同期するようになってしまうことです。そのため同期的に

実行されるコマンドの終了を待っているときには回収した子が終了を待っている子で

あると確認するまで wait を何度も呼ぶ必要があります。

while true do

let cmd = input_line Pervasives.stdin in

let words, ampersand = parse_command_line cmd in

match fork () with

| 0 -> exec_command words

| pid_son ->

if ampersand then ()

else

let rec wait_for_son () =

let pid, status = wait () in

if pid = pid_son then

print_status "Program" status

else

let p = "Background program " ^ (string_of_int pid) in

print_status p status;

wait_for_son () in

wait_for_son ()

done

練習問題 12 (58 ページ) 親プロセスは子プロセスを待たなくてはいけませんが、

そのまえに子プロセスが読み込むパイプへの入力を親が閉じる必要があります。これ

を閉じなかった場合子プロセスは次の整数が来るのを永遠に待つことになり、デッド

ロックを引き起こします (チャンネルを閉じると対応するディスクリプタのバッファ
が空になるので、データが失われることはありません)。具体的には sieve 関数の 10
行目を以下のようにします:

let output = out_channel_of_descr fd_out in

generate len output;

close_out output;

ignore(waitpid [] p);;

そして filter関数の 12–16行目 (...で表されている部分)を次のコードで覆います。

try

...

with End_of_file ->

close_out output;

ignore (waitpid [] p)

練習問題 13 (58 ページ) 子プロセスは親プロセスの完全なコピーであることから、

標準ライブラリの i/oバッファは fork のときに複製されます。そのため各書き込み

ごとにバッファを空にしない場合、fork の直前にバッファを空にしておく必要があり

ます。read_first_prime 関数の 14 行目の後に flush Pervasives.stdout を追加すれ

ばよいです。

練習問題 14 (61 ページ)

• >> については、ファイルを開くときのフラグを [O_WRONLY; O_APPEND; O_CREAT] に

変えれば他は > と同じです。

• 2> については、dup2 fd stdout を dup2 fd stderr に変えれば他は > と同じです。

• 2>1については、コマンドを実行する前に dup2 stderr stdoutを実行すればよいです。

• << については、シェル sh は /tmp フォルダに一時的なファイルを作って << に続く文

字列を保存し、標準入力をこのファイルにリダイレクトしてからコマンドを実行する

必要があります。これとは別の解法はコマンドの標準入力をパイプの出力につなぎ、

子プロセスを使ってそのパイプの入力に << に続く文字列を入力することです。

練習問題 18 (98 ページ) クイックソートは目的の配列を分割し、分割した配列を

それぞれ独立にソートするので並列化はしやすいです。クイックソートを実行するス

レッドにおける唯一の同期は分割した配列のソートが終わるのを待つ部分だけです。

1 let qsort cmp arr =

2 let rec qsort lo hi =

3 if hi - lo > 0 then

4 begin

5 let mid = (lo + hi) lsr 1 in

6 if cmp arr.(mid) arr.(lo) then swap arr mid lo;

7 if cmp arr.(hi) arr.(mid) then

8 begin

9 swap arr mid hi;

10 if cmp arr.(mid) arr.(lo) then swap arr mid lo

11 end;

12 let pivot = arr.(mid) in

13 let i = ref (lo + 1) and j = ref (hi - 1) in

14 while !i < !j do

15 while not (cmp pivot arr.(!i)) do incr i done;

16 while not (cmp arr.(!j) pivot) do decr j done;

17 if !i < !j then swap arr !i !j;

18 done;

19 let u = Thread.create (qsort lo) (!i-1) in

20 let v = Thread.create (qsort (!i+1)) hi in

21 Thread.join u;

22 Thread.join v

23 end in

24 qsort 0 (Array.length arr - 1);;

20 行目と 21 行目を入れ替えても正しい答えは得られますが、あまり意味がないで
しょう。配列の前半分がソートされてから後半分のソートが始まるので、動作はシー

ケンシャルなプログラムと同じになりますが、スレッドの生成の分だけ実行が遅くな

ります。

並列クイックソートを実際のプログラムで使う場合には、配列の長さがある程度

よりも小さくなったらスレッドを作らずにシーケンシャルにソートしたほうが良いで

しょう。

練習問題 21 (104 ページ) 新しい条件変数 non_full と size フィールドを追加し

ます。

type 'a t =

{ queue : 'a Queue.t; size : int; lock : Mutex.t;

non_empty : Condition.t; non_full : Condition.t; }

let create k =

if k > 0 then

{ queue = Queue.create (); size = k; lock = Mutex.create ();

non_empty = Condition.create (); non_full = Condition.create () }

else failwith "Tqueue.create: empty size";;

要素の追加は add と take の処理を組み合わせたものになります。

let add x q =

Mutex.lock q.lock;

while Queue.length q.queue = q.size

do Condition.wait q.non_full q.lock done;

if Queue.is_empty q.queue then Condition.broadcast q.non_empty;

Queue.add q x;

Mutex.unlock q.lock;;

要素の削除は追加と似ていますが、削除する前のキューが満杯な場合には non_full を

シグナルする必要があります。size を max_int とすれば動作が制限のないキューと

同じになります。

ボールド体のリンクは posix システムコールの説明へのリンクです。

accept, 76
access, 13
alarm, 46
argv, 3
at_exit, 4
bind, 76, 79
broadcast, 103
chdir, 8
chmod, 13
choose, 105
chown, 10
chroot, 8
clear_nonblock, 59
closedir, 14
close, 20, 74, 79
connect, 73, 81
create, 95
delay, 98
dup2, 59
dup, 61
environment, 3
error_message, 5
establish_server, 82
execve, 41, 47
execvp, 41, 47
execv, 41
exit, 3, 95, 95

fork, 38, 47, 94, 108

getgid, 12

getgroups, 10
gethostbyname, 73

getitimer, 52

getpid, 38
getprotobyname, 72
getpwnam, 10
getpwuid, 10
getservbyname, 73
getsockopt, 79
gettimeofday, 52
getuid, 12
gmtime, 53
handle_unix_error, 4
in_channel_of_descr, 57
inet_addr_of_string, 73
input_value, 84
ioctl, 28
join, 96
kill, 46
link, 8
listen, 76
localtime, 39, 53
lockf, 30
lock, 99
lseek, 28
lstat, 9
mkdir, 14
mkfifo, 59
mktime, 53
new_channel, 105
open_connection, 82
opendir, 13, 27
openfile, 27
out_channel_of_descr, 57
output_value, 84
pipe, 54
poll, 106
readdir, 13
readlink, 27
read, 13, 18, 45, 73, 81
receive, 105

recvfrom, 81
recv, 81, 82
rename, 8
rewinddir, 14
rmdir, 14
seek, 26
select, 49, 53, 65, 98
self, 95
sendto, 81
send, 81, 82, 105
set_nonblock, 59
setitimer, 52
setsid, 30
setsockopt, 79
shutdown_connection, 82
shutdown, 74
signal, 46, 103
sigprocmask, 48
sleep, 51
socket, 72
stat, 9, 27
stderr, 9
stdin, 9
stdout, 9
string_of_inet_addr, 73
symlink, 8, 27
sync, 105
system, 49
tcdrain, 30
tcflow, 30
tcflush, 30
tcgetattr, 28
tcsendbreak, 29
tcsetattr, 28
time, 39, 51
truncate, 27, 27
trylock, 99
umask, 13
unlink, 8
unlock, 99

waitpid, 39, 49

wrap_abort, 106

